Proofs without words I

Exercises in METAPOST

Toby Thurston

March 2021 — September 2022



Contents

Geometry and Algebra

```
path s, t;
s = unitsquare shifted -(1/2, 1/2) scaled 144;
t = point 0 of s -- point 2/3 of s -- point -1/3 of s -- cycle;
picture P[];
P2 = image(
    for i=0 upto 3:
        fill t rotated 90i withcolor if odd i: Blues 7 2 else: Oranges 7 2 fi;
        draw t rotated 90i;
    endfor
    draw s;
);
P1 = image(
    fill t withcolor Oranges 7 2; draw t;
    t := t rotatedabout(point 3/2 of t, 180);
    fill t withcolor Oranges 7 2; draw t;
    t := t shifted (point 0 of t - point 2 of t);
    t := t rotatedabout(point 2 of t, -90);
    fill t withcolor Blues 7 2; draw t;
    t := t rotatedabout(point 3/2 of t, 180);
    fill t withcolor Blues 7 2; draw t;
    draw s;
);
draw P1;
draw P2 shifted 200 right;
```

The Pythagorean theorem I



— adapted from the *Chou pei san ching*

```
path s, t;
s = fullcircle scaled 144;
t = (point 4 of s -- point 0 of s -- point sqrt(2) of s -- cycle) shifted point 6 of s;
s := for i=0 upto 3: point 2 of t rotated 90i -- endfor cycle;
picture P[];
P1 = image(
    fill s withcolor Greens 7 1;
    for i=0 upto 3:
        fill t rotated 90i withcolor if odd i: Oranges 7 2 else: Blues 7 2 fi;
        draw t rotated 90i;
    endfor
);
numeric theta; theta = angle (point 2 of t - point 0 of t);
s := s rotated -theta;
t := t rotated -theta;
P2 = image(
    fill s withcolor Greens 7 1; draw subpath (1, 3) of s;
    fill t withcolor Blues 7 2; draw t;
    t := t rotatedabout(point 1/2 of t, 180);
    fill t withcolor Blues 7 2; draw t;
    t := t rotatedabout(point 0 of t, -90);
    fill t withcolor Oranges 7 2; draw t;
    t := t rotatedabout(point 1/2 of t, 180);
    fill t withcolor Oranges 7 2; draw t;
);

label.ulft(P1, 10 left);
label.urt(P2, 10 right);
label.bot("\textit{Behold!}", point 1/2 of bbox currentpicture shifted 36 down);
```

The Pythagorean theorem II



Behold!

— Bhāskara (12th century)

```
path s, t, a, b, c;
s = fullcircle scaled 72;
t = (point 4 of s -- point 0 of s -- point sqrt(6) of s -- cycle) shifted point 6 of s;
a = unitsquare zscaled (point 2 of t - point 0 of t) shifted point 0 of t;
b = unitsquare zscaled (point 1 of t - point 2 of t) shifted point 2 of t;
c = unitsquare zscaled (point 0 of t - point 1 of t) shifted point 1 of t;

color v, w; v = Oranges 7 1; w = Greens 7 1;

picture P[];
P0 = image(draw a; draw b; draw c);
P1 = image(fill a withcolor v; fill b withcolor v; draw P0);
z0 = whatever[point 2 of a, point 3 of a] = whatever[point 2 of b, point 3 of b];
z1 = whatever[z0, point 3 of a]; x1 = xpart point 0 of a;
z2 = whatever[z0, point 2 of b]; x2 = xpart point 1 of b;
path wedge; wedge = subpath (0,1) of a -- subpath (0, 1) of b -- z2 -- z0 -- z1 -- cycle;

P2 = image(
    draw point 2 of a -- z0 -- point 3 of b dashed evenly scaled 1/2;
    path a', b'; numeric t, u;
    t = angle (point 1 of a - point 0 of a);
    u = angle (point 1 of b - point 0 of b);
    a' = a shifted - point 0 of a rotated -t slanted 1/2 rotated t shifted point 0 of a;
    b' = b shifted - point 0 of b rotated -u slanted -1/3 rotated u shifted point 0 of b;
    fill a' withcolor 1/4[v,w]; draw a';
    fill b' withcolor 1/4[v,w]; draw b';
    draw P0
);
P3 = image(
    draw point 2 of a -- z0 -- point 3 of b dashed evenly scaled 1/2;
    fill wedge withcolor 1/2[v,w]; draw wedge; draw point 1 of a -- z0;
    draw P0
);
P4 = image(
    fill wedge shifted (point 0 of a - z1) withcolor 3/4[v,w];
    draw wedge shifted (point 0 of a - z1);
    draw P0
);
P5 = image(fill c withcolor w; draw P0);

draw P1;
draw P2 shifted (144,0);
draw P3 shifted (288,0);
draw P4 shifted (72, -200);
draw P5 shifted (216, -200);
```

The Pythagorean theorem III



— based on Euclid's proof

```
path c, a, a', bq, bq'; numeric r; r = 59;
c = unitsquare shifted -(1/2, 1/2) scaled 160;
a = c scaled cosd(r) rotated r;
pair p, q;
p = whatever[point 0 of a, point 1 of a] = whatever[point 0 of c, point 1 of c];
q = whatever[point 0 of a, point 3 of a] = whatever[point 0 of c, point 3 of c];
bq = point 0 of c -- p -- point 0 of a -- q -- cycle;

fill a withcolor Blues 7 2;
for i=0 upto 3:
    fill bq rotated 90i withcolor if odd i: Oranges 7 2 else: Purples 7 2 fi;
    draw bq rotated 90i;
endfor

a' = a shifted (point 3 of c - point 0 of a);
fill a' withcolor Blues 7 2;
draw a';

bq' = bq rotated 180 shifted (point 1 of a' - point 2 of (bq rotated 180));
pair o; o = point 0 of bq';
for i=0 upto 3:
    fill bq' rotatedabout(o, 90i) withcolor if odd i: Oranges 7 2 else: Purples 7 2 fi;
    draw bq' rotatedabout(o, 90i);
endfor
```

The Pythagorean theorem IV



— H. E. Dudeney (1917)

```
path t, t';
t = (origin -- 377 right -- 144 up -- cycle) scaled 3/4;
t' = t rotated -90 shifted (point 2 of t + point 1 of t rotated 90);

draw unitsquare scaled 8 withcolor 1/2;
draw unitsquare scaled 8 rotated -90 shifted point 0 of t' withcolor 1/2;
draw unitsquare scaled 8 rotated angle (point 1 of t - point 2 of t)
      shifted point 2 of t withcolor 1/2;

draw t;
draw t';
draw point 1 of t -- point 2 of t';

label.lft("$a$", point -1/2 of t);
label.bot("$b$", point 1/2 of t);
label.urt("$c$", point 3/2 of t);
label.top("$a$", point -1/2 of t');
label.lft("$b$", point 1/2 of t');
label.lrt("$c$", point 3/2 of t');

label.bot(btex \vbox{\openup 24pt\halign{\hfil $#$ \hfil\cr
A = 2 \cdot \frac12 ab + \frac12 c^2 = \frac12\left(a+b\right)^2\cr
c^2 = a^2 + b^2\cr}} etex, (xpart point 1 of t, ypart point 2 of t' - 12));
```

The Pythagorean theorem V



$$A = 2 \cdot \frac{1}{2}ab + \frac{1}{2}c^2 = \frac{1}{2}\left(a+b\right)^2$$

$$c^2 = a^2 + b^2$$

— James A. Garfield (1876)

```
numeric r;
r = 144;  z1 = r * dir 66;

draw unitsquare scaled 8 rotated 90 shifted (x1, 0) withcolor 1/2;
draw (left--right) scaled r withcolor Blues 7 7;
draw origin -- z1 -- (x1, 0) withcolor Blues 7 7;
draw fullcircle scaled 2r withcolor Reds 7 7;

label.top("$a$", (1/2 x1, 0));
label.rt("$b$", (x1, 1/2 y1));
label.ulft("$c$", 1/2 z1);
label.top("$c$", (-1/2 r, 0));
label.top("$c-a$", (1/2(r+x1), 0));

label.lft(btex \vbox{\openup 24pt\halign{\hfil $\displaystyle#$ \hfil\cr
\frac{c+a}{b} = \frac{b}{c-a} \cr
a^2 + b^2 = c^2\cr}} etex, point -1/2 of bbox currentpicture + 16 left);
```

The Pythagorean theorem VI



$$\frac{c+a}{b} = \frac{b}{c-a}$$

$$a^2 + b^2 = c^2$$

— Michael Hardy

```
path c; c = fullcircle scaled 377;
z0 = point 4 of c; z1 = point 0 of c; z2 = point 2.828 of c;
z3 = 5/16[z0, z1];
z4 = whatever [z1, z2];
z4 - z3 = whatever * (z2 - z0);
picture P;
P = image(
    draw unitsquare scaled 6 rotated angle (z0 - z2) shifted z2 withcolor 1/2;
    draw unitsquare scaled 6 rotated angle (z3 - z4) shifted z4 withcolor 1/2;
    draw z3 -- z1 -- z2 -- z0 -- z3 -- z4;

    label.bot ("$a$", 1/2[z0, z1] shifted 10 down); label.top("$a'$", 7/16[z3, z1]);
    label.ulft("$b$", 1/2[z0, z2]); label.ulft("$b'$", 1/2[z3, z4]);
    label.urt ("$c$", 1/2[z1, z2]); label.llft("$c'$", 9/16[z1, z4]);

);

draw P shifted 200 up;
x5 = x4; y5 = 0;
draw unitsquare scaled 6 shifted z5 withcolor Blues 7 4;
draw z4--z5 withcolor Blues 7 4;
draw P;
label.bot("$\scriptstyle x$", 1/2[z3, z5]) withcolor Blues 7 6;
label.bot("$\scriptstyle y$", 1/2[z1, z5]) withcolor Blues 7 6;

label.bot(btex \vbox{\openup 8pt\halign{\hfil $\displaystyle # $\hfil\cr
\frac{x}{b'} = \frac{b}{a} \implies \frac{x}{b} = \frac{b'}{a} \implies ax = bb';\cr
\frac{y}{c'} = \frac{c}{a} \implies \frac{y}{c} = \frac{c'}{a} \implies ay = cc';\cr
\therefore\quad aa' = a\left(x+y\right) = bb' + cc'.\cr
}} etex, point 1/2 of bbox currentpicture shifted 24 down);
```

A Pythagorean theorem: $aa' = bb' + cc'$





$$\frac{x}{b'} = \frac{b}{a} \implies \frac{x}{b} = \frac{b'}{a} \implies ax = bb';$$
$$\frac{y}{c'} = \frac{c}{a} \implies \frac{y}{c} = \frac{c'}{a} \implies ay = cc';$$
$$\therefore \quad aa' = a\,(x+y) = bb' + cc'.$$

— Enzo R. Gentile

```
numeric r; r = 64;
numeric pi; pi = 3.141592653589793;
path base, h, c, c', s;

base = (left--right) scaled 7/2r;
h = halfcircle rotated 180 scaled (pi * r + r);
c = fullcircle scaled 2r rotated 90 shifted point 0 of h shifted (0, r);
c' = fullcircle scaled 2r rotated 270 shifted point 4 of h shifted (-r, r);
s = unitsquare scaled (sqrt(pi) * r) rotated -90 shifted point 0 of c';

fill c withcolor Blues 7 1;
fill s withcolor Blues 7 1;

draw base withcolor 1/2;
draw subpath (0, 4 + 1/45 angle point 1 of s) of h withcolor 1/2;
draw subpath (4 + 1/45 angle point 1 of s, 4) of h withcolor Blues 7 3;
draw s;
draw point infinity of h -- point 2 of c' dashed evenly;

forsuffixes $=c, c':
    draw point 0 of $ -- center $ -- point 2 of $ dashed evenly;
    draw $; drawdot point 0 of $ withpen pencircle scaled dotlabeldiam;
endfor

drawarrow subpath (5/4, 1/4) of fullcircle scaled (2r + 16)
    shifted center c withcolor Reds 7 7;
```

The rolling circle squares itself



— Thomas Elsner

```
picture link, pointer, pointer_groove;
color metal, light_metal;
metal = 1/256 (181, 166, 66);
light_metal = 3/4[metal, white];

link = image(
    path a, b, a', b', c;
    a = fullcircle scaled 3; a' = a shifted (98,0);
    b = fullcircle scaled 5; b' = b shifted center a';
    c = subpath(2,6) of b -- subpath(-2,2) of b' -- cycle;
    fill c withcolor light_metal; draw c;
    fill a withcolor metal; draw a;
    fill a' withcolor metal; draw a';
);
pointer = image(
    path a, b, c; numeric r;
    a = fullcircle scaled 18;
    b = fullcircle scaled 24;
    r = 1/3;
    c = subpath(r,8-r) of b --
        point 8-r of b shifted (10cm,0) --
        point 0   of b shifted (116mm,0) --
        point 8+r of b shifted (10cm,0) -- cycle;
    fill c withcolor light_metal;
    fill subpath (9,11) of c -- cycle withcolor 7/8 white;
    draw point 9 of c -- point 11 of c;
    draw c;
    fill a withcolor white; draw a;
);
pointer_groove = image(
    draw pointer;
    path g;
    g = (halfcircle scaled 4 rotated 90 --
         halfcircle scaled 4 rotated 270 shifted (4cm,0) --
         cycle) shifted (5cm,0);
    fill g withcolor 7/8[metal,white]; draw g;
);
draw pointer        rotated 42;
draw pointer_groove rotated 28;
draw pointer_groove rotated 14;
draw pointer        rotated 0;
z0 = 210 right rotated 14;
z1 = 120 right;
numeric t; t = angle (z0-z1);
draw link rotated t shifted z1 rotatedabout(z0,-34.5);
draw link rotated t shifted z1 rotatedabout(z0,-34.5) rotated 14;
draw link rotated t shifted z1;
draw link rotated t shifted z1 rotated 14;
```

On trisecting an angle



— Rufus Isaacs

```
numeric alpha, beta;
alpha = 144;
beta = 0;
for i=1 upto 9:
    beta := beta if odd i: + else: - fi alpha * (2 ** -i);
    path ray;
    ray = origin -- (130 + 10i) * dir beta;
    draw ray withcolor 3/4;
    if i < 7:
        picture t;
        t = thelabel("$\frac1{" & decimal (2**i) & "}$", origin)
                scaled (1 - i/8) shifted point 1 of ray;
        unfill bbox t; draw t;
    fi
endfor

for i = 0 upto 3:
    draw origin -- 240 dir (i * alpha/3)
        if i mod 3 > 0: dashed evenly fi
        withcolor Reds 6 6;
endfor
draw origin withpen pencircle scaled dotlabeldiam;
label.bot("$\displaystyle \frac13=\frac12-\frac14+\frac18-\frac1{16}+\cdots $",
    point 1/2 of bbox currentpicture shifted 24 down);
```

Trisection in an infinite number of steps



$$\frac{1}{3} = \frac{1}{2} - \frac{1}{4} + \frac{1}{8} - \frac{1}{16} + \cdots$$

— Eric Kincanon

```
picture P[];
pair A, B, C, D, E, F;
A = origin; B = 80 right; C = B rotated 60; D = C rotated 180;
E = 1/2 [B, D]; F = p[A,B] = q[E, C];

path ca, cb;
ca = fullcircle scaled 2 abs (A-B);
cb = ca rotated 180 shifted B;

P0 = image(
    drawoptions(withpen pencircle scaled 1/4 withcolor Blues 7 4);
    draw subpath 1/45(50, 70) of ca; draw subpath 1/45(50, 70) of cb;
    draw subpath -1/45(50, 70) of ca; draw subpath -1/45(50, 70) of cb;
    drawoptions();
);
P9 = image(draw A -- B; dotlabel.ulft("$A$", A); dotlabel.urt("$B$", B));
P1 = image(draw P0; draw P9);
P2 = image(
    draw P0;
    drawoptions(withpen pencircle scaled 1/4 withcolor Blues 7 4);
    draw subpath 1/45(230, 250) of ca;
    drawoptions();
    draw C -- D withcolor 1/2;
    dotlabel.top("$C$", C);
    dotlabel.llft("$D$", D);
    draw P9;
);
P3 = image(
    draw B--D withcolor 1/2;
    draw A -- C reflectedabout(A,B) dashed evenly withcolor 1/2;
    draw E withpen pencircle scaled dotlabeldiam;
    label("$E$", E-(2,12));
    draw P2;
);
P4 = image(
    draw B--C dashed evenly withcolor 1/2;
    draw C--E withcolor 1/2;
    draw P3;
    dotlabel.lrt("$F$", F);
);

draw P1;
draw P2 shifted (250, 0);
draw P3 shifted (0, -220);
draw P4 shifted (250, -220);

label.bot("$\overline{AF} = \frac13\cdot\overline{AB}$",
    point 1/2 of bbox currentpicture shifted 36 down);
```

Trisection of a line segment



$$\overline{AF} = \frac{1}{3} \cdot \overline{AB}$$

— Scott Cobel

```
z3 = -z5 = 120 left;
z1 = 180 dir 81;
z2 = 250 dir 130;
z4 =  90 dir -100;

z6 = z5 + 72 right;
z7 = whatever [z2, z5] = whatever [z1, z4];
z8 = whatever [z3, z5] = whatever [z1, z4];
y9 = y1; z9 - z5 = whatever * (z1 - z4);

path star;
star = z3 -- z5 -- z2 -- z4 -- z1 -- cycle;
draw star withcolor Blues 7 7;
draw z6 -- z5 -- z9 dashed evenly withcolor Blues 7 5;

def angle_point(expr a, b, c, r) =
    b + r * (unitvector(a-b) + unitvector(c-b))
enddef;

label("$\alpha$", angle_point(z3, z1, z4, 12));
label("$\beta$", angle_point(z5, z2, z4, 16));
label("$\gamma$", angle_point(z5, z3, z1, 10));
label("$\delta$", angle_point(z1, z4, z2, 12));
label("$\epsilon$", angle_point(z2, z5, z3, 12));

label("$\alpha+\gamma$", angle_point(z1, z8, z5, 16) + 8 down);
label("$\alpha+\gamma$", angle_point(z9, z5, z6, 16) + 8 down);
label("$\beta+\delta$", angle_point(z5, z7, z4, 18) + 1 up);
label("$\beta+\delta$", angle_point(z2, z5, z9, 18) + 2 down);
```

The vertex angles of a star sum to 180°



— Fouad Nakhli

```
pair A', B', C', A, B, C, D, E, F, G, P, Q, R;
A' = origin; B' = 300 right; C' = B' rotated 60; P = 3/8[B', C'];
xpart Q = xpart C';
ypart Q = ypart E = ypart P;
D = whatever[A', B']; xpart D = xpart P;
E = whatever[A', C'];
R = whatever[A', C']; R - P = whatever * (A' - C') rotated 90;
G = whatever[C', Q] = whatever [R, P];
A = whatever[A', B']; G-A = whatever * (C' - A');
B - B' = A - A' = C - C';
F = whatever[B, C]; F-P = whatever * (B-C) rotated 90;

def right_angle_mark(expr a, b, s) =
    subpath (1,3) of unitsquare scaled s rotated angle(b-a) shifted a
enddef;

drawoptions(withcolor 1/2);
draw right_angle_mark(D, B, 6);
draw right_angle_mark(F, P, 6);
draw right_angle_mark(G, P, 6);
draw right_angle_mark(Q, P, 6);
draw right_angle_mark(R, A', 6);
draw E--P;
draw A'--B'--C'--cycle;
drawoptions();

draw P--F withcolor Reds 7 7;
draw R--G--Q withcolor 1/2[Reds 7 7, white];
draw G--P dashed evenly scaled 3/4 withcolor Greens 7 7;
draw G--C' dashed evenly scaled 3/4 withcolor 1/2[Greens 7 7, white];

draw P--D dashed withdots scaled 1/4 withcolor Blues 7 7;
draw A--B--C--cycle;

drawdot P withpen pencircle scaled dotlabeldiam;

forsuffixes $=A, A', B, B', D, Q: label.bot("\strut$" & str $ & "$", $); endfor
forsuffixes $=C, C': label.top("$" & str $ & "$", $); endfor
label.urt("$F$", F);
label("$G$", G + 10 dir 192);

label.top(btex \vbox{\halign{\hss #\hss\cr
The perpendiculars to the sides from a point on\cr
the boundary or within an equilateral triangle\cr
add up to the height of the triangle.\cr
}} etex, point 5/2 of bbox currentpicture shifted 42 up);

label(btex \textit{This shows a particular example, with $C'GQ$ collinear, rather
than the general case} etex, point 1/2 of bbox currentpicture shifted 42 down);
```

Viviani's theorem I

The perpendiculars to the sides from a point on
the boundary or within an equilateral triangle
add up to the height of the triangle.



This shows a particular example, with $C'GQ$ collinear, rather than the general case

— Samuel Wolf

```
def distance(expr a, b, c) = abs ypart ((a-b) rotated -angle (c-b)) enddef;

pair a, b, c, p;
a = 89 up; b = a rotated 120; c = b rotated 120; p = 21 dir 42;

numeric h[];
h0 = distance(a, b, c); h1 = distance(p, a, b);
h2 = distance(p, b, c); h3 = distance(p, c, a);

path t[];
t0 = a--b--c--cycle;
t1 = t0 rotated -120 shifted -point 2 of t0 scaled (h1/h0) shifted p;
t2 = t0 shifted -point 0 of t0 scaled (h2/h0) shifted p;
t3 = t0 rotated +120 shifted -point 1 of t0 scaled (h3/h0) shifted p;

z0 = 1/3[1/2[point 2 of t1, point 1 of t3], point 0 of t0];
z1 = 2/3[point 0 of t1, point 3/2 of t1];

color s[];
s1 = Reds 7 2; s2 = Oranges 7 2; s3 = Blues 7 2;
picture p[];
forsuffixes $=1,2,3: p$ = image(fill t$ withcolor s$; draw t$--point 3/2 of t$); endfor

picture P[];
P1 = image(draw p1; draw p2; draw p3; draw t0;);
P2 = image(
    path cor;
    cor = reverse fullcircle rotated 90 scaled 4/3 h1 scaled 15/16 shifted z1;
    drawarrow subpath 1/45(20, 100) of cor withcolor Reds 7 7;
    drawarrow subpath 1/45(140, 220) of cor withcolor Reds 7 7;
    drawarrow subpath 1/45(260, 340) of cor withcolor Reds 7 7;
    draw p1 rotatedabout(z1, -120); draw p2; draw p3; draw t0);
P3 = image(
    path cor;
    cor = reverse fullcircle rotated 90 scaled 4/3 (h1+h3) scaled 7/8 shifted z0;
    drawarrow subpath 1/45(20, 100) of cor withcolor Reds 7 7;
    drawarrow subpath 1/45(140, 220) of cor withcolor Reds 7 7;
    drawarrow subpath 1/45(260, 340) of cor withcolor Reds 7 7;
    draw point 2 of t1 -- point 1 of t3 dashed withdots scaled 1/2;
    draw p2; draw p1 rotatedabout(z1, -120) rotatedabout(z0, -120);
    draw p3 rotatedabout(z0, -120); draw t0);

draw P1 shifted 160 up;
draw P2 shifted 90 left;
draw P3 shifted 90 right;
label.top(btex \vbox{\halign{\hss #\hss\cr
The perpendiculars to the sides from a point on\cr
the boundary or within an equilateral triangle\cr
add up to the height of the triangle.\cr
}} etex, point 5/2 of bbox currentpicture shifted 42 up);
```

Viviani's theorem II

The perpendiculars to the sides from a point on
the boundary or within an equilateral triangle
add up to the height of the triangle.



— Ken-Ichiroh Kawasaki

```
path s, t;
s = unitsquare shifted -(1/2,1/2) scaled 210;
t = subpath (3, 2) of s -- point 1.732 of fullcircle scaled 210
  shifted point 5/2 of s -- cycle;

fill t withcolor Blues 7 1;
fill s withcolor Oranges 7 1;
for i=0 upto 3: draw t rotated 90i; endfor;
draw point 2 of t -- point 2 of t rotated 180;

label.top(btex \vbox{\halign{\hss #\hss\cr
The internal bisector of the right angle of a right\cr
triangle bisects the square on the hypotenuse\cr
}} etex, point 5/2 of bbox currentpicture shifted 42 up);
```

A theorem about right angles

The internal bisector of the right angle of a right
triangle bisects the square on the hypotenuse



— Roland H. Eddy

```
def angle_arc(expr a, o, b, r) =
    fullcircle scaled 2r rotated angle (a-o) shifted o cutafter (o--b)
enddef;

path c; c = fullcircle scaled 180; pair A, B, C, D, E, F;
A = point 4 of c; B = point 0 of c; C = point 1.7 of c; D = (xpart C, ypart A);
E = C rotatedabout(D, -90); F = B rotatedabout(D, -90);

color r, b, g; r = Reds 7 1; g = Greens 7 1; b = Blues 7 1; picture P[];
P1 = image(
    fill A--B--C--cycle withcolor r;
    draw unitsquare scaled 6 rotated angle (A-C) shifted C withcolor 3/4 r;
    draw unitsquare scaled 6 rotated angle (C-D) shifted D withcolor 3/4 r;
    draw angle_arc(D, A, C, 12) withpen pencircle scaled 1 withcolor Reds 7 5;
    draw angle_arc(D, C, B, 12) withpen pencircle scaled 1 withcolor Reds 7 5;
    draw D--C--B--A--C;
    label.bot("$A$", A); label.bot("$B$", B); label.top("$C$", C); label.bot("$D$", D));

P2 = image(
    fill A--D--C--cycle withcolor r; fill A--D--F--cycle withcolor g;
    fill F--D--E--cycle withcolor r; fill C--D--E--cycle withcolor b;
    draw unitsquare scaled 6 rotated angle (C-D) shifted D withcolor 3/4 r;
    draw angle_arc(D, A, C, 12) withpen pencircle scaled 1 withcolor Reds 7 5;
    draw angle_arc(D, E, F, 12) withpen pencircle scaled 1 withcolor Reds 7 5;
    drawarrow subpath (7/4, 1/4) of quartercircle scaled 42 shifted D withcolor Blues 6 6;
    draw A--F--E--C--A--E; draw C--F;
    label.lft ("$A$", A); label.top ("$C$", C); label.llft("$D$", D);
    label.rt  ("$E$", E); label.bot ("$F$", F));

P3 = image(
    fill A--D--C--cycle withcolor r; fill C--D--E--cycle withcolor b;
    z3 = whatever[A,C]; z3 - E = whatever * (A-C) rotated 90;
    begingroup; interim ahangle := 180;
    drawarrow E--z3 dashed evenly scaled 3/4 withpen pencircle scaled 1/4;
    label.urt("$h$", 1/4[z3, E]);
    endgroup;
    draw A--E--C--A; draw C--D;
    label.bot("$A$", A); label.top("$C$", C); label.bot("$D$", D); label.bot("$E$", E));

P4 = image(
    fill A--D--C--cycle withcolor r; fill A--D--F--cycle withcolor g;
    z4 = whatever[A,C]; z4 - F = whatever * (A-C) rotated 90;
    draw z4--F dashed evenly scaled 3/4 withpen pencircle scaled 1/4;
    label.urt("$h$", 1/4[z4, F]);
    draw A--C--F--A--D;
    label.lft ("$A$", A); label.urt ("$C$", C); label.rt  ("$D$", D); label.lrt ("$F$", F));

P5 = image(fill C--D--E--cycle withcolor b; draw D--E--C--D;
    label.top("$C$", C); label.bot("$D$", D); label.bot("$E$", E));

P6 = image(fill A--D--F--cycle withcolor g; draw A--F--D--A;
    label.lft("$A$", A); label.rt("$D$", D); label.lrt("$F$", F));

draw P1 shifted 120 left; draw P2 shifted 120 right;
numeric y; y = -190;
draw P3 shifted (-120, y); label("${}={}$", (12, y+16)); draw P4 shifted (+120, y);
y := y - 112;
draw P5 shifted (-120, y-abs(D-B)); label("${}={}$", (12, y-28)); draw P6 shifted (+120, y);
label("$CD^2 = AD\cdot DB$", point 1/2 of bbox currentpicture shifted 42 down);
```

Area and the projection theorem of a right triangle



$$CD^2 = AD \cdot DB$$

— Sidney H. Kung

```
def angle_arc(expr a, o, b, r) =
    fullcircle scaled 2r rotated angle (a-o) shifted o cutafter (o--b)
enddef;

path C[]; pair O, P, Q, R;
C1 = fullcircle scaled 280; O = point 0 of C1;
C2 = fullcircle scaled 200 shifted O;

numeric t, u;
(t, u) = C1 intersectiontimes C2;
P = point t of C1;
Q = point 8-t of C1;
z0 = whatever[P, P + direction t of C1]; y0 = ypart point 6 of C1;
R = C2 intersectionpoint (z0--P);

draw center C1 -- P -- point 4 of C1 -- O withcolor 7/8;

forsuffixes $=P, Q, R:
    draw O -- $ withcolor Blues 7 6;
endfor
draw 5/4[Q, P] -- 5/4[P, Q] withcolor Reds 7 6;
draw 5/4[P, R] -- 5/4[R, P] withcolor Reds 7 6;

draw angle_arc(O, Q, P, 30);
draw angle_arc(O, Q, P, 28);
draw angle_arc(O, P, R, 30);
draw angle_arc(O, P, R, 28);

draw C1 withcolor 1/2;
draw C2 withcolor 1/2;;

dotlabel.urt("$O$", O);
dotlabel.urt("\strut $P$", P);
dotlabel.lrt("\strut $Q$", Q);
dotlabel.rt("$\;R$", R);

label.top(btex \vbox{\openup6pt\halign{\hss #\hss\cr
If circle $C_1$ passes through the center $O$ of circle $C_2$, the length\cr
of the common chord $\overline{PQ}$ is equal to the tangent segment $\overline{PR}$.\cr
}} etex, point 5/2 of bbox currentpicture shifted 42 up);
```

Chords and tangents of equal length

If circle C_1 passes through the center O of circle C_2, the length of the common chord $\overline{PQ}$ is equal to the tangent segment $\overline{PR}$.



— Roland H. Eddy

```
path xx, ax, hax, haha;
numeric x, a;
x = 89; a = 34;
xx   = unitsquare shifted 1/2 left scaled x shifted 12 up;
ax   = unitsquare shifted 1/2 left xscaled x yscaled -a shifted 12 down;
hax  = unitsquare shifted 1/2 left xscaled x yscaled -1/2 a shifted 12 down;
haha = unitsquare scaled 1/2 a rotated -90 shifted point 1 of xx shifted (8, -8);

picture P[];
P1 = image(
    fill xx withcolor Oranges 7 1; draw xx;
    label.top("$x$", point 5/2 of xx);
    label.lft("$x$", point 7/2 of xx);
    label("${}+{}$", origin);
    fill ax withcolor Blues 7 2; draw ax;
    label.lft("$a$", point 7/2 of ax);
);

P2 = image(
    fill xx withcolor Oranges 7 1; draw xx;
    label("${}+{}$", origin);
    for i=0, 1:
        fill hax shifted (0, -24i) withcolor Blues 7 2;
        draw hax shifted (0, -24i);
    endfor
);

P3 = image(
    fill xx withcolor Oranges 7 1; draw xx;
    hax := hax shifted (point 0 of xx - point 0 of hax);
    fill hax withcolor Blues 7 2; draw hax;
    hax := hax shifted - point 0 of hax rotated 90 shifted point 1 of xx;
    fill hax withcolor Blues 7 2; draw hax;

    fill haha withcolor Blues 7 1;
    draw haha dashed withdots scaled 1/4;
);


draw P1 shifted 144 left;
label("$=$", (-72, 16));
draw P2;
label("$=$", (72, 16));
draw P3 shifted 144 right;

label.top("$x^2 + ax = \left(x + a/2\right)^2 - \left(a/2\right)^2$",
point 5/2 of bbox currentpicture shifted 42 up);
```

Completing the square

$$x^2 + ax = (x + a/2)^2 - (a/2)^2$$



— Charles D. Gallant

```
numeric a, b; a = 89; b = 21; picture P[];
P1 = image(
    fill unitsquare xscaled a yscaled b shifted (0, a) withcolor Greens 7 1;
    fill unitsquare xscaled b yscaled a shifted (a, 0) withcolor Greens 7 1;
    draw (a, 0) -- (a, a+b) dashed withdots scaled 1/4;
    draw (0, a) -- (a+b, a) dashed withdots scaled 1/4;
    draw (a-b, a) -- (a-b, a+b) dashed withdots scaled 1/4;
    draw (a, a-b) -- (a+b, a-b) dashed withdots scaled 1/4;
    draw unitsquare scaled (a+b);
    label.bot("\strut $a$", (1/2a, 0));
    label.bot("\strut $b$", (a+1/2b, 0));
    label.lft("$a$", (0, 1/2a));
    label.lft("$b$", (0, a+1/2b));
);
P2 = image( draw unitsquare scaled (a-b); label.bot("\strut $a-b$", 1/2(a-b, 0)));
P3 = image(
    draw unitsquare scaled a;
    draw unitsquare scaled b shifted (a,a);
    label.bot("\strut $a$", (1/2a, 0));
    label("$b$", (a + 1/2b, a + 1/2b));
);
P4 = image(
    fill unitsquare scaled a withcolor Greens 7 1;
    fill unitsquare scaled b shifted (a,a) withcolor Greens 7 1;
    fill unitsquare scaled (a-b) withcolor background;
    draw (a-b, a) -- (a-b, a-b) -- (a, a-b) dashed withdots scaled 1/4;
    draw (0, a-b) -- (a-b, a-b) -- (a-b, 0);
    draw unitsquare scaled a;
    draw unitsquare scaled b shifted (a,a);
    label.bot("\strut $a-b$", 1/2(a-b, 0));
    label.bot("\strut $b$", (a-1/2b, 0));
    label("$b$", (a + 1/2b, a + 1/2b));
);


draw P1;
numeric x, y; y = 3/4 (a-b);
x := a + b + 14;     label("$+$", (x,y)); x := x + 14; draw P2 shifted (x,0);
x := x + a - b + 16; label("$=$", (x,y)); x := x + 16; draw P3 shifted (x,0);
x := x + a + 14;     label("$+$", (x,y)); x := x + 14; draw P4 shifted (x,0);

label.top("$\left(a+b\right)^2 + \left(a-b\right)^2 = 2\left(a^2 + b^2\right)$",
point 5/2 of bbox currentpicture shifted 42 up);
```

Algebraic areas I

$$(a + b)^2 + (a - b)^2 = 2\left(a^2 + b^2\right)$$



— Shirley Wakin

```
input arrow_label

numeric a, b, c; a = 80; 2b = a; 2c = b;

def make_box(expr p, shade) = image(fill p withcolor shade; draw p) enddef;

path s[]; picture t[];
s1 = unitsquare scaled (a-b-c);                      t1 = make_box(s1, Reds 7 2);
s2 = unitsquare scaled (2c);                         t2 = make_box(s2, Oranges 7 2);
s3 = unitsquare scaled (a-b+c);                      t3 = make_box(s3, YlGn 7 2);
s4 = unitsquare scaled (2b);                         t4 = make_box(s4, Greens 7 2);
s5 = unitsquare scaled (a+b-c);                      t5 = make_box(s5, Blues 7 2);
s6 = unitsquare xscaled (a+b-c) yscaled (a-b+c);     t6 = make_box(s6, Purples 7 2);
s7 = unitsquare xscaled (a-b+c) yscaled (a+b-c);     t7 = make_box(s7, Purples 7 2);

picture P[];
P1 = image(
    draw t4;
    draw t7 shifted point 1 of s4; draw t6 shifted point 3 of s4;
    draw t2 shifted ((1,1) scaled (a+b-c));

    draw t5 shifted (a + b + c + 20, 0);
    draw t3 shifted (2a + 2b + 40, 0);
    draw t1 shifted (3a + b + c + 60, 0);

    arrow_label(origin, 2b * up, "$2b$", -12);
    arrow_label((0, a+b+c), (a+b+c, a+b+c), "\strut$a+b+c$", -12);
    label.rt("$2c$", (a+b+c, a+b));
    label.top("$a+b-c$", (3/2a + 3/2b + 1/2c + 20, a + b - c + 4));
    label.top("$a-b+c$", (5/2a + 3/2b + 1/2c + 40, a - b + c + 4));
    label.top("$a-b-c$", (7/2a + 1/2b + 1/2c + 60, a - b - c + 4));
);
P2 = image(
    draw t5;
    draw t7 shifted point 1 of s5;
    draw t1 shifted (point 2 of s5 - point 3 of s1);
    draw t3 shifted point 2 of s5;
    draw t6 shifted point 3 of s5;

    draw t4 shifted (2a + 30, 0);
    draw t2 shifted (2a + 30 + 2b + 30, 0);

    arrow_label(origin, 2a * right, "$2a$", 9);
    label.bot("\strut$2b$", (2a + 30 + b, 0));
    label.bot("\strut$2c$", (2a + 30 + 2b + 30 + c, 0));
);

label.top(P1, (0, 2a+2b));
label.top(P2, origin);


label.top(btex $\left(a+b+c\right)^2 + \left(a+b-c\right)^2
          + \left(a-b+c\right)^2 + \left(a-b-c\right)^2
          = \left(2a\right)^2 + \left(2b\right)^2 + \left(2c\right)^2$ etex,
point 5/2 of bbox currentpicture shifted 42 up);
```

Algebraic areas II

$$(a+b+c)^2 + (a+b-c)^2 + (a-b+c)^2 + (a-b-c)^2 = (2a)^2 + (2b)^2 + (2c)^2$$





— Sam Pooley and K. Ann Drude

```
input arrow_label
picture P[];
numeric a,b,c,d;
a = sqrt(90); 1.732a = 1.414b; c + d = 3/2a; 1.414c = d;

P1 = image(
    path s[];
    s1 = unitsquare xscaled -(a*a) yscaled -(d*d);
    s2 = unitsquare xscaled  (b*b) yscaled -(d*d);
    s3 = unitsquare xscaled  (b*b) yscaled (c*c);
    s4 = unitsquare xscaled -(a*a) yscaled (c*c);

    fill s1 withcolor Reds 7 1;     draw s1;
    fill s2 withcolor Greens 7 1;   draw s2;
    fill s3 withcolor Oranges 7 1;  draw s3;
    fill s4 withcolor Blues 7 1;    draw s4;

    label("$a^2d^2$", center s1);
    label("$b^2d^2$", center s2);
    label("$b^2c^2$", center s3);
    label("$a^2c^2$", center s4);

    arrow_label(point 3 of s4, point 2 of s4, "$a^2$", 8);
    arrow_label(point 2 of s4, point 1 of s4, "$c^2$", 8);
    arrow_label(point 2 of s3, point 3 of s3, "$b^2$", 8);
    arrow_label(point 1 of s1, point 2 of s1, "$d^2$", 8);

);
% ... and so on for P2, P3, P4, P5, and P6.
draw P1;
draw P2;
draw P3;
draw P4;
draw P5;
draw P6;

def connect_with_arrow(expr a, b) =
    drawarrow (left-- 4 right) scaled 4
              rotated angle (b-a) shifted 1/2[a,b]
              withcolor Blues 5 5;
enddef;
connect_with_arrow(center P1 + 10 right, center P2);
connect_with_arrow(center P2, center P3);
connect_with_arrow(center P3, center P4);
connect_with_arrow(center P4, center P5);
connect_with_arrow(center P5, center P6);

label.top(btex $\left(a^2+b^2\right)\left(c^2+d^2\right)
         = \left(ab + bc\right)^2
         + \left(bd-ac\right)^2 $ etex,
point 5/2 of bbox currentpicture shifted 42 up);
```

Sum of squares identity

$$\left(a^2+b^2\right)\left(c^2+d^2\right) = (ab+bc)^2 + (bd-ac)^2$$



— Diophantus of Alexandria

```
vardef around(expr p, r) =
  if pair p:                         fullcircle scaled 2r shifted p
  elseif path p and (length p = 0): fullcircle scaled 2r shifted point 0 of p
  elseif path p:
    for i = 1 upto length p:
      subpath (i-1, i) of p
      shifted (r * unitvector(direction i-1/2 of p rotated -90)) ..
    endfor
    if not cycle p:
      for i = length p downto 1:
        subpath (i, i-1) of p
        shifted (r * unitvector(direction i-1/2 of p rotated 90)) ..
      endfor
    fi cycle
  fi
enddef;
% k-th n-gonal number...
numeric k, n;  k = 6; n = 6;
path gon[]; for i=2 upto k:
    gon[i] = (origin for j=1 upto n-1: -- dir (240/n*j) endfor -- cycle) scaled 50(i-1);
endfor
numeric r; r = 8;
path a; a = around(origin, r);
fill a withcolor Blues 7 3; draw a;
for i=1 upto n-1:
    a := around(point i of gon2 -- point i of gon[k], r);
    fill a withcolor Oranges 7 1; draw a;
endfor
for i=1 upto n-2:
    a := around(point i+1/2 of gon[3] -- point i+1/(k-1) of gon[k] --
             point i+1-1/(k-1) of gon[k] -- cycle, r);
    fill a withcolor Greens 7 1; draw a;
endfor
for i=2 upto n-2:
    draw origin -- point i of gon[k] dashed evenly;
endfor

for i=2 upto k:
    draw gon[i];
    for j = i-1 upto (n-1)*i:
        draw point j/(i-1) of gon[i] withpen pencircle scaled r;
    endfor
endfor
draw origin withpen pencircle scaled r;

label.top(btex The $k$\textsuperscript{th} $n$-gonal number is $1 +
    \bigl(k-1\bigr)\bigl(n-1\bigr) +
    \frac12\bigl(k-2\bigr)\bigl(k-1\bigr)\bigl(n-2\bigr)$ etex,
    point 5/2 of bbox currentpicture shifted 42 up);
```

Polygonal numbers

The k^{th} n-gonal number is $1 + (k-1)(n-1) + \frac{1}{2}(k-2)(k-1)(n-2)$



— Dave Logothetti

```
input isometric_projection
set_projection(22, -34);

path base, hlid, mlid;
numeric h, a, b; h = 6; b = 7; a = 3;
base = p(0, 0, 0) -- p(0, 0, b) -- p(-b, 0, b) -- p(-b, 0, 0) -- cycle;
hlid  = p(0, h, 0) -- p(0, h, a) -- p(-a, h, a) -- p(-a, h, 0) -- cycle;
mlid  = p(0, b-a, 0) -- p(0, b-a, a) -- p(-a, b-a, a) -- p(-a, b-a, 0) -- cycle;

picture P[];
P1 = image(
    path lid; lid = hlid;
    fill subpath (0, 1) of base -- subpath (1, 0) of lid -- cycle withcolor Blues 8 1;
    fill lid withcolor Blues 8 2;
    drawoptions(dashed withdots scaled 1/2 withcolor 1/2);
    draw subpath (1, 3) of base;
    draw point 2 of base -- point 2 of lid;

    drawoptions(withcolor 1/2);
    numeric t; t = 1/2;
    draw p(-t, 0, 0) -- p(-t, t, 0) -- p(0, t, 0) -- p(0, t, t) -- p(0, 0, t);
    draw p(-t, h, 0) -- p(-t, h-t, 0) -- p(0, h-t, 0) -- p(0, h-t, t) -- p(0, h, t);

    drawoptions();
    draw lid -- point 0 of base;
    draw point 3 of lid -- subpath (-1, 1) of base -- point 1 of lid;

    label.lft("$h$", p(0, 1/2 h, 0));
    label.urt("$a$", point 7/2 of lid);
    label.ulft("$a$", point 1/2 of lid);
    label.lrt("$b$", point 1/2 of base);
    label.llft("$b$", point 7/2 of base);

    label("$P_1$", p(0, -1, 0));
);
% more of the same for P2 and P3...
draw P1 shifted 120 left;
draw P2 shifted 120 right;
draw P3 shifted 240 down;

label.top(btex  $\displaystyle
    V_\frac{P_1} = {h}{b-a}\cdot V_\frac{P_2} = {h}{b-a}\cdot\frac13\left(b^3-a^3\right)
     = \frac{h}{3}\left(a^2+ab+b^2\right)$ etex,
     point 1/2 of bbox currentpicture shifted 42 down);
```

The volume of a frustrum of a square pyramid



$$V_{\underset{=}{P_1}} hb - a \cdot V_{\underset{=}{P_2}} hb - a \cdot \frac{1}{3}\left(b^3 - a^3\right) = \frac{h}{3}\left(a^2 + ab + b^2\right)$$

— *The Moscow Papyrus*, c. 1850 BCE

```
input arrow_label
input isometric_projection
set_projection(18, -32);

numeric r, h, s, tau;
tau = 6.283185307179586;
r * tau = 400 / ipscale;
h = 3/4 r;
s = r +-+ h;

z0 = p(0,0,0); z1 = p(0,0,r); z2 = p(0,r,r); z3 = p(0,r,0); z4 = p(tau * r, 0, 0);

z5 = p(tau * (r-h), h, 0); z6 = p(tau * (r-h), h, h); z7 = p(0, h, r); z8 = p(0, h, 0);

z9 = p(0, 0, 5r); z10 = p(0, r, 5r); z11 = p(0, h, 5r);

z12 = z9 shifted p(2r, 0, 0);
z13 = z10 shifted p(r, 0, 0);
z14 = 1/2[z9, z12];
z15 = z14 shifted p(0, 0, -r);
z16 = z14 shifted p(0, 0, +r);

z17 = z14 + p(-s, h, 0);
z18 = z14 + p(0, h, -s);
z19 = z14 + p(+s, h, 0);
z20 = z14 + p(0, h, +s);

path disc, base, arc, arch;
base = for i=0 upto 11: z14 + p(r*cosd(30i), 0, r*sind(30i)) .. endfor cycle;
disc = for i=0 upto 11: z14 + p(s*cosd(30i), h, s*sind(30i)) .. endfor cycle;
numeric a, b;
a = directiontime down of base;
b = directiontime up of base;

arc = point a of base .. point a of disc .. z13 .. point b of disc .. point b of base;

drawoptions(dashed evenly withpen pencircle scaled 1/4 withcolor 1/2);
draw z1--z9--z12; draw z2--z10--z13; draw z7--z11;
draw z9--z10;

drawoptions(dashed withdots scaled 1/4 withcolor 1/2);
draw z0--z1--z4; draw z1--z2;
draw z14 -- center disc;
fill disc withcolor Blues 7 1;
draw z13 -- center disc -- z19 -- z14;
draw subpath (b, a) of base;
draw subpath (b, a) of disc;

fill z5--z6--z7--z8--cycle withcolor Oranges 7 1;
draw z6--z7--z8;

drawoptions(dashed evenly withpen pencircle scaled 1/4 withcolor 1/2);
draw z11 -- center disc;

drawoptions();
draw z0--z4--z2--z3--z0; draw z3--z4; draw z8--z5--z6;
draw arc;
draw subpath (a, 12 + b) of base;
draw subpath (a, 12 + b) of disc;
% ... you will need to browse the source for the rest of this
% drawing, which is stretching the limits of `isometric_projection`
```

The volume of a hemisphere via Cavalieri's Principle



$$V_S = V_P = \frac{1}{3}r^2 \cdot 2\pi r = \frac{2}{3}\pi r^3$$

— Sidney H. Kung

Trigonometry, Calculus, & Analytic Geometry

```
numeric x, y, z, r;
pair A, B, C, P;

x = 75; y = 46; x + y + z = 180;
r = 180;
A = r * dir (270 - z);
B = r * dir (270 + z);
C - A = whatever * dir x;
C - B = whatever * dir (180-y);
P = whatever[A,B]; C - P = whatever * up;

path am[];
am1 = fullcircle scaled 42 rotated angle (B-A) shifted A cutafter (A--C);
am2 = fullcircle scaled 42 rotated angle (C-B) shifted B cutafter (B--A);
am3 = fullcircle scaled 42 rotated angle (A-C) shifted C cutafter (C--B);
am4 = fullcircle scaled 42 rotated angle A cutafter (origin -- 1/2[A,B]);
forsuffixes $=1,2: draw am$ withcolor 3/4; endfor
forsuffixes $=3,4: draw am$ withcolor Reds 6 5; endfor

draw subpath (1,3) of unitsquare scaled 6 shifted 1/2[A,B] withcolor 3/4;
draw subpath (1,3) of unitsquare scaled 6 shifted P withcolor 3/4;

draw fullcircle scaled 2r withcolor Blues 7 6;
fill fullcircle scaled 2  withcolor Blues 7 6;

draw C--P dashed evenly withcolor 3/4;
draw A--B--C--A--origin--1/2[A,B];

label.urt("$\alpha$", point arctime 3/4 arclength am1 of am1 of am1);
label.lft("$\beta$", point arctime 1/2 arclength am2 of am2 of am2);
label.lrt("$\gamma$", point arctime 1/2 arclength am3 of am3 of am3);
label.llft("$\gamma$", point arctime 1/2 arclength am4 of am4 of am4);
label.urt("$a$", 1/2[B,C]);
label.ulft("$b$", 1/2[C, A]);
label.bot("$c$", 1/2[A, B]);
label.top("$c/2$", 5/16[A, B]);
label.top("$r$", 1/2 A);

label.top("$\sin(\alpha+\beta) = \sin\alpha\cos\beta + \cos\alpha\sin\beta$"
        & "\enspace for \  $\alpha+\beta < \pi$",
   point 5/2 of bbox currentpicture shifted 42 up);

label.bot(btex \vbox{\openup 6pt\halign{\hss # \hss\cr
    $c = a \cos\beta + b \cos\alpha$\cr
    $r=1/2$ \quad $\Longrightarrow$ \quad
    $\sin\gamma = \frac{c/2}{1/2} = c$,\enspace $\sin\alpha=a$,\enspace $\sin\beta=b$\cr
    $\sin\bigl(\alpha+\beta\bigr) = \sin\bigl(\pi - (\alpha+\beta)\bigr) =
    \sin\gamma = \sin\alpha\cos\beta + \sin\beta\cos\alpha$\cr}} etex,
    point 1/2 of bbox currentpicture shifted 12 down);
```

Sine of the sum

$$\sin(\alpha+\beta) = \sin\alpha\cos\beta + \cos\alpha\sin\beta \;\text{ for }\; \alpha+\beta<\pi$$



$$c = a\cos\beta + b\cos\alpha$$

$$r = 1/2 \quad\Longrightarrow\quad \sin\gamma = \frac{c/2}{1/2} = c,\; \sin\alpha = a,\; \sin\beta = b$$

$$\sin(\alpha+\beta) = \sin(\pi-(\alpha+\beta)) = \sin\gamma = \sin\alpha\cos\beta + \sin\beta\cos\alpha$$

— Sidney H. Kung

```
numeric x, y, a, b;
x = 56; y = 42; a = 120 / cosd(x); a * cosd(x) = b * cosd(y);

path t[];
t1 = origin -- b * dir y -- a * dir x -- cycle;
t2 = origin -- (xpart point 1 of t1, 0) -- point 1 of t1 -- cycle;

path a[];
a1 = fullcircle scaled 64 cutafter subpath (2,3) of t1;
a2 = fullcircle scaled 36 cutafter subpath (2,3) of t2;
a3 = fullcircle scaled 64 cutbefore subpath (2,3) of t2 cutafter subpath (2,3) of t1;

picture P[];
P1 = image(
    fill t1 withcolor Blues 8 1;
    pair p; p = whatever[point 2 of t1, point 3 of t1];
    p - point 1 of t1 = whatever * ((point 3 of t1 - point2 of t1) rotated 90);
    drawoptions(withcolor Blues 7 3);
    draw subpath (1,3) of unitsquare scaled 5
        rotated angle (point 3 of t1 - point 2 of t1)
        shifted p withpen pencircle scaled 1/4;
    draw p -- point 1 of t1;
    drawoptions(withcolor Blues 7 5);
    pair q, r;
    q = 7/8 point arctime 1/2 arclength a3 of a3 of a3; r = q + (8, -8);
    draw r .. q dashed withdots scaled 1/4 withpen pencircle scaled 1/4;
    draw a3; label("$\alpha-\beta$", r+(4,-4));
    drawoptions();
    draw t1;
    label.ulft("$a$", point -5/8 of t1);
    label.lrt("$b$", point   5/8 of t1);
);
% .. and so on for P2 and P3
draw P1;
draw P2 shifted 124 right;
draw P3 shifted 280 right;
label("$=$", (128, 60));
label("$-$", (288, 60));

label.bot(btex \vbox{\openup 6pt\halign{\hfil $#$&${}=#$\hfil\cr
\frac12\cdot a\cdot b\sin\bigl(\alpha-\beta\bigr)&\frac12\cdot a\sin\alpha \cdot
b\cos\beta - \frac12\cdot a\cos\alpha\cdot b\sin\beta\cr
        \sin\bigl(\alpha-\beta\bigr)&\sin\alpha\cos\beta -
        \cos\alpha\sin\beta\cr}} etex,
point 1/2 of bbox currentpicture shifted 13 down);
% Browse the source for the lower drawing
```

Area and difference formulas



$$\frac{1}{2} \cdot a \cdot b \sin(\alpha - \beta) = \frac{1}{2} \cdot a \sin \alpha \cdot b \cos \beta - \frac{1}{2} \cdot a \cos \alpha \cdot b \sin \beta$$
$$\sin(\alpha - \beta) = \sin \alpha \cos \beta - \cos \alpha \sin \beta$$



$$\frac{1}{2} \cdot a \cdot b \cos(\alpha - \beta) = \frac{1}{2} \cdot a \cos \alpha \cdot b \cos \beta + \frac{1}{2} \cdot a \sin \alpha \cdot b \sin \beta$$
$$\cos(\alpha - \beta) = \cos \alpha \cos \beta + \sin \alpha \sin \beta$$

— Sidney H. Kung

```
numeric a, b, theta; a = 136; b = 9/16 a; theta = 40;

path A, B, C, arc, Am, Bm;
A = unitsquare scaled a rotated -90;
B = unitsquare scaled b rotated theta;
C = point 3 of A
 -- point 1 of B rotatedabout(point 3 of A, -90)
 -- point 3 of A rotatedabout(point 1 of B, +90)
 -- point 1 of B -- cycle;

z0 = whatever[point 0 of A, point 3 of A]; point 1 of B - z0 = whatever * up;
arc = quartercircle rotated 180 scaled 2 abs(point 1 of B - z0)
      shifted point 1 of B
      cutbefore subpath (0,1) of B;
Am = unitsquare scaled -abs(z0 - point 3 of A) shifted point 3 of A;
Bm = unitsquare scaled abs(point 0 of arc - point 1 of B) rotated theta shifted point 0 of arc;

picture P[];

P1 = image(
  draw subpath (1,3) of unitsquare scaled 6 shifted z0 withcolor 1/2;
  draw z0 -- point 1 of B dashed evenly scaled 1/2;

  draw arc dashed withdots scaled 1/4;

  fill A withcolor Oranges 7 1;
  fill B withcolor Oranges 7 1;
  fill C withcolor Blues 7 1;

  fill Am withcolor Blues 7 1;
  fill Bm withcolor Blues 7 1;

  draw A; draw B; draw C; draw Am; draw Bm;

  label.bot ("$a$", point 3/2 of A);
  label.ulft("$b$", point 5/2 of B);
  label.urt ("$c$", point 3/2 of C);

  label("$\theta$", 16 dir 1/2 theta);
  label.bot("\strut$b\cos\theta$", 1/2 z0);
  draw thelabel.top("$b\sin\theta$", origin) rotated theta shifted point 1/2 of Bm;
);
P2 = image(
    Bm := Bm rotatedabout(point 1 of B, 90-theta);
    draw subpath (1,3) of unitsquare scaled 6 shifted z0 withcolor 1/2;
    forsuffixes $=Am, Bm, C: fill $ withcolor Blues 7 1; draw $; endfor
    label("$\left(a - b \cos\theta \right)^2$", center Am);
    label("$\left(b \sin\theta \right)^2$", center Bm);
    label("$c^2$", center C);
);

draw P1; draw P2 shifted 200 right;
drawarrow 160 right -- 190 right withpen pencircle scaled 2 withcolor Blues 7 5;
label.bot(btex \vbox{\openup 8pt\halign{\hfil $#$&${}=#$\hfil\cr
c^2 &\left(b\sin\theta\right)^2 + \left(a - b\cos\theta\right)^2\cr
    &b^2\sin^2\theta + a^2 - 2ab\cos\theta + b^2\cos^2\theta\cr
    &a^2 + b^2\left(\sin^2\theta + \cos^2\theta\right) - 2ab\cos\theta\cr
    &a^2 + b^2 - 2ab\cos\theta\cr}} etex,
point 1/2 of bbox currentpicture shifted 32 down);
```

The law of cosines I



$$\begin{aligned} c^2 &= (b\sin\theta)^2 + (a-b\cos\theta)^2 \\ &= b^2\sin^2\theta + a^2 - 2ab\cos\theta + b^2\cos^2\theta \\ &= a^2 + b^2\left(\sin^2\theta + \cos^2\theta\right) - 2ab\cos\theta \\ &= a^2 + b^2 - 2ab\cos\theta \end{aligned}$$

— Timothy A. Sipka

```
numeric a, b;  a = 0.98; b = 2.718;
path c; c = fullcircle scaled 421;
z0 = whatever[point a of c, point a+4 of c] = whatever[point 0 of c, point b of c];

fill center c -- point 0 of c -- z0 -- cycle withcolor Greens 7 1;

% mark the angles
draw unitsquare scaled 8 rotated angle (point 4 of c - point b of c)
                shifted point b of c withcolor 3/4;
draw halfcircle scaled 64 shifted point 0 of c
                cutbefore (point 0 of c -- point b of c) withcolor Reds 7 7;
label("$\theta$", 26 dir (180 - 1/4(180 - 45b)) shifted point 0 of c);

draw point a of c -- point a + 4 of c;
draw point 4 of c -- point 0 of c -- point b of c -- cycle;
draw c withcolor Blues 7 7;
draw center c withpen pencircle scaled dotlabeldiam withcolor Blues 7 7;

% add some labels with a one-off macro
vardef midlabel@#(expr t, a, b) =
    draw thelabel@#(t, origin) rotated angle (b-a) shifted 1/2[a,b]
enddef;

midlabel.top("$a$", point 4+a of c, origin);
midlabel.bot("$a$", point 4 of c, origin);
midlabel.bot("$a$", origin, point 0 of c);
midlabel.top("$b$", z0, point 0 of c);
midlabel.top("$c$", origin, z0);
midlabel.top("$a-c$", z0, point a of c);
midlabel.top("$2a\cos\theta-b$", point b of c, z0);

label.bot("$\bigl(2a\cos\theta - b\bigr) \cdot b"
    & "= \bigl( a - c \bigr) \cdot \bigl(a + c \bigr)$",
    point 1/2 of bbox currentpicture shifted 42 down);
label.bot("$c^2 = a^2 + b^2 - 2ab\cos\theta$",
    point 1/2 of bbox currentpicture shifted 12 down);
```

The law of cosines II



$$(2a\cos\theta - b)\cdot b = (a-c)\cdot(a+c)$$

$$c^2 = a^2 + b^2 - 2ab\cos\theta$$

— Sidney H. Kung

```
path c; numeric a, b;
c = fullcircle scaled 421;
a = 1/4; b = -7/8;

z0 = point a of c; z1 = point 4-a of c;
z3 = point b of c; z2 = point 4-b of c;

draw fullcircle scaled 42 rotated angle (z0-z3) shifted z3
    cutafter (z2--z3) withcolor Reds 7 6;
draw fullcircle scaled 42 rotated angle (z3-z2) shifted z2
    cutafter (z1--z2) withcolor Reds 7 6;

draw z0--z2--z3--z0--z1;
draw z2--z1--z3 dashed evenly;

draw c withcolor Blues 7 7;

label.top("$a+2b\cos\bigl(\pi-\theta\bigr)$", 1/2[z0, z1]);
label.bot("$a$", 1/2[z2, z3]);
label.lft("$b$", 1/2[z1, z2]);
label.rt ("$b$", 1/2[z3, z0]);
label.ulft("$c$", 3/4[z2, z0]);
label.urt ("$c$", 3/4[z3, z1]);

label("$\theta$", z3 + 8 (unitvector(z0-z3)+unitvector(z1-z3)));
label("$\theta$", z2 + 8 (unitvector(z0-z2)+unitvector(z1-z2)));

label.bot(btex \vbox{\openup 8pt\halign{\hfil $##$ \hfil\cr
  c \cdot c = b \cdot b + \Bigl(a + 2b \cos\bigl(\pi-\theta\bigr)\Bigr) \cdot a\cr
  c^2 = a^2 + b^2 - 2ab\cos\theta\cr}} etex,
  point 1/2 of bbox currentpicture shifted 42 down);
```

The law of cosines III (via Ptolemy's theorem)



$$c \cdot c = b \cdot b + \big(a + 2b\cos(\pi - \theta)\big) \cdot a$$

$$c^2 = a^2 + b^2 - 2ab\cos\theta$$

— Sidney H. Kung

```
path h; pair A, B, C, D, O; numeric theta;

h = halfcircle scaled 420;

O = origin;
A = point 4 of h;
B = point 0 of h;
C = point 5/4 of h;
D = (xpart C, ypart A);

2 theta = angle C;

draw unitsquare scaled 8 rotated angle (C-D) shifted D withcolor 3/4;
draw unitsquare scaled 8 rotated angle (A-C) shifted C withcolor 3/4;

draw A--C--B withcolor Reds 7 7;
draw O--C--D withcolor Reds 7 7;

drawoptions(withcolor Blues 7 6);
draw h;
label.ulft("$x^2 + y^2 = 1$", point 3 of h);
drawoptions();

primarydef o through p =
    (1+o/arclength(p))[point 1 of p, point 0 of p] --
    (1+o/arclength(p))[point 0 of p, point 1 of p]
enddef;
drawarrow 16 through (A--B);
drawarrow 16 through (O--point 2 of h);

dotlabel.bot("$A$", A);
dotlabel.bot("$B$", B);
dotlabel.urt("$C \smash{\;\bigl(\cos2\theta, \sin2\theta\bigr)}$", C);
dotlabel.bot("$D$", D);
dotlabel.llft("$O$", O);

label("$\theta$", 28 dir 1/2 theta shifted A);
label("$2\theta$", 20 dir theta);

label("$x$", B shifted 24 right);
label("$y$", point 2 of h shifted 24 up);

vardef midlabel@#(expr t, a, b) =
    draw thelabel@#(t, origin) rotated angle (b-a) shifted 1/2[a,b]
enddef;
midlabel.top("$2\cos\theta$", A, C);
midlabel.top("$2\sin\theta$", C, B);

label.bot("$\triangle ACD \sim \triangle ABC$",
    point 1/2 of bbox currentpicture shifted 42 down);

% fix bbox path in order to draw labels side by side
path p; p = bbox currentpicture shifted 20 down;

label.bot(btex \vbox{\openup 8pt\halign{\hfil $#$\hfil\cr
  CD \Big/ AC = BC \Big/ AB\cr
 \sin 2\theta \big/ 2 \cos\theta = 2 \sin\theta \big/ 2\cr
 \sin 2\theta = 2\sin\theta \cos\theta\cr}} etex, point 1/4 of p);

label.bot(btex \vbox{\openup 8pt\halign{\hfil $#$\hfil\cr
  AD \Big/ AC = AC \Big/ AB\cr
  \bigl(1 + \cos 2\theta \bigr) \big/ 2 \cos\theta = 2 \cos\theta \big/ 2\cr
  \cos 2\theta = 2\cos^2\theta - 1\cr}} etex, point 3/4 of p);
```

The double-angle formulae



$$\triangle ACD \sim \triangle ABC$$

$$CD/AC = BC/AB$$
$$\sin 2\theta / 2\cos\theta = 2\sin\theta/2$$
$$\sin 2\theta = 2\sin\theta\cos\theta$$

$$AD/AC = AC/AB$$
$$(1+\cos 2\theta)/2\cos\theta = 2\cos\theta/2$$
$$\cos 2\theta = 2\cos^2\theta - 1$$

— Roger B. Nelsen

```
path h; pair A, B, C, D, O; numeric theta;

h = halfcircle scaled 420;

O = origin;
A = point 4 of h;
B = point 0 of h;
C = point 5/4 of h;
D = (xpart C, ypart A);

theta = angle C;

draw unitsquare scaled 8 rotated angle (C-D) shifted D withcolor 3/4;
draw unitsquare scaled 8 rotated angle (A-C) shifted C withcolor 3/4;

drawoptions(withcolor Reds 7 7);
draw A--C--B;
draw O--C--D;

drawoptions();
label("$\theta/2$", 38 dir 1/4 theta shifted A);
label("$\theta/2$", 42 dir (270 + 1/4 theta) shifted C);
label("$\theta$", 20 dir 1/2 theta);

drawoptions(withcolor Blues 7 6);
draw h;
label.ulft("$x^2 + y^2 = 1$", point 3 of h);

drawoptions();
primarydef o through p =
    (1+o/arclength(p))[point 1 of p, point 0 of p] --
    (1+o/arclength(p))[point 0 of p, point 1 of p]
enddef;
drawarrow 16 through (A--B);
drawarrow 16 through (O--point 2 of h);

label("$x$", B shifted 24 right);
label("$y$", point 2 of h shifted 24 up);

vardef midlabel@#(expr t, a, b) =
    draw thelabel@#(t, origin) rotated angle (b-a) shifted 1/2[a,b]
enddef;

midlabel.bot("$1$", A, origin);
midlabel.top("$1$", origin, C);
midlabel.bot("$\cos\theta$", origin, D);
midlabel.bot("$1-\cos\theta$", D, B);
midlabel.top("$\sin\theta$", D, C);

label.bot(btex $\displaystyle
\tan \frac\theta2 = \frac{\sin\theta}{1+\cos\theta} = \frac{1-\cos\theta}{\sin\theta}
$ etex, point 1/2 of bbox currentpicture shifted 42 down);
```

The half-angle tangent formulae



$$\tan\frac{\theta}{2} = \frac{\sin\theta}{1+\cos\theta} = \frac{1-\cos\theta}{\sin\theta}$$

— R. J. Walker

```
z1 = 210 left; z2 = 210 right; z3 = 180 dir 113;

pair t; t = unitvector(z1-z3) + unitvector(z2-z3);
z4 - z3 = whatever * t;
z4 - z1 = whatever * t rotated 90;

z5 = whatever[z1,z4] = whatever[z2,z3];
z6 = whatever[z1,z4]; z6 - z2 = whatever * (z3 - z4);
z7 = whatever[z1,z4]; z7 - z2 = whatever * (z3 - z1);

draw subpath (1,3) of unitsquare scaled 8 rotated angle (z4-z1) shifted z4 withcolor 3/4;
draw subpath (1,3) of unitsquare scaled 8 rotated angle (z1-z6) shifted z6 withcolor 3/4;

draw z3--z4 dashed withdots scaled 1/2;
draw z2--z6 dashed withdots scaled 1/2;

drawoptions(withcolor Blues 8 7);
draw halfcircle scaled 48 shifted z1 cutafter (z1--z3);
label("$\alpha$", z1 + 32 dir 1/2 angle (z3-z1));

drawoptions(withcolor Greens 8 7);
draw reverse halfcircle scaled 48 shifted z2 cutafter (z2--z3);
label("$\beta$", z2 + 32 dir (90 + 1/2 angle (z3-z2)));

drawoptions(withcolor Oranges 8 7);
draw halfcircle scaled 48 rotated angle (z4-z3) shifted z3 cutafter (z2--z3);
draw halfcircle scaled 48 rotated angle (z7-z2) shifted z2 cutafter (z2--z6);
label("$\frac{\gamma}{2}$", z3 + 20 (unitvector(z4-z3) + unitvector(z2-z3)));

drawoptions(withcolor Reds 8 7);
picture a; a = image(
    for s=48,52:
        draw halfcircle scaled s rotated angle (z3-z2) shifted z5 cutafter (z5--z1);
    endfor
);
draw a;
draw a rotatedabout(z5, 180);
draw a rotatedabout(z5, 180) reflectedabout(z2,z6);
label("$\frac{\alpha+\beta}{2}$", z5 + 22 (unitvector(z1-z5) + unitvector(z3-z5)));

drawoptions(withcolor Purples 8 7);
draw halfcircle scaled 108 shifted z1 cutafter (z1--z4);
pair s, t; s = z1 + 58 dir 1/2 angle (z4-z1); t = s + (32, -18);
label.bot("$\frac{\alpha-\beta}{2}$", t);
drawarrow t {up} .. {left} s withpen pencircle scaled 1/4;

drawoptions();
draw z1--z2--z3--cycle;
draw z1--z7--z2;

label.urt ("$a$", 1/2[z2, z3]);
label.ulft("$b$", 1/2[z3, z1]);
label.bot ("$c$", 1/2[z1, z2]);
label.lrt ("$a-b$", 1/2[z2, z7]);

label.top(btex $\displaystyle
(a-b)\cos\frac{\gamma}{2} = c \sin\left(\frac{\alpha-\beta}{2}\right)
$ etex, point 5/2 of bbox currentpicture shifted 42 up);
```

Mollweide's equation

$$(a-b)\cos\frac{\gamma}{2} = c\sin\left(\frac{\alpha-\beta}{2}\right)$$



— H. Arthur DeKleine

```
numeric s, theta; theta = 28; s = 120;
path c; c = fullcircle scaled 2s;

z0 = whatever * dir theta;
z1 = whatever * dir theta;
z2 = (x1, y0) = (xpart point 0 of c, ypart point 6 of c);

drawoptions(withcolor 3/4);
draw unitsquare scaled 6 rotated 0   shifted point 6 of c;
draw unitsquare scaled 6 rotated 90  shifted z2;
draw unitsquare scaled 6 rotated 180 shifted point 0 of c;
draw unitsquare scaled 6 rotated 270 shifted center c;

drawoptions(withcolor Blues 8 8);
draw c;

drawoptions();
draw z0--z1--z2--cycle;
draw point 0 of c -- center c -- point 6 of c;

drawoptions(withcolor Blues 8 8);
draw center c withpen pencircle scaled dotlabeldiam;

drawoptions(withcolor Reds 8 7);
label("$\theta$", z0 + 24 dir 1/2 theta);
label("$\theta$", center c + 24 dir 1/2 theta);

drawoptions();
label.top("$1$", 1/2[point 0 of c, center c]);
label.lft("$1$", 1/2[point 6 of c, center c]);
label.bot("$1$", 1/2[point 6 of c, z2]);
label.rt ("$1$", 1/2[point 0 of c, z2]);
label.rt ("$\tan\theta$", 1/2[point 0 of c, z1]);
label.bot("$\cot\theta$", 1/2[point 6 of c, z0]);

picture p;
p = thelabel.top("$\csc\theta$", origin);
unfill bbox p rotated theta shifted 1/2[z0, center c];
       draw p rotated theta shifted 1/2[z0, center c];
draw thelabel.top("$\sec\theta$", origin) rotated theta shifted 1/2[z1, center c];

label.bot(btex \vbox{\openup 8pt\halign{\hfil $##$ \hfil\cr
    \tan^2\theta + 1 = \sec^2 \theta\cr
    \cot^2\theta + 1 = \csc^2 \theta\cr
    \left(\tan\theta + 1\right)^2 +
    \left(\cot\theta + 1\right)^2 =
    \left(\sec\theta + \csc\theta\right)^2\cr}} etex,
    point 1/2 of bbox currentpicture shifted 32 down);

label.bot(btex also \quad $\displaystyle \tan\theta =
    \frac{\tan\theta+1}{\cot\theta + 1}$ etex,
    point 1/2 of bbox currentpicture shifted 24 down);
```

Tangent, cotangent, secant, and cosecant



$$\tan^2\theta + 1 = \sec^2\theta$$

$$\cot^2\theta + 1 = \csc^2\theta$$

$$(\tan\theta + 1)^2 + (\cot\theta + 1)^2 = (\sec\theta + \csc\theta)^2$$

$$\text{also} \qquad \tan\theta = \frac{\tan\theta + 1}{\cot\theta + 1}$$

— William Romaine

```
numeric theta, u; theta = 60; u = 144;
z0 = u * left;
z1 = origin;
x2 = x1; z2 = whatever * dir 1/2 theta shifted z0;
y3 = y1; z2 - z3 = whatever * (z2-z0) rotated 90;
z4 = z1 shifted (z2-z3);
z5 = z2 shifted (z2-z3);
x6 = x5; y6 = y1;
picture P[];
P0 = image(
    draw z0--z1--z2--cycle withpen pencircle scaled 1;
    label.bot("$1$", 1/2[z0, z1]);
    label.rt ("$z$", 1/2[z1, z2]);
    label("$\theta/2$", 32 dir 1/4 theta shifted z0 shifted 2 down)
          withcolor Reds 8 7;
);

P1 = image(
    draw unitsquare scaled 6 rotated 90 withcolor 1/2;
    draw P0;
);
% ... and so on, with more complication, for P2, P3, P4

P2 := P2 shifted (7/4u, 0);
P3 := P3 shifted (0,    -7/4u);
P4 := P4 shifted (7/4u, -7/4u);

draw P1; draw P2; draw P3; draw P4;

drawoptions(withpen pencircle scaled 2 withcolor Blues 8 7);
interim linecap := butt;
interim linejoin := mitered;
interim bboxmargin := 16;
picture a; a = image(drawarrow (left--right) scaled 21);
drawoptions();

for i=1 upto 3:
    draw a rotated angle (center P[i+1] - center P[i])
           shifted 1/2[center P[i], center P[i+1]];
endfor

label.bot(btex $\displaystyle
  z = \tan\frac\theta2 \quad\Longrightarrow\quad
  \sin\theta = \frac{2z}{1 + z^2} \quad\hbox{and}\quad
  \cos\theta = \frac{1-z^2}{1 + z^2}
  $ etex, point 1/2 of bbox currentpicture shifted 42 down);
```

Substitution to make a rational function of sine and cosine



$$z = \tan\frac{\theta}{2} \quad \Longrightarrow \quad \sin\theta = \frac{2z}{1+z^2} \quad \text{and} \quad \cos\theta = \frac{1-z^2}{1+z^2}$$

— Roger B. Nelsen

```
numeric u; u = 42;
picture P[];
P1 = image(
    path t[];
    t1 = origin -- (2u, 0) -- (2u, u) -- cycle;
    t2 = origin -- (3u, -u) -- (3u, 0) -- cycle;

    fill t1 withcolor Oranges 7 1;
    fill t2 withcolor Blues 7 1;

    for x= 0 upto 3: draw (down -- up) shifted (x, 0) scaled u withcolor 3/4; endfor
    for y=-1 upto 1: draw (origin -- 3 right) shifted (0, y) scaled u withcolor 3/4; endfor

    draw fullcircle scaled 3/2 u
        rotated angle point 1 of t2
        cutafter subpath (2, 3) of t1
        withcolor Reds 7 6;

    draw t1 -- subpath (0, 2) of t2 -- point 1 of t1 withpen pencircle scaled 1;
    draw point 1 of t2 -- point 2 of t1 withpen pencircle scaled 1;

    label.bot("$\displaystyle \arctan \frac12 + \arctan \frac13 = \frac\pi4$",
       point 1/2 of bbox currentpicture shifted 21 down);
);
P2 = image(
    path t[];
    t1 = origin -- (-u, 0) -- (-u, -u) -- cycle;
    t2 = origin -- (-u, 2u) -- (-u, 0) -- cycle;
    t3 = origin -- (5u, 5u) -- (-u, 2u) -- cycle;

    fill t1 withcolor Greens 7 1;
    fill t2 withcolor Oranges 7 1;
    fill t3 withcolor Blues 7 1;

    numeric y; y = -2;
    for ss = (-1, 1), (-1, 1), (-1, 2), (-1, 3), (-1, 5), (1, 5), (3, 5):
        draw ((xpart ss, incr y) -- (ypart ss, y)) scaled u withcolor 3/4;
    endfor
    numeric x; x = -2;
    for ss = (-1, 3), (-1, 3), (-1, 4), (1, 4), (2, 5), (3, 5), (3, 5):
        draw ((incr x, xpart ss) -- (x, ypart ss)) scaled u withcolor 3/4;
    endfor

    draw halfcircle scaled 3/2 u
        rotated angle point 1 of t3
        withcolor Reds 7 6;

    draw                  t1 withpen pencircle scaled 1;
    draw subpath (0,2) of t2 withpen pencircle scaled 1;
    draw subpath (0,2) of t3 withpen pencircle scaled 1;

    label.bot("$\displaystyle \arctan 1 + \arctan 2 + \arctan 3 = \pi$",
       point 1/2 of bbox currentpicture shifted 21 down);

);
label.top(P1, 21 up); label.bot(P2, 21 down);
```

Sums of arctangents



$$\arctan\frac{1}{2} + \arctan\frac{1}{3} = \frac{\pi}{4}$$



$$\arctan 1 + \arctan 2 + \arctan 3 = \pi$$

— Edward M. Harris

```
numeric u; u = 33;
path xx, yy;
xx = (left -- 8 right) scaled u;
yy = (down -- 11 up) scaled u;

numeric a, b, c, m;
m = 1.9;
c = -2;
a = 6.5;
b = 1;

z0 = (0, c) scaled u;
z1 = (a, a*m + c) scaled u;
z2 = (a, b) scaled u;
z3 = whatever[z0, z1]; z2 - z3 = whatever * (z0 - z1) rotated 90;

path p, t;
p = ((1/2, 1/2 m + c) -- (a+1/2, (a+1/2)*m + c)) scaled u;
t = (origin -- right -- (1, m)) scaled u shifted 1/4[z3, z1];

drawoptions(withpen pencircle scaled 1/4 withcolor Reds 8 6);
draw subpath (1,3) of unitsquare scaled 6 rotated 90 shifted point 1 of t;
draw subpath (1,3) of unitsquare scaled 6 rotated angle (z2-z3) shifted z3;

drawoptions(withcolor Reds 8 8);
draw t;
draw z1 -- z2 -- z3;
draw p;
draw thelabel.top("$y=mx + c$", origin) rotated angle (z1-z0) shifted point 1/5 of p;

drawoptions(withcolor Blues 8 7);
label.llft("$d$", 1/2[z2, z3]);
label.rt("$\big| ma + c - b \big|$", 1/2[z2, z1]);
label.bot("$1$", point 1/2 of t);
label.rt("$m$", point 3/2 of t);
label.ulft("$\sqrt{1+m^2}$", 1/2[point 0 of t, point 2 of t]);

drawoptions();
dotlabel.lrt ("$(a, ma+c)$", z1);
dotlabel.lrt ("$(a, b)$", z2);

drawarrow xx;
drawarrow yy;
label.rt  ("$x$", point 1 of xx);
label.top ("$y$", point 1 of yy);

label.bot(btex $\displaystyle
    \frac{d}{1} = \frac{\left|ma+c-b\right|}{\sqrt{1+m^2}}$ etex,
    point 1/2 of bbox currentpicture shifted 42 down);
```

The distance between a point and a line



$$\frac{d}{1} = \frac{|ma + c - b|}{\sqrt{1+m^2}}$$

— R. L. Eisenman

```
numeric u; u = 12;
pair a, b, fa, fb, ta, tb, m, fm, am, bm;
xpart a = xpart fa = xpart ta = -4u = -xpart b = - xpart fb = - xpart tb;
ypart a = ypart b = 0;
ypart ta = 8u;
ypart tb = 11u;
ypart fa = 2u;
ypart fb = 9u;
m = 1/2[a, b];
fm = 1/2[ta, tb];
xpart am = xpart bm = xpart m;
am - fa = whatever * (tb - ta);
bm - fb = whatever * (tb - ta);

path base, lid, curve;
base = fa -- a -- b -- fb;
lid = fa -- ta -- tb -- fb;
curve = fa {dir 85} .. fm {tb-ta} .. fb;

picture P[];
P1 = image(
    fill lid & reverse curve & cycle withcolor Blues 8 2;
    draw base;
    draw lid;
    draw curve withcolor Blues 8 7;
    label.bot("$a$", a);
    label.bot("$b$", b);
    dotlabel.bot("$\frac{a+b }{2}$", m);
    dotlabel.ulft("$f\left(\frac{a+b}{2}\right)$", fm);
);
% ... and so on for P2 .. P5

draw P1; draw P2 shifted (12u, 0); draw P3 shifted (24u, 0);
label("$\le$", (6u, 6u)); label("$=$", (18u, 6u));

draw P4 shifted (6u, -16u); draw P5 shifted (18u, -16u);
label("$=$", (0u, -10u)); label("$\le$", (12u, -10u));
```

The midpoint rule is better than the trapezoidal rule for concave functions



— Frank Burk

```
path ff, xx, yy;
pair p, q, r, s;

ff = (80, 55) {dir 10} .. (333, 233);
xx = 5 left -- 13 right + (xpart point 1 of ff, 0);
yy = 5 down -- 13 up    + (0, ypart point 1 of ff);

numeric t; t = 1/12;

p = (xpart point t of ff, 0);
q = (xpart point 1-t of ff, 0);
r = (0, ypart point t of ff);
s = (0, ypart point 1-t of ff);

fill p -- subpath(t, 1-t) of ff -- q -- cycle withcolor Oranges 7 1;
fill r -- subpath(t, 1-t) of ff -- s -- cycle withcolor Blues 7 1;

drawoptions(withpen pencircle scaled 1/4);
draw p -- point t   of ff -- r;
draw q -- point 1-t of ff -- s;

drawarrow xx;
drawarrow yy;
draw ff withpen pencircle scaled 3/4 withcolor Reds 8 7;

label.rt("$\left\{\:\vcenter{\halign{$#$\hfil\cr u=f(x)\cr v=g(x)\cr}}\right.$",
    point 1 of ff);

label.rt ("$u$", point 1 of xx);
label.top("$v$", point 1 of yy);

interim labeloffset := 8;
dotlabel.lft("$s=g(b)$", s);
dotlabel.lft("$r=g(a)$", r);
dotlabel.bot("$p=f(a)$", p);
dotlabel.bot("$q=f(b)$", q);
dotlabel.llft("$0$", origin);

def box(expr s) =
    "\pdfliteral{" &
    decimal redpart s & " " & decimal greenpart s & " " & decimal bluepart s &
    " rg}\vrule height 5mm width 8mm depth 2mm\pdfliteral{0 g}"
enddef;

label.bot("\vbox{\openup 16pt\halign{\hfil $\displaystyle # $\hfil\cr" &
 "\hbox{Area\ " & box(Blues 7 1) & "}+\hbox{Area\ " & box(Oranges 7 1) & "}=qs-pr\cr" &
 "\int_r^s u\:dv + \int_p^q v\: du = uv \: \bigg|_{(p, r)}^{(q, s)}\cr" &
 "\int_a^b f(x) g'(x) \:dx = f(x) g(x) \: \Big|_a^b - \int_a^b g(x) f'(x) \: dx\cr}}",
 point 1/2 of bbox currentpicture shifted 21 down);
```

Integration by parts



Area + Area $= qs - pr$

$$\int_r^s u\,dv + \int_p^q v\,du = uv\Big|_{(p,r)}^{(q,s)}$$

$$\int_a^b f(x)g'(x)\,dx = f(x)g(x)\Big|_a^b - \int_a^b g(x)f'(x)\,dx$$

— Richard Courant

```
z0 = -8(1,1); z1 = 221(1,1);
path xx, yy, xy;
xx = (x0, 0) -- (x1, 0); yy = xx rotated 90; xy = z0 -- z1;

picture P[];
P0 = image(
    drawarrow xx;
    drawarrow yy;
    draw xy withcolor Blues 8 7;
    draw thelabel.ulft("$y=x$", origin) rotated 45
        shifted point 1 of xy withcolor Blues 8 7;
);

P1 = image(
    path ff, ff';
    ff = point 1/3 of xx shifted 12 up .. {dir 5} point 1 of xx shifted 72 up;
    ff' = ff reflectedabout(z0, z1);
    numeric a, b; (a, b) = point 3/8 of ff;
    drawoptions(dashed evenly withpen pencircle scaled 1/4);
    draw (a,b) -- (b, a) -- (b, b) -- (a,b) -- (a, a) -- (b, a);
    drawoptions(withcolor Reds 8 7);
    draw ff; label.rt("$y=f(x)$", point 1 of ff);
    draw ff' dashed withdots scaled 1/4;
    label.top("$x=f(y)$, $y=f^{-1}(x)$", point 1 of ff');
    drawoptions();
    draw P0;

    dotlabel.lrt("$(a, b)$", (a, b));
    dotlabel.lrt("$(a, a)$", (a, a));
    dotlabel.lrt("$(b, b)$", (b, b));
    dotlabel.ulft("$(b, a)$", (b, a));
    label.top("$a-b$", 1/2(a+b, 2a));
);
P2 = image(
    path ff, ff';
    ff = point 7/8 of yy shifted 21 right .. {dir -8} point 1 of xx shifted 55 up;
    ff' = ff reflectedabout(z0, z1);
    numeric a, b; (a, b) = point 1/4 of ff;
    drawoptions(dashed evenly withpen pencircle scaled 1/4);
    draw (a,b) -- (b, b) -- (b, a) -- (a, a) -- cycle;
    drawoptions(withcolor Reds 8 7);
    draw ff; label.rt("$y=f(x)$", point 1 of ff);
    draw ff' dashed withdots scaled 1/4;
    label.top("$x=f(y)$, $y=f^{-1}(x)$", point 1 of ff');
    drawoptions();
    draw P0;
    dotlabel.llft("$(a, b)$", (a, b));
    dotlabel.lrt("$(a, a)$", (a, a));
    dotlabel.lrt("$(b, b)$", (b, b));
    dotlabel.llft("$(b, a)$", (b, a));
);

label.top(P1, origin); label.bot(P2, 21 down);
```

The graphs of f and f^{-1} are reflections about the line $y = x$





— Ayoub B. Ayoub

```
path para, dirx, xx, yy;
numeric p, s, t, u;
p = 3/2; s = 1/8; u = 28; t = 12/s;
para = ((49/4/p, -7) for y=s - 7 step s until 7: -- (y * y / 4p, y) endfor) scaled u;
dirx = (-p*u, ypart point 0 of para) -- (-p*u, ypart point infinity of para);
xx = (-u-p*u, 0) -- (p*u + xpart point 0 of para, 0);
yy = dirx shifted (p*u, 0);

z0 = p * u * right;
z1 = point t of para;
z2 = z0 reflectedabout(z1, direction t of para rotated 90 shifted z1);
z3 = z2 rotatedabout(z1, 180);
z4 = 1/2[z0, z3];
z5 = z4 rotatedabout(z1, 180);

drawoptions(withcolor 3/4);
draw subpath (3.8, 4.2 + 1/45 angle (z1-z0)) of fullcircle scaled 2 abs(z0-z1) shifted z1;
draw subpath (1, 3) of unitsquare scaled 6 rotated angle (z0-z3) shifted z4;

drawoptions(withcolor Blues 8 4);
draw z4--z5;
draw z3--z1;
draw 1.2[z0,z3] -- 1.6[z3,z0];
drawoptions(withcolor Blues 8 6);
draw thelabel.ulft("$m_1 = y' = 2p/y$", origin) rotated angle (z5-z4) shifted z5;
draw thelabel.ulft("$m_2 = -y/2p$", origin) rotated angle (z0-z3) shifted 1.5[z3, z0];
numeric a; a = 1/2 angle (z1-z0);
label("$\alpha$", z1 + 36 dir (180 + 3/2 a));
label("$\beta$",  z1 + 36 dir (180 + 1/2 a));
label("$\gamma$", z1 + 36 dir (1/2 a));

drawoptions(withcolor Reds 8 7);
draw para; draw dirx;
draw thelabel.top("$y^2=4px$", origin)
    rotated (180 + angle direction 8 of para) shifted point 8 of para;
draw thelabel.top("$x=-p$", origin)
    rotated 90 shifted point 1/8 of dirx;


drawoptions();
drawarrow z0 -- z1 -- z2;
drawarrow xx;
drawarrow yy;

dotlabel.llft("$F(p,0)$", z0);
dotlabel.lrt("$Q(x,y)$", z1);
dotlabel.lft("$D(-p,y)$", z3);
label.bot("\mathsurround 6pt" &
    "$QF=QD$ and $m_1\cdot m_2=-1,$ therefore $\alpha=\beta=\gamma$",
    point 1/2 of bbox currentpicture shifted 42 down);
```

The reflection property of the parabola



$$QF = QD \quad \text{and} \quad m_1 \cdot m_2 = -1, \quad \text{therefore} \quad \alpha = \beta = \gamma$$

— Ayoub B. Ayoub

```
input arrow_label
picture P[];
path c, cycloid, base;
numeric pi, r, s; r = 42; s = 1; pi = 3.141592653589793;
c = fullcircle scaled 2r rotated 90;
cycloid = point 0 of c rotated -180 shifted (-pi * r, 0) for t = s-180 step s until 180:
    -- point 0 of c rotated -t shifted (t / 180 * pi * r, 0) endfor;
base = point 0 of cycloid shifted 21 left -- point infinity of cycloid shifted 21 right;
P0 = image(
    draw center c -- c withcolor Blues 8 7;
    label.rt("$r$", 1/2 point 0 of c);
);
P1 = image(
    draw P0;
    draw cycloid withcolor Blues 8 7;
    draw base;
    drawarrow subpath (7, 5.8) of c scaled 1.1 withcolor Reds 8 7;
    dotlabel.top("$P$", point 0 of c);
    arrow_label(point 0 of cycloid, point infinity of cycloid, "$2\pi r$", 10);
);
P2 = image(
    draw P0;
    label("$+$", (-3/2r, 0));
    label("$+$", (+3/2r, 0));
    % see source for the rest of this part ...
);
P3 = image(
    draw P0;
    label("$+$", (-3/2r, 0));
    label("$+$", (+3/2r, 0));
    % see source for the rest of this part ...
);
draw P1;
draw P2 shifted (0, -3r);
draw P3 shifted (0, -6r);

label.bot(btex \vbox{\openup 12pt\halign{\hfil $\displaystyle #$\hfil\cr
\frac12\pi r \cdot 2r \qquad\qquad + \qquad\qquad
\pi r^2               \qquad\qquad + \qquad\qquad
\frac12\pi r \cdot 2r \cr
\hbox to 0pt{\hss\small therefore\qquad}A = 3\pi r^2\cr}} etex,
point 1/2 of bbox currentpicture shifted 42 down);
```

Area under an arch of the cycloid



$$\frac{1}{2}\pi r \cdot 2r \qquad + \qquad \pi r^2 \qquad + \qquad \frac{1}{2}\pi r \cdot 2r$$

therefore $\quad A = 3\pi r^2$

— Richard M. Beekman

Inequalities

```
input arrow_label
path h; h = halfcircle scaled .95 \mpdim{\textwidth};
z0 = point 2.7818 of h;
x1 = x0; y1 = 0;

draw point 4 of h -- z0 -- point 0 of h dashed evenly withpen pencircle scaled 1/4;
drawoptions(withcolor Blues 8 7);
draw h -- cycle; draw origin -- point 2 of h; draw z0--z1;

drawoptions();
arrow_label(point 4 of h, z1, "$a$", 8);
arrow_label(z1, point 0 of h, "$b$", 8);

label.rt("$\sqrt{ab}$", 1/2[z0, z1]);
label.rt("$\frac{a+b}{2}$", 3/4 point 2 of h);
label.bot("$\displaystyle \sqrt\frac{ab} \le {a+b}{2}$",
    point 1/2 of bbox currentpicture shifted 42 down);
```

The arithmetic mean – geometric mean inequality I



$$\sqrt{\frac{ab}{\leq}a + b2}$$

— Charles D. Gallant

```
numeric a, b; a + b = 7/8 \mpdim{\textwidth}; a = 610/377 b;
path r, s;
s = unitsquare shifted -(1/2, 1/2) scaled (a-b);
r = unitsquare xscaled a yscaled -b shifted point 0 of s;
fill s withcolor Blues 8 2;
for t=0 upto 3:
    fill r rotated 90t withcolor Oranges 8 1; draw r rotated 90t;
endfor
label.bot("$a$", point 5/2 of r);
label.lft("$a$", point 5/2 of r rotated 270);
label.lft("$b$", point 3/2 of r rotated 180);
label.bot("$b$", point 3/2 of r rotated 270);

label.bot("$(a+b)^2 - (a-b)^2 = 4ab$",
    point 1/2 of bbox currentpicture shifted 21 down);
label.bot("$\displaystyle \frac{a+b}{2} \ge \sqrt{ab}$",
    point 1/2 of bbox currentpicture shifted 13 down);
```

The arithmetic mean – geometric mean inequality II



$$(a+b)^2 - (a-b)^2 = 4ab$$

$$\frac{a+b}{2} \geq \sqrt{ab}$$

— Doris Schattschneider

```
input arrow_label
numeric a, b;
5b = 3a; a + b = \mpdim{\textwidth};
z1 = (1/2(a+b) +-+ 1/2(a-b), 1/2(b-a));
path C, c;
C = fullcircle scaled a rotated 90;
c = fullcircle scaled b rotated 90 shifted z1;
drawoptions(withcolor Blues 8 5);
draw origin -- z1;
draw (0, y1) -- z1 dashed evenly scaled 1/2;
draw C -- point 4 of C -- point 4 of c -- c;
drawoptions();
arrow_label(point 0 of C, point 4 of C, "$a$", 10);
arrow_label(point 0 of c, point 4 of c, "$b$", -10);
arrow_label(point 4 of C, point 4 of c, "$\sqrt{ab}$", 10);
path aa; aa = (center C -- center c) shifted 12 up;
drawdblarrow aa;
picture t; t = thelabel("$\frac{a+b}{2}$", point 1/2 of aa);
    unfill bbox t shifted 2.7 up; draw t;

arrow_label((0, y1), center C, "$\frac{a-b}{2}$", 12);

label.top("\mathsurround 6pt"
        & "$\displaystyle \frac{a+b}{2} \ge \sqrt{ab},$"
        & "with equality iff $a=b$",
    point 5 /2 of bbox currentpicture shifted 42 up);
```

The arithmetic mean – geometric mean inequality III

$$\frac{a+b}{2} \geq \sqrt{ab}, \text{ with equality iff } \quad a = b$$



— Roland H. Eddy

```
numeric P, s, minx, maxx; s = 2; P = 1440; minx = 4s; maxx = P/minx;
path pp, xx, yy;
pp = (minx, P/minx) for x=minx+s step s until maxx+eps: -- (x, P/x) endfor;
xx = (-minx, 0) -- (maxx+minx, 0); yy = xx rotated 90;

z0 = point 2/3 length pp of pp;
z1 = (x0+y0, 0); z2 = (0, x0+y0); z3 = (2 sqrt(P), 0); z4 = (0, 2 sqrt(P));

picture T[];

T0 = image(
    undraw (left--right) scaled 1/2 \mpdim{\textwidth} shifted point 1/2 of xx;
    draw origin -- point 1 of xx rotated 45 dashed evenly withpen pencircle scaled 1/4;
    drawarrow xx; drawarrow yy;
    draw z1 -- z2 withcolor Blues 8 7;
    label.rt("$x$", point 1 of xx);
    label.top("$y$", point 1 of yy);
    dotlabel.urt("$\bigl(x, y\bigr)$", z0);
    dotlabel.bot("$\bigl(S, 0\bigr)$", z1);
    dotlabel.lft("$\bigl(0, S\bigr)$", z2));
T1 = image(
    draw T0;
    draw pp withcolor Reds 8 7;
    label.rt("$xy=P$", point 1/2 of pp) withcolor Reds 8 7;
    draw z3 -- z4 withcolor Blues 8 7;
    dotlabel.urt("$\bigl(\sqrt{\scriptstyle P}, \sqrt{\scriptstyle P}\bigr)$", 1/2(x3, y4));
    dotlabel.bot("$\bigl(2\sqrt{\scriptstyle P}, 0\bigr)$", z3);
    dotlabel.lft("$\bigl(0, 2\sqrt{\scriptstyle P}\bigr)$", z4);

    label.top(btex \vbox{\hsize 3.7in\centering For a given product,
        the sum of two positive numbers is minimal when the numbers are
        equal.} etex, point 5 /2 of bbox currentpicture shifted 13 up));
T2 = image(
    fill unitsquare xscaled x0 yscaled y0 withcolor Reds 7 1;
    fill unitsquare scaled 1/2(x0+y0) withcolor Blues 7 1;
    fill unitsquare xscaled 1/2(x0+y0) yscaled y0 withcolor 1/2[Reds 7 1, Blues 7 1];
    draw unitsquare xscaled x0 yscaled y0 withpen pencircle scaled 1/4;
    draw unitsquare scaled 1/2(x0+y0)     withpen pencircle scaled 1/4;

    picture eq; eq = image(
        for t=-1/2, 1/2:
            draw (up--down) rotated -5 scaled 2 shifted (t, 0) withpen pencircle scaled 1/4;
        endfor
    );
    draw eq shifted (1/4x0+3/4y0, y0);
    draw eq shifted (3/4x0+1/4y0, y0);
    draw eq rotated 90 shifted (1/2x0+1/2y0, 1/4x0+3/4y0);

    draw T0;
    dotlabel.urt("$\frac12\bigl(S, S\bigr)$", 1/2[z1,z2]);
    label.top(btex \vbox{\hsize 3.7in\centering For a given sum,
        the product of two positive numbers is maximal when the numbers are
        equal.} etex, point 5 /2 of bbox currentpicture shifted 13 up));

label.top(T1, 9 up); label.bot(T2, 9 down);
```

Two extremum problems

For a given product, the sum of two positive numbers
is minimal when the numbers are equal.



For a given sum, the product of two positive numbers
is maximal when the numbers are equal.



— Paulo Montuchi and Warren Page

```
path c; c = fullcircle scaled 233;
pair A, G, H, M, P, Q, R;
A = center c;
P = point 4 of c;
Q = point 6 of c;
R = point 8 of c;
M - P = (tw-18, 0);
G = c intersectionpoint halfcircle scaled abs(M-A) shifted 1/2[A, M];
H = (xpart G, ypart A);

drawoptions(withcolor 3/4);
  draw unitsquare scaled 8 shifted H;
  draw unitsquare scaled 8 rotated angle (Q-A) shifted A;
  draw unitsquare scaled 8 rotated angle (A-G) shifted G;

drawoptions(withcolor Blues 8 4);
  draw c;

drawoptions(withcolor Blues 8 8);
  draw P -- (M -- G -- A -- Q -- cycle);
  draw G -- H;

drawoptions(withcolor Reds 8 7);
  label.ulft("$\frac{a-b}{2}$",                                 1/2[A, G]);
  label.lft ("$\frac{a-b}{2}$",                                 1/2[A, Q]);
  label.bot ("$\frac{a+b}{2}$",                                 1/2[A, M]);
  label.top ("$\frac{2ab}{a+b}$",                               1/2[H, M]);
  label.urt ("$\scriptstyle\sqrt{ab}$",                         1/2[G, M]);
  label.lrt ("$\scriptstyle\sqrt{\frac12\left(a^2+b^2\right)}$", 1/2[Q, M]);
drawoptions();
dotlabel.llft("$A$", A);
dotlabel.llft("$R$", R);
dotlabel.lft ("$P$", P);
dotlabel.rt  ("$M$", M);
dotlabel.bot ("$Q$", Q);
dotlabel.bot ("$H$", H);
dotlabel.urt ("$G$", G);

label.bot("$PM=a$, \quad $RM=b$, \quad $a>b>0$",
    point 1/2 of bbox currentpicture shifted 21 down);
label.bot("$HM < GM < AM < QM$",
    point 1/2 of bbox currentpicture shifted 13 down);
label.bot("$\displaystyle\frac{2ab}{a+b}<\sqrt{ab}" &
          "<\frac{a+b}{2}<\sqrt{\frac12\left(a^2+b^2\right)}$",
    point 1/2 of bbox currentpicture shifted 13 down);
```

The HM–GM–AM–QM inequalities I



$$PM = a, \quad RM = b, \quad a > b > 0$$

$$HM < GM < AM < QM$$

$$\frac{2ab}{a+b} < \sqrt{ab} < \frac{a+b}{2} < \sqrt{\frac{1}{2}\left(a^2+b^2\right)}$$

— Roger B. Nelsen

```
pair A, B, C, D, E, F, G;
D = origin; A = - C;
C - A = (tw, 0);
B = 7/26[A, C];
E = (B -- B shifted 400 up) intersectionpoint halfcircle scaled abs(A-C);
F - B = whatever * (E - D);
F - E = whatever * (E - D) rotated 90;
G = E - B rotated angle (E - F);


draw halfcircle scaled abs (A-C) withcolor 7/8;

drawoptions(withcolor Blues 8 4);
draw subpath (1, 3) of unitsquare scaled 8 rotated angle (D-B) shifted B;
draw subpath (1, 3) of unitsquare scaled 8 rotated angle (B-F) shifted F;
draw subpath (1, 3) of unitsquare scaled 8 rotated angle (D-E) shifted E;

drawoptions(withcolor Blues 8 7);
draw A -- C;
draw B -- E -- D;
draw B -- F -- G -- D;

drawoptions(withcolor Reds 8 7);

label.llft("$\frac{2ab}{a+b}$",         1/2[B, F]);
label.lft("$\scriptstyle\sqrt{ab}$", 1/2[B, E]);
label.urt("$\frac{a+b}{2}$",            1/2[D, E]);
label.rt ("$\scriptstyle\sqrt{\frac12\left(a^2+b^2\right)}$", 1/2[D, G]);

drawoptions();
input arrow_label;
arrow_label(A, B, "$a$", 10);
arrow_label(B, C, "$b$", 10);
label.top("$\frac{b-a}{2}$", 1/2[B, D]);
label.top("$\frac{b-a}{2}$", 1/2[E, G]);
interim labeloffset := 6;
dotlabel.urt("$A$", A);
dotlabel.ulft("$B$\enspace", B);
dotlabel.ulft("$C$", C);
dotlabel.urt("$D$", D);
dotlabel.ulft("$E$", E);
dotlabel.ulft("$F$", F);
dotlabel.ulft("$G$", G);

label.bot("$AB=a$,\quad $BC=b$,\quad $AD=DC=\frac{a+b}{2}$",
    point 1/2 of bbox currentpicture shifted 42 down);
label.bot("$BE \perp AB$,\quad $DE=AD$",
    point 1/2 of bbox currentpicture shifted 13 down);
label.bot("$FE \perp ED$,\quad $FB \parallel ED$, \quad $EG=BD=\frac{b-a}{2}$",
    point 1/2 of bbox currentpicture shifted 13 down);
```

The HM–GM–AM–QM inequalities II



$$AB = a, \quad BC = b, \quad AD = DC = \tfrac{a+b}{2}$$

$$BE \perp AB, \quad DE = AD$$

$$FE \perp ED, \quad FB \parallel ED, \quad EG = BD = \tfrac{b-a}{2}$$

— Sidney H. Kung

```
numeric u; u = 20;
path U, A, B;
U = unitsquare shifted -(1/2, 1/2) scaled u;
A = unitsquare scaled 3u shifted point 2 of U rotated 90;
B = unitsquare scaled 4u shifted point 0 of U;

picture P[];
P1 = image(
    for t=0, 180:
        fill A rotated t withcolor Greens 8 1;
        fill B rotated t withcolor Reds 8 1;
    endfor
    fill U withcolor Reds 8 2;
    for t=0, 180:
        draw subpath (1, 3) of A rotated t;
        draw B rotated t;
    endfor

    label.top("$a$", point 3/2 of A);
    label.lft("$a$", point 5/2 of A);
    label.top("$b$", point 5/2 of B);
    label.lft("$b$", point 3/2 of B rotated 180);

    label.rt(btex
    \vbox{\openup 12pt\halign{\hbox to 64pt{\hfil$#$}&${}\ge #$\hfil\cr
    2a^2 + 2b^2 & \left(a+b\right)^2\cr
    \sqrt{\frac12\left(a^2+b^2\right)} & \displaystyle \frac{a+b}{2}\cr}}
    etex, point 3/2 of bbox currentpicture shifted 34 right);
);

% ... and similar for P2, P3

draw P1;
draw P2 shifted (9u * down);
draw P3 shifted (18u * down);
```

The HM–GM–AM–QM inequalities III



$$2a^2 + 2b^2 \geq (a+b)^2$$

$$\sqrt{\tfrac{1}{2}\left(a^2+b^2\right)} \geq \frac{a+b}{2}$$



$$\left(\sqrt{a+b}\right)^2 \geq 4 \cdot \tfrac{1}{2}\sqrt{a}\sqrt{b}$$

$$\frac{a+b}{2} \geq \sqrt{ab}$$



$$1 \geq 4 \cdot \frac{a}{a+b} \cdot \frac{b}{a+b}$$

$$\sqrt{ab} \geq \frac{2ab}{a+b}$$

— Roger B. Nelsen

```
numeric a, b, am, gm, rms, hm, ch; a + b + 32 = tw; b = 4a;
am = 1/2(a+b); gm = sqrt(a*b); rms = 1/2 sqrt 2 * (a++b);
hm = 2a / (a+b) * b; ch = a / (a+b) * a + b / (a+b) * b;
path xx, yy, qq, ss, pp, arc, harc, hpp;
xx = 10 left -- (a+b+20) * right; qq = xx rotated 45; yy = xx rotated 90;
ss = (-10, a + b + 10) -- (a + b + 10, -10);
arc = quartercircle scaled 2 (a++b);
harc = subpath(1/2,3/2) of quartercircle scaled (sqrt(2) * (a+b));

numeric s, ix; ix = sqrt(a*b); s = 4;
pp = (ix, ix) for x=s+ix step s until a+b-8: -- (x, b/x*a) endfor;
pp := reverse pp reflectedabout(origin, point 1 of qq) & pp;
numeric s, ix; ix = 1/2 sqrt(2) * (a++b); s = 4;
hpp = (ix, ix) for x=s+ix step s until b+eps: -- (x, a / 2x * a + b / 2x * b) endfor;
hpp := reverse hpp reflectedabout(origin, point 1 of qq) & hpp;

drawoptions(withcolor Greens 8 7);
 draw pp; label.rt("$xy=ab$", point infinity of pp);
 draw hpp; label.rt("$2xy=a^2+b^2$", point infinity of hpp);

drawoptions(withcolor Oranges 8 7);
 draw arc; draw harc; begingroup; interim ahlength := 2;
  z0 = 1/3[point infinity of pp, point infinity of hpp];
  z1 = 2/3[point infinity of pp, point infinity of hpp];
  path aa, bb;
  aa = (z0 -- point 0 of harc) cutafter fullcircle scaled 4 shifted point 0 of harc;
  numeric t, u; (t, u) = arc intersectiontimes (aa shifted (z1-z0));
  bb = (z1 -- point t of arc) cutafter fullcircle scaled 4 shifted point t of arc;

  drawarrow aa withpen pencircle scaled 1/4 withcolor Oranges 8 5;
  drawarrow bb withpen pencircle scaled 1/4 withcolor Oranges 8 5;

  label.rt("$2\left(x^2 + y^2\right) = \left(a+b\right)^2$", z0);
  label.rt("$x^2 + y^2 = a^2 + b^2$", z1);
 endgroup;

drawoptions(withcolor Blues 8 7);
 draw qq; draw thelabel.top("$x=y$", origin) rotated 45 shifted point 0.9 of qq;
 draw ss; draw thelabel.top("$x+y=a+b$", origin) rotated -45 shifted point 0.9 of ss;

drawoptions(dashed evenly scaled 1/2);
 draw (hm, ch) -- (am, ch) -- (am, am) -- (ch, am) -- (ch, hm);
 draw (gm, rms) -- (gm, gm) -- (rms, gm);

def connect(expr p, q, P, Q) =
    draw (p, 0) -- (p, q) -- (0, q) dashed evenly scaled 1/2;
    label.bot("\strut" & P, (p, 0)); label.lft("\strut" & Q, (0, q));
    draw (p, q) withpen pencircle scaled dotlabeldiam;
enddef;

drawoptions();
connect(a, b, "$a$", "$b$");  connect(am, am, "$am$", "$am$");
connect(b, a, "$b$", "$a$");  connect(gm, gm, "$gm$", "$gm$");
connect(rms, rms, "$rms$", "$rms$");
connect(ch, hm, "$cm$", "$hm$"); connect(hm, ch, "$hm$", "$cm$");

for p = (hm, am), (gm, rms), (am, ch), (am, hm), (rms, gm), (ch, am):
    draw p withpen pencircle scaled dotlabeldiam;
    undraw p withpen pencircle scaled 1/2 dotlabeldiam;
endfor

drawarrow xx; label.rt("$x$", point 1 of xx);
drawarrow yy; label.top("$y$", point 1 of yy);
```

Five means — and their means



— Roger B. Nelsen

```
path ff, xx, yy;
def f(expr x) = 1/256 mlog(x) / x enddef;
numeric minx, maxx, s, u, v;
minx = 13/8; s = 1/16; maxx = 19/4;
u = 89;
v = 3328-256;
ff = ((minx, f(minx)) for x=minx+s step s until maxx:
      .. (x, f(x))
    endfor) xscaled u yscaled v;
xx = (point 0 of ff -- (xpart point infinity of ff, ypart point 0 of ff)) shifted 30 down;
yy = (point 0 of ff -- point 0 of ff shifted (0, 0.1v)) shifted 30 left;

numeric pi, e, fpi, fe;
pi = 3.141592653589793 u; fpi = f(3.141592653589793) * v;
e  = 2.718281828459045 u; fe  = f(2.718281828459045) * v;

path ee, pp;
ee = (e, ypart point 0 of xx) -- (e, fe) -- (xpart point 0 of yy, fe);
pp = (pi, ypart point 0 of xx) -- (pi, fpi) -- (xpart point 0 of yy, fpi);

draw ee withcolor Reds 8 4;
draw pp withcolor Oranges 8 4;
draw ff withcolor Blues 8 7;

drawarrow xx;
drawarrow yy;

for x=2 upto 4:
    draw (down--up) scaled 2 shifted (x * u, ypart point 0 of xx);
    label.bot("$" & decimal x & "$", (x * u, ypart point 0 of xx - 2));
endfor
for y=32, 35, 38,:
    draw (left--right) scaled 2 shifted (xpart point 0 of yy, y/100 * v);
    label.lft("$" & decimal (y/100) & "$", (xpart point 0 of yy - 2, y/100 * v));
endfor

drawoptions(withcolor Reds 8 7);
label.bot("$e$", point 0 of ee shifted 4 down);
label.lft("$1/e$", point 2 of ee shifted 2 left);

drawoptions(withcolor Oranges 8 7);
label.bot("$\pi$", point 0 of pp shifted 4 down);
label.lft("$\ln\pi/\pi$", point 2 of pp shifted 2 left);

drawoptions(withcolor Blues 8 7);
label.urt("$\displaystyle y=\frac{\ln x}{x}$", point 42 of ff);

drawoptions();
label.rt("$x$", point 1 of xx);
label.top("$y$", point 1 of yy);
```

$e^\pi > \pi^e$



— Fouad Nakhli

```
path ff, xx, yy;
def f(expr x) = 1/256 mlog(x) enddef;
numeric minx, maxx, s, u, v, A, B, e;
minx = 1/4; s = 1/4; maxx = 12;
u = (tw-40)/maxx;
v = 89;
ff = ((minx, f(minx)) for x=minx+s step s until maxx:
       .. (x, f(x))
     endfor) xscaled u yscaled v;
xx = 24 left -- (xpart point infinity of ff + 10, 0);
yy = (0, ypart point 0 of ff) -- (0, ypart point infinity of ff + 10);
A = 1/2 maxx;
B = 3/4 maxx;
e = 2.718281828459;
primarydef o through p =
    (1+o/arclength(p))[point 1 of p, point 0 of p] --
    (1+o/arclength(p))[point 0 of p, point 1 of p]
enddef;

forsuffixes $=e, A, B:
    z$ = ($ * u, f($) * v);
    draw ($ * u, 0) -- z$;
    path p; p = 32 through (origin -- z$);
    draw p withcolor Reds 8 7;
    if not (str $ = "e"):
        draw thelabel.rt("$y=m_" & str $ & "x$", origin)
            rotated angle z$ shifted point 1 of p withcolor Reds 8 8;
    fi
endfor

drawoptions(withcolor Blues 8 7);
draw ff withpen pencircle scaled 3/4;
label.rt("$y=\ln x$", point 1/2 of ff shifted 4 right);

drawoptions();
drawarrow xx;
drawarrow yy;
label.rt("$x$", point 1 of xx);
label.top("$y$", point 1 of yy);

dotlabel.lrt("\strut $0$", (0, 0));
dotlabel.lrt("\strut $1$", (u, 0));
dotlabel.bot("\strut $e$", (e*u, 0));
dotlabel.bot("\strut $A$", (A*u, 0));
dotlabel.bot("\strut $B$", (B*u, 0));

label.bot(btex \vbox{\openup 12pt\halign{\hfil $#$&${}\quad
    \mathbin{\Longrightarrow}\quad #$\cr
    e \le A < B & m_A > m_B \cr
                & \frac{\ln A}{A} > \frac{\ln B}{B}\cr
                & A^B > B^A\cr
}} etex, point 1/2 of bbox currentpicture shifted 42 down);
```

$A^B > B^A$ for $e \leq A < B$



$$
\begin{aligned}
e \leq A < B \quad &\Longrightarrow \quad m_A > m_B \\
&\Longrightarrow \quad \frac{\ln A}{A} > \frac{\ln B}{B} \\
&\Longrightarrow \quad A^B > B^A
\end{aligned}
$$

— Charles D. Gallant

```
numeric a, b, c, d;
b + d + 16 = tw;
b = 2a;
a = 3/4 c = d;

path t[];
t1 =  origin -- (b,   0) -- (b,   a)   -- cycle;
t2 = (origin -- (d,   0) -- (d,   c)   -- cycle) shifted point 2 of t1;
t3 =  origin -- (b+d, 0) -- (b+d, a+c) -- cycle;

fill t1 withcolor Blues 8 1;
fill t2 withcolor Oranges 8 1;

forsuffixes $=1,2,3:
    draw subpath (1,3) of unitsquare scaled 6 rotated 90
         shifted point 1 of t$ withcolor 3/4;
endfor

draw subpath(1, 3) of t1;
draw subpath(-1, 1) of t2;
draw t3;

label.rt("$a$", point 3/2 of t1);
label.rt("$c$", point 3/2 of t2);
label.rt("$a$", 1/2[point 1 of t3, point 1 of t2]);

label.bot("$b$", point 1/2 of t1);
label.bot("$d$", point 1/2 of t2);
label.bot("$b$", 1/2[point 1 of t3, point 1 of t1]);

label.top(btex $\displaystyle \frac{a}{b} < \frac{c}{d} \quad \Rightarrow \quad
    \frac{a}{b} < \frac{a+c}{b+d} < \frac{c}{d}$ etex,
    point 5/2 of bbox currentpicture shifted 42 up);
```

The mediant property

$$\frac{a}{b} < \frac{c}{d} \quad \Rightarrow \quad \frac{a}{b} < \frac{a+c}{b+d} < \frac{c}{d}$$



— Richard A. Gibbs

```
path xx, yy;
xx = 12 left -- 233 right;
yy = 12 down -- 144 up;

z1 = .95(1/4[point 1 of xx, point 1 of yy]);
z3 = .95(3/4[point 1 of xx, point 1 of yy]);
z2 = 1/2[z1, z3];

drawoptions(withcolor Blues 8 7);
draw origin -- z1 -- z3 -- cycle;
draw origin -- z2;
picture m; m = image(draw (up--down) scaled 2 rotated -5 shifted 1/2 left;
                     draw (up--down) scaled 2 rotated -5 shifted 1/2 right);
draw m rotated angle (z3 - z1) shifted 1/2[z1, z2];
draw m rotated angle (z3 - z1) shifted 1/2[z2, z3];

draw thelabel.top("$y=m_1 x$", origin) rotated angle z1 shifted 3/4 z1;
draw thelabel.top("$y=m_2 x$", origin) rotated angle z2 shifted 3/4 z2;
draw thelabel.top("$y=m_3 x$", origin) rotated angle z3 shifted 3/4 z3;

drawoptions();
drawarrow xx; label.rt("$x$", point 1 of xx);
drawarrow yy; label.top("$y$", point 1 of yy);

dotlabel.urt("$(b,a)$", z1);
dotlabel.urt("$\left(\frac{b+d}{2},\frac{a+c}{2}\right)$", z2);
dotlabel.urt("$(d,c)$", z3);
drawdot origin withpen pencircle scaled dotlabeldiam;

label.top(btex $\displaystyle
    a, b, c, d > 0; \quad \frac{a}{b} < \frac{c}{d}
    \quad \Longrightarrow \quad
    \frac{a}{b} < \frac{a+c}{b+d} < \frac{c}{d}$
    etex, point 5/2 of bbox currentpicture shifted 13 up);
label.bot(btex $m_1 < m_3 \quad \Longrightarrow \quad m_1 < m_2 < m_3$
    etex,  point 1/2 of bbox currentpicture shifted 13 down);
```

Regle des nombres moyens – I

$$a, b, c, d > 0; \quad \frac{a}{b} < \frac{c}{d} \quad \Longrightarrow \quad \frac{a}{b} < \frac{a+c}{b+d} < \frac{c}{d}$$



$$m_1 < m_3 \quad \Longrightarrow \quad m_1 < m_2 < m_3$$

— Li Changming

```
numeric a, b, c, d, u;
u = 89;
55a = 34b; b = d = 1;
64c = 61d;

path A, B, C, D, A', B';

A =  unitsquare scaled u yscaled (a/b);
A' = unitsquare scaled u yscaled (c/d);
B =  unitsquare scaled u yscaled (a/b) xscaled (b/(b+d))
     shifted point 1 of A shifted 42 right;
B' = A' shifted point 0 of B;
C =  unitsquare scaled u yscaled (c/d) xscaled (d/(b+d))
     shifted point 1 of B;
D = A' shifted point 1 of C shifted 42 right;

draw A' dashed evenly scaled 1/2;
draw B' dashed evenly scaled 1/2;

fill A withcolor Oranges 8 2;
fill B withcolor Oranges 8 2;
fill C withcolor Blues 8 2;
fill D withcolor Blues 8 2;

draw A; draw B; draw C; draw D;

vardef superlabel(expr t, z) =
    interim bboxmargin := 6;
    save P; picture P; P = thelabel(t, origin);
    save s; path s; s = superellipse(point 3/2 of bbox P, point 5/2 of bbox P,
        point 7/2 of bbox P, point 1/2 of bbox P, 0.78);
    unfill s shifted z; draw P shifted z;
enddef;

superlabel("$\frac{a}{b}$",   center A);
superlabel("$\frac{a}{b+d}$", center B);
superlabel("$\frac{c}{b+d}$", center C);
superlabel("$\frac{c}{d}$",   center D);

input arrow_label
arrow_label(point 0 of A, point 1 of A, "$1$",            14);
arrow_label(point 0 of B, point 1 of B, "$\frac{b}{b+d}$", 14);
arrow_label(point 0 of C, point 1 of C, "$\frac{d}{b+d}$", 14);
arrow_label(point 0 of D, point 1 of D, "$1$",            14);

arrow_label(1/2[point 1 of A, point 0 of B], 1/2[point 2 of A, point 3 of B],
    "$\frac a b$", 0);
arrow_label(1/2[point 1 of C, point 0 of D], 1/2[point 2 of C, point 3 of D],
    "$\frac c d$", 0);

label.bot(btex $\displaystyle
    \frac{a}{b} < \frac{a}{b+d} + \frac{c}{b+d} < \frac{c}{d}
    $ etex, point 1/2 of bbox currentpicture shifted 21 down);
```

Regle des nombres moyens – II



$$\frac{a}{b} < \frac{a}{b+d} + \frac{c}{b+d} < \frac{c}{d}$$

— Roger B. Nelsen

```
input superlabel

picture P[];

P1 = image(numeric x; x = 89/72; path box[];
  box1 = unitsquare scaled 72 xscaled x yscaled (1/x);
  box2 = box1 rotated 90;  box2 := box2 shifted (point 1 of box1 - point 3 of box2);
  box3 = box1 rotated 180; box3 := box3 shifted (point 1 of box2 - point 3 of box3);
  box4 = box1 rotated 270; box4 := box4 shifted (point 1 of box3 - point 3 of box4);

  forsuffixes $=1,2,3,4:
    fill box$ withcolor if odd $: Oranges else: Blues fi 8 2;
    superlabel("$1$", center box$);
    draw subpath (-2, 1) of box$;
  endfor
  label.bot("\strut$x$", point 1/2 of box1); label.bot("\strut$\frac1x$", point -1/2 of box2);
  label.lft("$x$", point 1/2 of box4); label.lft("$\frac1x$", point -1/2 of box1);
  label.top("I.", point 3 of bbox currentpicture shifted 3.25 up));
P2 = image(path xx, yy; xx = 8 left -- 150 right; yy = xx rotated 90;
  numeric u; u = 100/3;
  for i=1 upto 4:
    draw (left--right) shifted (0, i*u);
    draw (up--down) shifted (i*u, 0);
  endfor
  drawarrow xx; label.rt("$x$", point 1 of xx);
  drawarrow yy; label.top("$y$", point 1 of yy);
  path a, b;
  a = ((-24/100, 224/100) -- (224/100, -24/100)) scaled u;
  b = ((1,1) for x=1+1/8 step 1/8 until 4: -- (x, 1/x) endfor) scaled u;
  b := reverse b reflectedabout(origin, point 0 of b) & b;

  draw b withcolor Oranges 8 7;
  draw a withcolor Blues 8 7;
  label.urt("$y=\frac1x$", point 40 of b) withcolor Oranges 8 7;
  label.lrt("$y=2-x$", point 1 of a shifted 3 up) withcolor Blues 8 7;
  label.llft("II.", point 3 of bbox currentpicture));
numeric x, u; x = 7/4; u = 80;
z1 = (0, x - 1/x) scaled u; z2 = (-2, 0) scaled u; z3 = -z2;
z4 = origin rotatedabout(z1, 90); z5 = whatever[z1, z3]; x5 = x4;

P3 = image(draw unitsquare scaled 8 rotated 90 withcolor 1/2;
  draw origin -- z1 -- z2 -- cycle;
  label.bot("$2$", 1/2 z2); label.rt("$x-\frac{1}{x}$", 1/2z1);
  label.ulft("$x+\frac{1}{x}$", 1/2[z1, z2]);
  label("III.", point 3 of bbox currentpicture));
P4 = image( fill origin -- z1 -- z3 -- cycle withcolor Oranges 8 2;
  fill z4 -- z1 -- z5 -- cycle withcolor Blues 8 2;
  draw z1 -- origin -- z3 -- z1 -- z4 -- z5;
  draw z5 -- (x5, 0) dashed withdots scaled 1/2;

  label.bot("$x$", 1/2 z3); label.top("$1$", 1/2[z1, z4]);
  label.lft("$1$", 1/2 z1); label.rt("$\frac1x$", 1/2[z4, z5]);
  label("IV.", point 3 of bbox currentpicture shifted 4 left));

draw P1 shifted 180 left; draw P2 shifted 32 right;
draw P3 shifted (-32, -180); draw P4 shifted (+32, -180);
```

The sum of a positive number and its reciprocal is at least two



— Roger B. Nelsen

```
numeric u, v, a, b, halfpi; u = 160; v = 89; a = 1.2; b = 0.8a; halfpi = 1.570796;

vardef sin(expr x) = sind(57.2957795 x) enddef;
vardef cos(expr x) = cosd(57.2957795 x) enddef;
vardef tan(expr x) = sin(x) / cos(x) enddef;

numeric ma; ma = a + 1/8;
path ss, tt, sbb, tbb;
ss = origin for x=1/32 step 1/32 until ma+eps: -- (x * u, sin(x) * v) endfor;
tt = origin for x=1/32 step 1/32 until a+1/32: -- (x * u, tan(x) * v) endfor;
sbb = origin -- (ma * u, sin(b) / b * ma * v);
tbb = origin -- (ma * u, tan(b) / b * ma * v);

draw ss  withcolor Blues 8 7; draw sbb withcolor Blues 8 5;
draw tt  withcolor Oranges 8 7; draw tbb withcolor Oranges 8 5;

for $=a,b:
  draw ($*u, 0) -- ($*u, tan($)*v) dashed evenly scaled 1/2 withpen pencircle scaled 1/4;
  draw ($*u, sin($)*v) withpen pencircle scaled dotlabeldiam;
  draw ($*u, tan($)*v) withpen pencircle scaled dotlabeldiam;
endfor
draw (a*u, sin(b)/b*a*v) withpen pencircle scaled dotlabeldiam;
draw (a*u, tan(b)/b*a*v) withpen pencircle scaled dotlabeldiam;

path xx, yy; xx = 12 left -- 12 right shifted (halfpi * u, 0); yy = xx rotated 90;
drawarrow xx; label.rt("$x$", point 1 of xx);
drawarrow yy; label.top("$y$", point 1 of yy);

label.rt("$y=\sin x$", point infinity of ss shifted down);
label.top("$y=\tan x$", point infinity of tt);

label.urt("$y=\frac{\sin \beta}{\beta}x$", point infinity of sbb shifted 4 down);
label.urt("$y=\frac{\tan \beta}{\beta}x$", point infinity of tbb shifted 3 down);

vardef hbarlabel@#(expr t, z) =
  draw (left--right) scaled 3/2 shifted z;
  interim labeloffset := 5; label@#(t, z);
enddef;
vardef vbarlabel@#(expr t, z) =
  draw (down--up) scaled 3/2 shifted z;
  interim labeloffset := 5; label@#(t, z);
enddef;

hbarlabel.lft("$1$", (0, v));
dotlabel.llft("$0$", origin);
vbarlabel.bot("\strut $\beta$", (b * u, 0));
vbarlabel.bot("\strut $\alpha$", (a * u, 0));
vbarlabel.bot("\strut $\pi/2$", (halfpi * u, 0));

label.top(btex $\displaystyle
  0 < \beta < \alpha < \frac{\pi}{2} \implies
  \frac{\sin\alpha}{\sin\beta} < \frac{\alpha}{\beta} < \frac{\tan\alpha}{\tan\beta}
  $ etex, point 5/2 of bbox currentpicture shifted 42 up);
label.bot(btex $\displaystyle
  \sin\alpha < \frac{\sin\beta}{\beta}\alpha, \qquad
  \tan\alpha > \frac{\tan\beta}{\beta}\alpha $ etex,
  point 1/2 of bbox currentpicture shifted 42 down);
label.bot(btex $\displaystyle \therefore\quad
  \frac{\sin\alpha}{\sin\beta} < \frac{\alpha}{\beta} < \frac{\tan\alpha}{\tan\beta}
  $ etex, point 1/2 of bbox currentpicture shifted 8 down);
```

Aristarchus' inequalities

$$0 < \beta < \alpha < \frac{\pi}{2} \implies \frac{\sin\alpha}{\sin\beta} < \frac{\alpha}{\beta} < \frac{\tan\alpha}{\tan\beta}$$



$$\sin\alpha < \frac{\sin\beta}{\beta}\alpha, \qquad \tan\alpha > \frac{\tan\beta}{\beta}\alpha$$

$$\therefore \quad \frac{\sin\alpha}{\sin\beta} < \frac{\alpha}{\beta} < \frac{\tan\alpha}{\tan\beta}$$

— Roger B. Nelsen

```
path t[]; z0 = 89 dir 280; z1 = 75 dir 200;
t1 = (x1, y0) -- z0 -- z1 -- cycle;
t2 = (xpart point 2 of (t1 rotated 180), ypart point 1 of t1)
         -- point 2 of (t1 rotated 180)     -- point 1 of t1 -- cycle;

picture P[];

for i=1,2:
  if i = 2:
    t1 := t1 rotated - angle (point 1 of t1 - point 2 of t1);
    t1 := t1 shifted - point 3/2 of t1;
    t1 := t1 shifted - (0, 1/2 abs (point 1 of t2 - point 2 of t2));

    t2 := t2 rotated (90 - angle (point 1 of t2 - point 2 of t2));
    t2 := t2 shifted (point 1 of t1 - point 2 of t2);
  fi

  P[i] = image(
    if i = 2:
      draw unitsquare scaled 6 shifted point 2 of t1;
      label.top("$\sqrt{a^2+b^2}$", point 3/2 of t1);
      label.lft("$\sqrt{x^2+y^2}$", point 3/2 of t2);
    fi
    forsuffixes r=0, 180:
      fill t1 rotated r withcolor Blues 8 2;
      fill t2 rotated r withcolor Oranges 8 2;
      draw unitsquare scaled 6 rotated angle (point 1 of t1 - point 0 of t1)
                               shifted point 0 of t1 rotated r;
      draw unitsquare scaled 6 rotated angle (point 1 of t2 - point 0 of t2)
                               shifted point 0 of t2 rotated r;
      draw t1 rotated r; draw t2 rotated r;
      if i=1:
        label("$|a|$", point  3/7 of t1 shifted 8 down rotated r);
        label("$|b|$", point -3/7 of t1 shifted 8 left rotated r);
        label("$|x|$", point  3/7 of t2 shifted 8 right rotated r);
        label("$|y|$", point -3/7 of t2 shifted 8 down rotated r);
      fi
    endfor
  );
endfor

label.lft(P1, 12 left); label("$\le$", origin); label.rt(P2, 12 right);

label.top(btex $
\left|  \langle a,b \rangle \cdot            \langle x,y \rangle \right| \le
\left\| \langle a,b \rangle \right\| \left\| \langle x,y \rangle \right\|
$ etex, point 5/2 of bbox currentpicture shifted 42 up);

label.bot(btex $
\left(|a|+|y|\right)\left(|b|+|x|\right) \le
2\left(\frac12|a||b| + \frac12|x||y|\right) + \sqrt{a^2 + b^2} \sqrt{x^2 + y^2}
$ etex, point 1/2 of bbox currentpicture shifted 21 down);

label.bot(btex $\therefore\quad
\left| ax+by \right| \le |a||x|+|b||y| \le \sqrt{a^2 + b^2} \sqrt{x^2 + y^2}
$ etex, point 1/2 of bbox currentpicture shifted 16 down);
```

The Cauchy-Schwartz inequality

$$|\langle a, b\rangle \cdot \langle x, y\rangle| \leq \|\langle a, b\rangle\| \, \|\langle x, y\rangle\|$$



$$\left(|a| + |y|\right)\left(|b| + |x|\right) \leq 2\left(\tfrac{1}{2}|a||b| + \tfrac{1}{2}|x||y|\right) + \sqrt{a^2 + b^2}\sqrt{x^2 + y^2}$$

$$\therefore \quad |ax + by| \leq |a||x| + |b||y| \leq \sqrt{a^2 + b^2}\sqrt{x^2 + y^2}$$

— Roger B. Nelsen

```
numeric u; u = 53;
path xx, yy; xx = (-12, 0) -- (12 + 2.5u, 0); yy = (0, -12-u) -- (0, 12 + 2.5u);
path ff, dff; numeric s; s = 1/8;
vardef f(expr x) = x**1.9 - 1 enddef;
ff = ((0, f(0)) for x = s step s until 2: .. (x, f(x)) endfor) scaled u;
dff = (3/2 left -- 2 right) scaled u
  rotated angle direction 1/s of ff shifted point 1/s of ff;

picture P[];
P1 = image(
  drawoptions(withcolor Blues 7 5);
  draw dff; label.rt("$y=r(x-1)$", point infinity of dff);

  drawoptions(withcolor Blues 7 7);
  draw ff; label.rt("$y=x^r - 1$", point infinity of ff);

  drawoptions();
  drawarrow xx; label.rt("$x$", point 1 of xx);
  drawarrow yy; label.top("$y$", point 1 of yy);

  dotlabel.llft("$0$", origin); dotlabel.lft("$-1$", u * down);
  dotlabel.lrt("$1$", u * right);
  unfill fullcircle scaled 3/4 dotlabeldiam shifted (u * right);

  label.top("\hbox to \textwidth{I. First semester calculus\hss}",
        point 5/2 of bbox currentpicture shifted 8 up);
);
% P2 is made up of several subpictures ...
% ... see the document source for details.

P2 = image(
  label.lft(P21, 12 left); label.rt(P22, 12 right);
  label.top("\hbox to \textwidth{II. Second semester calculus\hss}",
    point 5/2 of bbox currentpicture shifted 8 up);
);

label.top(P1, 7 up); label.bot(P2, 7 down);
label.top("$x>0$, $x \ne 1$, $r>1$: $x^r - 1 > r(x-1)$",
    point 5/2 of bbox currentpicture shifted 13 up);
```

Bernoulli's inequality

$$x > 0, x \neq 1, r > 1\colon x^r - 1 > r(x-1)$$

I. First semester calculus



II. Second semester calculus



$$x^r - 1 = \int_1^x rt^{r-1}\,dt > r(x-1)$$



$$1 - x^r = \int_x^1 rt^{r-1}\,dt < r(1-x)$$

— Roger B. Nelsen

```
numeric u, v, a, b; u = 64; v = 42; a = 3; b = 21;

vardef f(expr x) = (x * u, 1/256 mlog(x) * v) enddef;
path ff; ff = f(1/8) for t = 1/4 step 1/8 until 4: .. f(t) endfor;
path yy; yy = (0, ypart point 0 of ff - 12) -- (0, ypart point infinity of ff + 12);
path xx; xx = 12 left -- 12 right + 4u * right; picture P[];
P1 = image(path L[];
  L1 = (left--right) scaled u rotated angle direction b of ff shifted point b of ff;
  L3 = (left--right) scaled u rotated angle direction a of ff shifted point a of ff;
  L2 = point a of ff -- point b of ff;

  drawoptions(dashed withdots scaled 1/4 withcolor Blues 7 6);
    forsuffixes $=1,2,3: draw L$; endfor
    label.ulft("$L_1$", point 7/8 of L1); label.ulft("$L_3$", point 7/8 of L3);
    label.lrt("$L_2$", point 1/4 of L2);

  drawoptions(withcolor Reds 7 7);
    draw ff; label.rt("\rlap{$y=\ln x$}", point infinity of ff);

  drawoptions();
  forsuffixes $=a, b:
    draw (up--down) shifted (xpart point $ of ff, 0);
    label.top("$" & str $ & "$", (xpart point $ of ff, 0));

    fill fullcircle scaled dotlabeldiam shifted point $ of ff;
    unfill fullcircle scaled 2/3 dotlabeldiam shifted point $ of ff;

  endfor

  drawarrow xx; label.rt("$x$", point 1 of xx);
  drawarrow yy; label.top("$y$", point 1 of yy);
  label.bot("$\hbox{slope of $L_1$}<\hbox{slope of $L_2$}<\hbox{slope of $L_3$}$",
    point 1/2 of bbox currentpicture);
  label.top("\hbox to \textwidth{I. First semester calculus\hss}",
    point 5/2 of bbox currentpicture shifted 8 up));
path ff; ff = (1/4 u, 4 v) for x=3/8 step 1/8 until 4: .. (x * u, v / x) endfor;
path yy; yy = 12 down -- (0, 12 + ypart point 0 of ff);

P2 = image(path A, B, C;
  z0 = point a of ff; z1 = point b of ff; x2=x0; x3=x1; y2=y3=0;
  A = unitsquare xscaled (x1-x0) yscaled y0 shifted z2;
  B = unitsquare xscaled (x1-x0) yscaled y1 shifted z2;
  C = z3 -- z2 -- subpath(a,b) of ff -- cycle;
  fill C withcolor Blues 7 1; fill B withcolor Blues 7 2;
  draw A withcolor 1/2; draw B withcolor 1/2;

  label.bot("\strut$a$", z2); label.bot("\strut$b$", z3);

  drawoptions(withcolor Reds 7 7);
  draw ff; label.top("\rlap{$y=\frac1x$}", point infinity of ff);
  drawoptions();
  drawarrow xx; label.rt("$x$", point 1 of xx);
  drawarrow yy; label.top("$y$", point 1 of yy);
  label.bot("$\frac1b(b-a) < \int_a^b \frac1x\,dx < \frac1a(b-a)$",
    point 1/2 of bbox currentpicture);
  label.top("\hbox to \textwidth{II. Second semester calculus\hss}",
    point 5/2 of bbox currentpicture shifted 8 up));


label.top(P1, 7 up); label.bot(P2, 7 down);

label.top("$b>a>0$ implies $\displaystyle\frac{1}{b}<\frac{\ln b-\ln a}{b-a}<\frac{1}{a}$",
  point 5/2 of bbox currentpicture shifted 13 up);
```

Napier's inequality

$$b > a > 0 \text{ implies } \frac{1}{b} < \frac{\ln b - \ln a}{b - a} < \frac{1}{a}$$

I. First semester calculus



slope of L_1 < slope of L_2 < slope of L_3

II. Second semester calculus



$$\frac{1}{b}(b-a) < \int_a^b \frac{1}{x}\,dx < \frac{1}{a}(b-a)$$

— Roger B. Nelsen

Integer sums

```
input paintball
numeric n; n = 7;
for i=1 upto n-1:
  for j=1 upto n:
    draw if j > i: bball else: rball fi shifted ((j, -i) scaled 280/n);
    if i=j:
      draw (up--origin--right) shifted (j+1/2, -i-1/2) scaled (280/n);
    fi
  endfor
endfor
% remove the extra parts of the stepped line
clip currentpicture to bbox currentpicture scaled 0.975;
label.bot("$1+2+\cdots + n = \frac12 n (n+1)$",
  point 1/2 of bbox currentpicture shifted 21 down);
```

Sums of integers I



$$1 + 2 + \cdots + n = \tfrac{1}{2}n(n+1)$$

— Ancient Greek

```
numeric n; n = 7;
path t; t = (origin -- right -- up -- cycle) rotatedabout(1/2[right, up], 180);
for i=1 upto n:
  for j=0 upto n-i:
    draw (up--origin--right) shifted (i, j) scaled (280/n);
  endfor
  fill t shifted (i, n-i) scaled (280/n) withcolor Purples 8 3;
  draw t shifted (i, n-i) scaled (280/n);
endfor
label.bot("$1+2+\cdots + n = \frac{n^2}2 + \frac n2$",
  point 1/2 of bbox currentpicture shifted 21 down);
```

Sums of integers II



$$1 + 2 + \cdots + n = \frac{n^2}{2} + \frac{n}{2}$$

— Ian Richards

```
input paintball
numeric n; n = 10;
picture half;
half = image(
  for i=1 upto n:
    for j=1 upto i:
      draw if odd i: rball else: bball fi shifted ((i, j) scaled (280/n));
    endfor
    if i < n:
      draw (origin -- i * up) shifted (i+1/2, 1/2) scaled (280/n);
    fi
  endfor
);
draw half; draw half reflectedabout(origin, (1, 1));

label.bot("$1+3+5+\cdots + (2n-1) = n^2$",
  point 1/2 of bbox currentpicture shifted 21 down);
```

Sums of odd integers I



$$1 + 3 + 5 + \cdots + (2n - 1) = n^2$$

— Nichomachus of Gerasa

```
input paintball
numeric n, u; n = 10; u = 280/n;
for i=0 upto 3:
  for j=1 upto floor (n/2):
    for k=j upto n-j:
      draw if i=0: bball else: rball fi
        shifted ((k, j) scaled u)
        rotatedabout((n/2+1/2,n/2+1/2) scaled u, 90i);
      if k=j:
        draw (down--origin--right)
          shifted (k-1/2, j+1/2) scaled u
          rotatedabout((n/2+1/2,n/2+1/2) scaled u, 90i);
      fi
    endfor
  endfor
endfor
interim bboxmargin := -1/4;
draw bbox currentpicture;
label.bot("$1+3+\cdots + (2n-1) = \frac14\left(2n\right)^2 = n^2$",
  point 1/2 of bbox currentpicture shifted 24 down);
```

Sums of odd integers II



$$1 + 3 + \cdots + (2n - 1) = \frac{1}{4}(2n)^2 = n^2$$

— Roger B. Nelsen

```
picture P[]; path t; numeric n; n = 6;
t = for i=0 upto 2: 16 down rotated 120i -- endfor cycle;

pair u, v;
u = point 2 of t - point 0 of t;
v = point 1 of t - point 2 of t;

P1 = image(
    fill t withcolor Blues 8 2; label("$\Delta$", origin);
    for i=0 upto n-1:
        for j=0 upto i:
            draw t shifted (j * v) shifted (i * u);
        endfor
        label.lrt(
            if i=n-1: "$2n-1$"
            elseif i=n-2: "\rotatebox{105}{$\ddots$}"
            else: "$" & decimal (2i+1) & "$" fi,
            point 2/3 of t shifted (i * v) shifted (i * u));
    endfor
);
P2 = image(
    path s, s';
    s = t shifted -(xpart point 1 of t, ypart point 0 of t);
    s' = s scaled n;

    forsuffixes $=s, s':
        fill $ withcolor Blues 8 2; draw $;
    endfor
    z1 = point 2 of s' scaled 1.1;
    drawoptions(dashed evenly scaled 1/2 withcolor 1/2);
    drawarrow origin -- z1;
    drawarrow origin -- (x1, 0);
    drawarrow origin -- (0, y1);
    drawoptions();
    label("$\Delta$", 1/3[point 3/2 of s, point 0 of s]);
    label("$n^2\Delta$", 1/3[point 3/2 of s', point 0 of s']);
    label.llft("$1$", point -2/3 of s);
    label.llft("$n$", point -1/2 of s');
);

interim bboxmargin := 12;
label.ulft(P1, origin);
label.urt(P2, origin shifted down);

label.bot("$\Delta+3\cdot\Delta+\cdots + (2n-1)\cdot\Delta = A = n^2\cdot\Delta$",
    point 1/2 of bbox currentpicture shifted 24 down);
label.bot("$\displaystyle\sum_{i=1}^n \left(2i - 1\right) = n^2$",
    point 1/2 of bbox currentpicture shifted 4 down);
```

Sums of odd integers III



$$\Delta + 3 \cdot \Delta + \cdots + (2n-1) \cdot \Delta = A = n^2 \cdot \Delta$$

$$\sum_{i=1}^{n} (2i-1) = n^2$$

— Jenő Lehel

```
input paintball
for n=2 upto 4:
    numeric y; y = - 20 n * n;
    picture p; p = image(
        for i=1 upto n:
            for j=1 upto n:
                draw rball shifted 28(i, j);
            endfor
        endfor
        path b; b = bbox currentpicture;
        drawoptions(dashed evenly withpen pencircle scaled 1/4);
        for i=1 upto n-1:
            for j=-1, 1:
                draw (left-- 2 right) scaled 40n rotated 45 shifted ((28i-14)*j, 0);
            endfor
        endfor
        drawoptions();
        clip currentpicture to b;
    );
    label(p, (-80, y));
    label("$"
        for i=1 upto n: & decimal i & "+" endfor
        for i=n-1 downto 2: & decimal i & "+" endfor
        & "1 =" & decimal n & "^2$",
        (80, y));
endfor

label.bot("$1+2+\cdots+(n-1) + n + (n-1) + \cdots + 2 + 1 = n^2$",
    point 1/2 of bbox currentpicture shifted 42 down);
```

Squares and sums of integers I



$$1+2+1=2^2$$



$$1+2+3+2+1=3^2$$



$$1+2+3+4+3+2+1=4^2$$

$$1+2+\cdots+(n-1)+n+(n-1)+\cdots+2+1=n^2$$

— Ancient Greek

```
input paintball
for n=2 upto 4:
    numeric y; y = - 24 n * n;
    picture p; p = image(
        for i=1 upto n:
            for j=1 upto n:
                draw rball shifted 28(i, j);
            endfor
        endfor
    );
    picture q; q = image(
        for i=1 upto n-1:
            for j=1 upto n-1:
                draw bball shifted 28(i+1/2, j+1/2);
            endfor
        endfor
    );
    picture r; r = image(
        drawoptions(dashed evenly withpen pencircle scaled 1/4);
        for i=1 upto n-1:
            for j=-1, 1:
                draw (left-- 2 right) scaled 40n rotated 45 shifted ((28i-14)*j, 0);
            endfor
        endfor
        drawoptions();
        clip currentpicture to bbox p;
    );
    label(p, (-144, y + 42));
    label(q, (-144, y + 42));
    label(r, (-144, y + 42));
    label("$=$", (-64, y+42));
    label(p, (144, y + 42));
    label("$+$", (64, y+42));
    label(q, (0, y + 42));

    label.bot("$"
        for i=1 step 2 until 2n-1: & decimal i & "+" endfor
        for i=2n-3 step -2 until 3: & decimal i & "+" endfor
        & "1 =" & decimal (n-1) & "^2 + " & decimal n & "^2$",
        (0, y - 3n));
endfor

label.bot("$\vdots$", point 1/2 of bbox currentpicture shifted 21 down);

label.bot("$1+3+\cdots+(2n-1) + (2n+1) + (2n-1) + \cdots + 3 + 1 = n^2 + (n+1)^2$",
    point 1/2 of bbox currentpicture shifted 21 down);
```

Squares and sums of integers II



$$1 + 3 + 1 = 1^2 + 2^2$$



$$1 + 3 + 5 + 3 + 1 = 2^2 + 3^2$$



$$1 + 3 + 5 + 7 + 5 + 3 + 1 = 3^2 + 4^2$$

$$\vdots$$

$$1 + 3 + \cdots + (2n - 1) + (2n + 1) + (2n - 1) + \cdots + 3 + 1 = n^2 + (n + 1)^2$$

— Hee Sik Kim

```
numeric u; u = 21;
vardef folded_bar(expr n, k) =
  if k < 2:
    unitsquare xscaled (n + k)
  else:
    numeric p, q;
    if odd k:
      p = n + 1/2 (k - 1);
      q = n + k - p - 1;
      origin -- (p, 0) -- (p, -q) -- (p+1, -q) -- (p+1, 1)
    else:
      p = n + 1/2 k;
      q = n + k - p + 1;
      origin -- (p, 0) -- (p, q) -- (p-1, q) -- (p-1, 1)
    fi -- up -- cycle
  fi
enddef;

numeric y; y = 0;
for n = 1 upto 4:
  y := y - 2(n-1) * u - u;
  for k = 0 upto 2n - 2:
    path s; s = folded_bar(n, k) scaled u
      shifted (0, y + 1/2 u * (if odd k: k+1 else: -k fi));
    fill s withcolor
      if k = 0: Purples 8 2
      elseif odd floor(k/2): Blues 8 2
      else: Oranges 8 2
      fi;
  endfor

  drawoptions(dashed withdots scaled 1/2);
  for i = 1 upto 2n - 2:
    draw ((n-1) * down -- n * up)   shifted (i, 0)     scaled u shifted (0, y);
    draw (origin -- (2n-1) * right) shifted (0, i-n+1) scaled u shifted (0, y);
  endfor
  drawoptions();

  for k = 0 upto 2n - 2:
    path s; s = folded_bar(n, k) scaled u
      shifted (0, y + 1/2 u * (if odd k: k+1 else: -k fi));
    draw s;
  endfor
endfor


label.rt(btex \vbox{\openup 6pt \halign{$#$\hss\cr
  n = 4\cr 4 + 5 + 6 + 7 + 8 + 9 + 10 = 7^2\cr}} etex,
  (8u, y + 1/2 u));

label.llft(btex
  $\displaystyle \sum_{k=n}^{3n-2} k = \left(2n-1\right)^2$;\quad
  $n=1,2,3,\dots$
  etex, urcorner currentpicture);
```

Arithmetic progressions with sum equal to square of number of terms

$$\sum_{k=n}^{3n-2} k = (2n-1)^2; \quad n = 1, 2, 3, \ldots$$



$n = 4$

$4 + 5 + 6 + 7 + 8 + 9 + 10 = 7^2$

— James O. Chilaka

```
input isometric_projection
set_projection(22, -34);
picture P[], b_cube, b_semicube, r_semicube;
b_cube = cube(Blues 8 4, Blues 8 2, background);
b_semicube = semicube(Blues 8 4, Blues 8 2, background);
r_semicube = semicube(Reds 8 6, Reds 8 4, Reds 8 2);

P1 = image(
    pair a, b; a = p(0, 2, 3); b = p(-5, 2, 0);
    for k= 0 upto 3:
        for j = k upto 3:
            for i = k upto 3:
                draw b_cube shifted p(i-3, k, 3-j);
                draw b_cube shifted a shifted p(i-3, j, 3-k);
                draw b_cube shifted b shifted p(k, i, 3-j);
            endfor
        endfor
    endfor
    label.lrt("$n$", p(0, 0, 2));
    label.llft("$n$", p(-2, 0, 0));
);
P2 = image(
    % .. as P1 but with a and b rather smaller
);
P3 = image(
    for i=-4 upto 0:
        for j = 0 upto 3:
            for k = 3 downto 0:
                draw b_cube shifted p(i, j, k);
            endfor
        endfor
    endfor
    for k = 3 downto 0:
        for i = -k upto 0:
            draw b_semicube shifted p(i, 4, k);
            draw r_semicube  shifted p(i, 4.5, k);
        endfor
    endfor
);
P4 = image(
    % .. as P3 except for top layer, and labels
);
draw P1 shifted 12 down; draw P2 shifted 243 right;
draw P3 shifted 233 down; draw P4 shifted 233 down shifted 243 right;

label.top(btex $
  1^2 + 2^2 + \cdots + n^2 =
  \frac13 n \left(n + 1\right)\left(n + \frac12\right)
  $ etex, point 5/2 of bbox currentpicture shifted 21 up);
```

Sums of squares I

$$1^2 + 2^2 + \cdots + n^2 = \frac{1}{3}n\,(n+1)\left(n+\frac{1}{2}\right)$$



— Man-Keung Siu

```
numeric u, N; u = 14.4; N = 6;
picture P[];
P1 = image(
    for n = 1 upto N:
        for m = n downto 1:
            path s; s = unitsquare scaled (-m * u) shifted (0, - n / 2 * n * u);
            fill s withcolor Spectral[7][m+1];
            for k = 1 upto m - 1:
                draw subpath (2, 3) of s shifted (0, k*u);
                draw subpath (1, 2) of s shifted (k*u, 0);
            endfor
            draw s withpen pencircle scaled 3/2;
        endfor
    endfor
);
P2 = image(
    for n = 1 upto N:
        path s; s = unitsquare xscaled (2n - 1) yscaled (N - n + 1) scaled u
        shifted ((N + 1 - n, -2 - N * n + n*(n-1)/2) scaled u);
        fill s withcolor Spectral[7][1+n];
        for k = 0 upto N - n:
            draw (left--right) scaled (N+1/2) scaled u
                shifted point 1/2 of s shifted (0, k*u)
                if k=0: withpen pencircle scaled 3/2 fi;
        endfor
        draw s withpen pencircle scaled 3/2;
    endfor
    for k=0 upto 2N + 1:
        draw ((k, -2) -- (k, - 2 - N / 2 * (N + 1))) scaled u
            if (k=0) or (k=2N+1): withpen pencircle scaled 3/2 fi;
    endfor
    draw ((0, -2) -- (2N+1, -2)) scaled u withpen pencircle scaled 3/2;
    label.bot("$2n+1$", point 1/2 of bbox currentpicture);
    label.lft(TEX("$1+2+\cdots+n$") rotated 90, point -1/2 of bbox currentpicture);
);


draw P1;
draw P2 shifted 80 right;

label.top(btex
$3\left(1^2 + 2^2 + \cdots + n^2\right) =
    \left(2n+1\right)\left(1 + 2 + \cdots + n\right)$ etex,
    point 5/2 of bbox currentpicture shifted 21 up);
```

Sums of squares II

$$3\left(1^2 + 2^2 + \cdots + n^2\right) = (2n+1)\,(1+2+\cdots+n)$$



— Dan Kalman

```
numeric u, N; u = 14.4; N = 6;

numeric x, y; x = 0; y = -1;
for n = 1 upto N:
    label("$" & decimal n & "$", (x + n / 2, 1/2) scaled u);
    draw unitsquare scaled 1/2 (n * n + n) scaled u rotated -90
        if n < N: dashed evenly scaled 1/2 fi;
    fill unitsquare scaled n shifted (x, y) scaled u withcolor Purples 8 3;
    for m = 0 upto n:
        draw (origin -- n * right) shifted (x, y + m) scaled u;
        draw (origin -- n * up)    shifted (x + m, y) scaled u;
    endfor
    if n > 3:
        label("$(1" for i=2 upto n-1: & "+" & decimal i endfor & ")\cdot(" & decimal n & ")$",
            1/2[(0, y+n) scaled u, (x,y) scaled u]);
    fi
    x := x + n;
    y := y - (n+1);
endfor


label.top(btex $\displaystyle
    \sum_{k=1}^n k^2 = \left(\sum_{k=1}^n k \right)^2
    - 2 \sum_{k=1}^{n-1} \left( \left(k+1\right) \sum_{i=1}^k i \right)
$ etex, point 5/2 of bbox currentpicture shifted 21 up);
```

Sums of squares IV

$$\sum_{k=1}^{n} k^2 = \left(\sum_{k=1}^{n} k\right)^2 - 2\sum_{k=1}^{n-1}\left((k+1)\sum_{i=1}^{k} i\right)$$



— James O.Chilaka

```
numeric u, N, m; u = 16; N = 6; 2m = N * N + N;
path s[];
for n=1 upto N:
    s[n] = unitsquare xscaled -m shifted (0, -n) scaled u;
    fill s[n] withcolor if odd n: Oranges else: Blues fi 8 2;
    m := m - n;
endfor
numeric m; 2m = N * N + N;
for n=1 upto N:
    for k=1 upto n-1:
        draw ((k - m, 0) -- (k - m, -n)) scaled u dashed evenly scaled 1/2;
    endfor
    draw (-m * u, 0) -- subpath (1, 0) of s[n]
        if n=N: -- subpath (3, 2) of s1 fi;
    m := m - n;
endfor

label.top(btex $\displaystyle
    \sum_{i=1}^n\sum_{j=i}^n j = \sum_{i=1}^{n} i^2
$ etex, point 5/2 of bbox currentpicture shifted 21 up);
```

Sums of squares V

$$\sum_{i=1}^{n}\sum_{j=i}^{n} j = \sum_{i=1}^{n} i^2$$



— Pi-Chun Chuang

```
input paintball
picture P[];
P1 = image(draw rball);
P2 = image(for i=0 upto 1:
    for j=0 upto 1:
        draw if i+j=0: rball else: bball fi shifted (18i, 18j);
    endfor
endfor);
P3 = image(for i=0 upto 2:
    for j=0 upto 2:
        draw if i+j<3: rball else: bball fi shifted (18i, 18j);
    endfor
endfor);
P4 = image(for i=0 upto 3:
    for j=0 upto 3:
        draw if i+j<3: rball else: bball fi shifted (18i, 18j);
    endfor
endfor);
P5 = image(for i=0 upto 3:
    for j=0 upto 3:
        if i+j >= 3: draw bball shifted (18i, 18j) fi;
    endfor
endfor);
label("${}-{}$", origin);
label.rt(P1,        point 3/2 of bbox currentpicture);
label.rt("${}+{}$", point 3/2 of bbox currentpicture);
label.rt(P2,        point 3/2 of bbox currentpicture);
label.rt("${}-{}$", point 3/2 of bbox currentpicture);
label.rt(P3,        point 3/2 of bbox currentpicture);
label.rt("${}+{}$", point 3/2 of bbox currentpicture);
label.rt(P4,        point 3/2 of bbox currentpicture);
label.rt("${}={}$", point 3/2 of bbox currentpicture);
label.rt(P5,        point 3/2 of bbox currentpicture);
label.bot(btex $\displaystyle \sum_{k=1}^n \left(-1\right)^k k^2
  = \left(-1\right)^n T_n = \left(-1\right)^n \frac{n(n+1)}{2}$ etex,
  point 1/2 of bbox currentpicture shifted 13 down);
```

Alternating sums of squares

I.



$$\sum_{k=1}^{n} (-1)^k k^2 = (-1)^n T_n = (-1)^n \frac{n(n+1)}{2}$$

— Dave Logothetti

II.



$$n^2 - (n-1)^2 + \cdots + (-1)^{n-1} (1)^2 = \sum_{k=0}^{n} (-1)^k (n-k)^2 = \frac{n(n+1)}{2}$$

— Steven L. Snover

```
numeric u; u = 12;
for i=1 upto 34:
    draw ((i, 0) -- (i, -21)) scaled u dashed withdots scaled 1/4;
endfor
for i=1 upto 20:
    draw ((0, -i) -- (34, -i)) scaled u dashed withdots scaled 1/4;
endfor
numeric y; y = 0;
for i=1, 2, 5, 13:
    y := y + i * u;
    draw unitsquare scaled (u * i) shifted (0, -y) withpen pencircle scaled 3/2;
    draw unitsquare scaled y shifted (u * i, -y) withpen pencircle scaled 3/2;
endfor

label.bot(btex $F_1=F_2=1$; $F_{n+2}=F_{n+1} + F_n$
  \quad hence\quad
  $F_1^2+F_2^2+\cdots+F_n^2=F_nF_{n+1}$
  etex, point 1/2 of bbox currentpicture shifted 13 down);
```

Sums of squares of Fibonacci numbers



$$F_1 = F_2 = 1; F_{n+2} = F_{n+1} + F_n \quad \text{hence} \quad F_1^2 + F_2^2 + \cdots + F_n^2 = F_n F_{n+1}$$

— Alfred Brousseau

```
numeric u, n, x, y;
u = 9;
n = 8;
x = y = 0;
for i=1 upto n:
    y := y - (i*u);
    for j=0 upto floor (i/2) - 1:
        path s; s = unitsquare scaled (i*u) shifted (i*u*j, y);
        fill s withcolor if odd i: Blues else: Oranges fi 8 2;
    endfor
    for j=1 upto ceiling (i/2):
        path s; s = unitsquare scaled (i*u) shifted (x, -i*u*j);
        fill s withcolor if odd i: Blues else: Oranges fi 8 2;
        if 2j = i:
            fill center s -- subpath (-1/2, 1/2) of s -- cycle
                withcolor Oranges 8 3;
        fi
    endfor
    x := x + (i*u);
endfor
numeric x, y;
x = y = 0;
for i=1 upto n:
    y := y - (i*u);
    for j=0 upto floor (i/2) - 1:
        path s; s = unitsquare scaled (i*u) shifted (i*u*j, y);
        draw s;
    endfor
    for j=1 upto ceiling (i/2):
        path s; s = unitsquare scaled (i*u) shifted (x, -i*u*j);
        draw s;
    endfor
    x := x + (i*u);
endfor

numeric N; N = 1/2 n * (n + 1);
for i=1 upto N-1:
    draw (i*u, 0) -- (i*u, -1/2 u);
    draw (0, -i*u) -- (1/2u, -i*u);
endfor
draw unitsquare scaled u xscaled N yscaled -N withpen pencircle scaled 1;

label.top("$1^3+2^3+3^3+\cdots+n^3 = \left(1+2+3+\cdots+n\right)^2$",
point 5/2 of bbox currentpicture shifted 13 up);
```

Sums of cubes I

$$1^3 + 2^3 + 3^3 + \cdots + n^3 = (1 + 2 + 3 + \cdots + n)^2$$



— Solomon W.Golomb

```
numeric u, N;
u = 9;
N = 8;
path s; s = unitsquare scaled u yscaled -1;
for n = N downto 1:
    numeric t; t = n / 2 * (n + 1);
    fill s scaled t withcolor if odd n: Blues else: Oranges fi 8 2;
endfor
numeric t; t = N / 2 * (N + 1);
for i = 1 upto t - 1:
    draw (i*u, 0) -- (i*u, -t*u);
    draw (0, -i*u) -- (t*u, -i*u);
endfor
for n = N downto 1:
    numeric t; t = n / 2 * (n + 1);
    draw s scaled t withpen pencircle scaled 3/2;
    for i=n step n until t:
        draw ((i, -t) -- (i, n-t)) scaled u withpen pencircle scaled 3/2;
    endfor
    for i=n step n until t-eps:
        numeric a; a = if t - i < n: n/2 else: 0 fi;
        draw ((t-n, a-i) -- (t-a, a-i)) scaled u withpen pencircle scaled 3/2;
    endfor
endfor

label.top("$1^3+2^3+3^3+\cdots+n^3 = \left(1+2+3+\cdots+n\right)^2$",
point 5/2 of bbox currentpicture shifted 13 up);
```

Sums of cubes II

$$1^3 + 2^3 + 3^3 + \cdots + n^3 = (1 + 2 + 3 + \cdots + n)^2$$



— J. Barry Love

```
input isometric_projection
set_projection(100/3, -45);
picture P[], b_cube, o_cube, r_cube;
b_cube = cube(Blues 8 4, Blues 8 2, Blues 8 1);
o_cube = cube(Oranges 8 4, Oranges 8 2, Oranges 8 1);
r_cube = cube(Reds 8 4, Reds 8 2, Reds 8 1);

P1 = image(
for n=1, 2, 3:
  for i=1 upto n:
    for j=1 upto n:
      for k=1 upto n:
        draw o_cube shifted p(i - 3/4 n * n, n * n / 2 + j, n - k);
      endfor
    endfor
  endfor
endfor);
P2 = image(
for n=1, 2, 3:
  for i=1 upto n:
    for j=1 upto n:
      for k=1 upto n:
        draw if k = 1: o_cube elseif k = 2: b_cube else: r_cube fi
          shifted p(i - 3/4 n * n, n * n / 2 + j, n - 1.4k);
      endfor
    endfor
  endfor
endfor);
P3 = image(
  draw o_cube;
  for i=1 upto 3:
    for j=1 upto 3:
      if (i*j) > 1:
        draw if (i>1) and (j>1): o_cube else: b_cube fi shifted p(i, j, .5);
      fi
    endfor
  endfor
  for i=1 upto 6:
    for j=1 upto 6:
      if (i>3) or (j>3):
        draw if i<=3: b_cube elseif j<=3: r_cube else: o_cube fi shifted p(i, j, 2);
      fi
    endfor
  endfor);
P4 = image(
  for i=1 upto 6:
    for j=1 upto 6:
      draw o_cube shifted p(i, j, 1);
    endfor
  endfor);

draw P1; draw P2 shifted (200, 0);
draw P3 shifted (-100, -160); draw P4 shifted (100, -160);

label.top("$1^3+2^3+3^3+\cdots+n^3 = \left(1+2+3+\cdots+n\right)^2$",
  point 5/2 of bbox currentpicture shifted 21 up);
```

Sums of cubes III

$$1^3 + 2^3 + 3^3 + \cdots + n^3 = (1 + 2 + 3 + \cdots + n)^2$$



— Alan L. Fry

```
color s[]; s1 = Reds 8 2; s2 = Oranges 8 2; s3 = Blues 8 2;
numeric u; 36 u = tw;
for i = 1 upto 5:
    numeric o; o = 1/2 i * (i + 1);
    path b; b = unitsquare scaled i shifted -(o, o) scaled u;
    for r=0 upto 3:
        for j = 0 upto i-1:
            if known s[r]:
                fill b shifted (j*i*u, 0) rotated 90r withcolor s[r];
            else:
                drawoptions(dashed evenly scaled 1/2 withpen pencircle scaled 1/4);
                for k=1 upto i-1:
                    draw subpath (3,4) of b shifted (j*i*u+k*u, 0);
                    draw subpath (0,1) of b shifted (j*i*u, k*u);
                endfor
                drawoptions();
            fi
            draw b shifted (j*i*u, 0) rotated 90r;
        endfor
    endfor
endfor
input arrow_label

arrow_label((-15u, -15u), (10u, -15u), "$n^2$", 9);
arrow_label((10u, -15u), (15u, -15u), "$n$", 9);

label.top("$1^3+2^3+3^3+\cdots+n^3 = \frac14 \left(n(n+1)\right)^2$",
point 5/2 of bbox currentpicture shifted 21 up);
```

Sums of cubes IV

$$1^3 + 2^3 + 3^3 + \cdots + n^3 = \frac{1}{4}\left(n(n+1)\right)^2$$



— Antonella Cupillari

```
input isometric_projection
set_projection(80/3, -30); ipscale := 16;
picture P[], Cube[]; numeric N; N = 4;
Cube0 = cube(background, background, background);
Cube1 = cube(Reds 8 4, Reds 8 2, Reds 8 1);
Cube2 = cube(Oranges 8 4, Oranges 8 2, Oranges 8 1);
Cube3 = cube(Purples 8 4, Purples 8 2, Purples 8 1);
Cube4 = cube(Blues 8 4, Blues 8 2, Blues 8 1);
P1 = image(for n = N downto 1:
        numeric a; a = 1/2 n * n - 1/2;
        for m = 1 upto n-1:
            numeric b; b = 1/2 m * m - 1/2;
            for z = 1 upto n:
                for x = 0 upto m-1:
                    draw Cube[if n < N: 0 else: 4 fi] shifted p(b + x, 0, a + n - z);
                endfor
            endfor
        endfor
        for x = 0 upto n-1:
            for z = 1 upto n:
                draw Cube[if n < N: 0 else: 4 fi] shifted p(a + x, 0, a + n - z);
            endfor
        endfor
    endfor
    for n = 1 upto N:
        numeric a; a = 1/2 n * n - 1/2;
        for m = n-1 downto 1:
            numeric b; b = 1/2 m * m - 1/2;
            for x = 0 upto n-1:
                for z = 1 upto m:
                    draw Cube[if n < N: 0 else: 4 fi] shifted p(a + x, 0, b + m - z);
                endfor
            endfor
        endfor
    endfor);
% ... and so on to assemble all the other pictures

P7 = image(
  draw P1; draw P2 shifted p(12.5, 0, 8);
  draw P3 shifted p(13.5, 0, 0); draw P4 shifted p(22, 0, 1);
  drawarrow p(11, 0, 9.5) -- p(12, 0, 9.5);
  drawarrow p(11, 0, 2.5) -- p(12, 0, 2.5);
  drawarrow p(18, 0, 6.5) -- p(19, 0, 6.5);
  label.bot("$t_n = 1 + 2 + \cdots + n$ \quad\Rightarrow\quad $t_n^2 - t_{n-1}^2 = n^3$",
    point 1/2 of bbox currentpicture));

P8 = image(
  draw P5; draw P6; drawarrow p(11, 0, 2.5) -- p(12, 0, 2.5);
  label.bot("$t_n^2 = \left(1 + 2 + \cdots + n\right)^2 = 1^3 + 2^3 + \cdots + n^3$",
    point 1/2 of bbox currentpicture));

draw P7; draw P8 shifted 240 down;
```

Sums of cubes V



$$t_n = 1 + 2 + \cdots + n \quad \Rightarrow \quad t_n^2 - t_{n-1}^2 = n^3$$



$$t_n^2 = (1 + 2 + \cdots + n)^2 = 1^3 + 2^3 + \cdots + n^3$$

— Roger Nelsen

```
vardef cartouche(expr w, d, r) = save p; path p; p =
  quartercircle rotated 180 shifted ( 1/2,  1/2) scaled r shifted (0-d, -d) --
  quartercircle rotated 270 shifted (-1/2,  1/2) scaled r shifted (w+d, -d) --
  quartercircle rotated   0 shifted (-1/2, -1/2) scaled r shifted (w+d, +d) --
  quartercircle rotated  90 shifted ( 1/2, -1/2) scaled r shifted (0-d, +d) --
  cycle; image(fill p withcolor Oranges 8 2; draw p;)
enddef;
vardef boomer(expr n, w, h, d, r) = save p; path p; p =
  quartercircle rotated 180 shifted ( 1/2,  1/2) scaled r shifted (  0-d, -h*n-d) --
  quartercircle rotated 270 shifted (-1/2,  1/2) scaled r shifted (w*n+d, -h*n-d) --
  quartercircle rotated   0 shifted (-1/2, -1/2) scaled r shifted (w*n+d,    0+d) --
  if n > 0:
  quartercircle rotated  90 shifted (+1/2, -1/2) scaled r shifted (w*n-d,    0+d) --
  reverse
  quartercircle rotated 270 shifted (-1/2,  1/2) scaled r shifted (w*n-d, -h*n+d) --
  fi
  quartercircle rotated  90 shifted (+1/2, -1/2) scaled r shifted (  0-d, -h*n+d) --
  cycle; image(fill p withcolor Blues 8 2; draw p)
enddef;

picture P[]; P0 = image(
  label.lrt(btex \vbox{\openup 16pt\halign{\hss ${}#{}$&&\hbox to 36pt{\hss$#$\hss}\cr
  & 1 & 2 & 3 & \cdots & n\cr
+ & 2 & 4 & 6 & \cdots & 2n\cr
+ &  3 & 6 & 9 & \cdots & 3n\cr
+ & \vdots & \vdots & \vdots & \ddots & \vdots \cr
+ & n & 2n & 3n& \cdots & n^2\cr
}} etex, origin));

P1 = image(
  picture c; c = cartouche(144, 8, 4);
  for i=0, 1, 2, 4: draw c shifted (34, -29.4i - 6); endfor
  draw P0;
  label.lrt("${}=\quad \sum_{i=1}^n i + 2\sum_{i=1}^n i + \cdots + n\sum_{i=1}^n i$",
    point 0 of bbox currentpicture shifted 16 down);
  label.lrt("${}=\quad \left(\sum_{i=1}^n i\right)^2$",
    point 0 of bbox currentpicture shifted 10 down);
);
P2 = image(
  numeric u, v; u = 36; v = 29.4;
  for i = 0, 1, 2, 4: draw boomer(i, u, v, 8, 4) shifted (33, -6); endfor
  draw P0;
  label.lrt("${}=\quad 1 + 2 \cdot 2^2 + 3 \cdot 3^2 + \cdots + n \cdot n^2$",
    point 0 of bbox currentpicture shifted 16 down);
  label.lrt("${}=\quad \sum_{i=1}^n i^3$",
    point 0 of bbox currentpicture shifted 16 down);
);

draw P1 shifted 112 left; draw P2 shifted 112 right;
```

Sums of cubes VI

$$
\begin{array}{cccccc}
 & 1 & 2 & 3 & \cdots & n \\
+ & 2 & 4 & 6 & \cdots & 2n \\
+ & 3 & 6 & 9 & \cdots & 3n \\
+ & \vdots & \vdots & \vdots & \ddots & \vdots \\
+ & n & 2n & 3n & \cdots & n^2 \\
= & \multicolumn{5}{l}{\sum_{i=1}^{n} i + 2\sum_{i=1}^{n} i + \cdots + n\sum_{i=1}^{n} i} \\
= & \multicolumn{5}{l}{\left(\sum_{i=1}^{n} i\right)^2}
\end{array}
\qquad
\begin{array}{cccccc}
 & 1 & 2 & 3 & \cdots & n \\
+ & 2 & 4 & 6 & \cdots & 2n \\
+ & 3 & 6 & 9 & \cdots & 3n \\
+ & \vdots & \vdots & \vdots & \ddots & \vdots \\
+ & n & 2n & 3n & \cdots & n^2 \\
= & \multicolumn{5}{l}{1 + 2 \cdot 2^2 + 3 \cdot 3^2 + \cdots + n \cdot n^2} \\
= & \multicolumn{5}{l}{\sum_{i=1}^{n} i^3}
\end{array}
$$

— Farhood Pouryoussefi

```
numeric N, u, y;
N = 5; y = 0; u = 12;
for n = 1 upto N:
    numeric w; w = n * u;
    path s, t;
    s = unitsquare scaled w;
    y := y - w;
    label("$" &
      if n = N: "n"
      elseif n = N - 1: "\vdots"
      else: decimal n
      fi & "$", (-8, y + 1/2 w));
    label("$" &
      if n = N: "n \times n^2"
      elseif n = N - 1: "\vdots"
      else: decimal n & "\times" & decimal n & "^2"
      fi & "$", (N * N * u + 4 u, y + 1/2 w));
    for x = 0 upto n - 1:
      draw s shifted (x * w, y);
    endfor
    t = subpath (3/2, 3) of s shifted ((n - 1) * w, y) -- cycle;
    fill t withcolor Oranges 8 2;
    fill t rotatedabout(point 0 of t, 180) withcolor Blues 8 2;
endfor
draw origin -- (0, y) -- (N * (N + 1) * u, y) -- cycle withpen pencircle scaled 3/4;
label.bot("$n(n+1)$", (1/2 N * (N + 1) * u, y));
label.top(btex
    \vbox{\openup 12pt\halign{\hss $#$&${}=#$ \hss \cr
        1 + 2 + \cdots + n & \frac12 n(n+1)\cr
    1^3 + 2^3 + \cdots + n^3 & \left(\frac12 n (n+1)\right)^2\cr
    }} etex, point 5/2 of bbox currentpicture shifted 42 up);
```

Sums of integers and sums of cubes

$$1 + 2 + \cdots + n = \tfrac{1}{2}n(n+1)$$

$$1^3 + 2^3 + \cdots + n^3 = \left(\tfrac{1}{2}n(n+1)\right)^2$$



— Georg Schrage

```
numeric N, u, y;
N = 4; y = 0; u = 6;

% see document source for definitions of sq, sqmark, and sqclip

picture P[];
P1 = image(
  for n = 1 upto N:
    string s, S;
    s = if n = N: "\left(2n-1\right)" else: decimal (2n - 1) fi;
    S = "$" & s & "^3 = " & s & "\times" & s & "^2 = {}$";
    numeric y; y = (n * n + n + n) * -u;
    label.lft(S, (80, y));
    for i = 1 upto 2n - 1:
      draw if (i=1) or (i=2n-1): sqmark(2n - 1) else: sq(2n-1) fi
        shifted (80, y + 1/2 u)
        shifted (if i > n: (n-1, i - n - 1/2) else: (i-1, -1/2) fi * (2n+1) * u);
    endfor
  endfor);

input arrow_label
P2 = image(
  numeric x, y; x = y = 0;
  for n = 1 upto N:
    picture C, Up, Left; C = sq(2n - 1); Left = sqclip(2n - 1);
    Up = Left rotated -90 reflectedabout(up, down) shifted ((2n-1, 3n-2) * u);
    draw C shifted ((x, y) * u);
    for i = 2 upto n:
      draw if i < n: C else: Up fi
          shifted ((x, y + ((i-1) * (2n - 7/8))) * u);
      draw if i < n: C else: Left shifted ((1-n)*u,0) fi
          shifted ((x - ((i-1) * (2n - 7/8)), y) * u);
    endfor
    x := x + 2n - 1 + 1/8;
    y := y - 2n - 1 - 1/8;
  endfor
  arrow_label(lrcorner currentpicture, urcorner currentpicture,
            TEX("$n(2n-1) + (n-1)$") rotated 90, 12));

label.top(P1, 10 up);
label.bot(P2, 10 down);

label.bot(btex $
    1^3 + 3^3 + 5^3 + \cdots + \left( 2n   - 1 \right)^3
    = 1 +   2 +   3 + \cdots + \left( 2n^2 - 1 \right)
    = n^2 \left( 2n^2 - 1 \right)
$ etex, point 1/2 of bbox currentpicture shifted 34 down);
```

Sums of odd cubes are triangular numbers



$$1^3 + 3^3 + 5^3 + \cdots + (2n-1)^3 = 1 + 2 + 3 + \cdots + \left(2n^2 - 1\right) = n^2\left(2n^2 - 1\right)$$

— Monte J. Zerger

```
numeric u, o; u = 12; o = 0;
for n = 1 upto 4:
    path s; s = unitsquare rotated -90 scaled n scaled n shifted (o, -o) scaled u;
    fill s withcolor Greens 8 2;
    if n=1:
        label("$\scriptstyle 1^4$", center s);
    else:
        path t, tt;
        t = subpath (4, 3) of s -- subpath (3, 4) of s shifted (0, o*u) -- cycle;
        tt = t reflectedabout(origin, dir -45);
        fill t withcolor Purples 8 2;
        fill tt withcolor Purples 8 2;
        draw subpath (-1, 1) of t;
        draw subpath (-1, 1) of tt;
        label("$" if n=2: & "\scriptstyle" fi & decimal n & "^2\left(1^2"
            for i=2 upto n-1:
                & "+" & decimal i & "^2"
            endfor & "\right)$", center t);
            fill (superellipse(right, up, left, down, 0.78)) scaled u shifted center s
                withcolor Greens 9 1;
        label("$" & decimal n & "^4$", center s);
    fi
    o := o + n * n;
endfor
path s; s = unitsquare xscaled o yscaled -o scaled u; draw s;
for i=1 upto o-1:
    draw (origin -- 2 up) shifted ((i, -o)*u);
    if i > 5: draw (origin -- 2 down) shifted ((i, 0)*u); fi
    draw (origin -- 2 left) shifted ((o, -i)*u);
    draw (origin -- 2 right) shifted ((0, -i)*u);
endfor

input arrow_label
arrow_label(point 4 of s, point 3 of s, "$\sum_{i=1}^n i^2$", 24);
arrow_label(point 3 of s, point 2 of s, "$\sum_{i=1}^n i^2$", 16);

label.top(btex $
\sum_{i=1}^n i^4
= \left(\sum_{i=1}^n i^2 \right)^2
- 2 \left(\sum_{k=2}^n \left(k^2 \sum_{i=1}^{k-1} i^2 \right)\right)$
etex, point 5/2 of bbox currentpicture shifted 34 up);
```

Sums of fourth powers

$$\sum_{i=1}^{n} i^4 = \left(\sum_{i=1}^{n} i^2\right)^2 - 2\left(\sum_{k=2}^{n}\left(k^2 \sum_{i=1}^{k-1} i^2\right)\right)$$



— Elizabeth M. Markham

```
numeric u, N;
u = 16; N = 5;

for n = N downto 1:
    path s; s = unitsquare scaled u xscaled (N * N - N + n) yscaled n;
    fill s withcolor if odd n: Blues else: Oranges fi 8 2;
endfor

for i=1 upto N * N - N - 1:
    draw (origin -- up * N * u) shifted (i*u, 0) dashed evenly scaled 1/2;
endfor

z0 = ((N * N - N) * u, 0);
draw (origin -- up * N * u) shifted z0;

for i = 1 upto N - 1:
    draw (x0 + i * u, N * u) -- (x0 + i * u, y0 + i * u) -- (N * N * u, y0 + i * u)
        dashed evenly scaled 1/2;
endfor

for n = N downto 1:
    path s; s = unitsquare scaled u xscaled (N * N - N + n) yscaled n;
    draw s;
endfor
input arrow_label

arrow_label(origin, z0, "$n^{k-1}-n$", 12);
arrow_label(z0, (N*N*u, 0), "$n$", 12);
arrow_label(origin, (0, N*u), "$n$", -12);

label.top(btex $
n^k = \left(n^{k-1} - n + 1\right)
    + \left(n^{k-1} - n + 3\right) + \cdots
    + \left(n^{k-1} - n + 2n - 1\right)$ for $k=2, 3, \dots$
etex, point 5/2 of bbox currentpicture shifted 34 up);
```

k-th powers as sums of consecutive odd numbers

$$n^k = \left(n^{k-1} - n + 1\right) + \left(n^{k-1} - n + 3\right) + \cdots + \left(n^{k-1} - n + 2n - 1\right) \text{ for } k = 2, 3, \ldots$$



— N. Gopalakrishnan Nair

```
numeric t, u, N; t = 0; u = 16; N = 6;
for n = 1 upto N:
    path W, M, B;
    W = (origin -- (n, 0) -- (n, n) for i = 1 upto n:
          -- (n-i, n-i+1) -- (n-i, n-i)
        endfor -- cycle) shifted (t, 0) scaled u;
    M = W rotatedabout(point n + 3/2 of W, 180);
    fill M withcolor Reds 8 2; draw M;
    fill W withcolor Greens 8 2; draw W;
    t := t + n;
    B = unitsquare xscaled -t scaled u shifted point 0 of M;
    fill B withcolor Blues 8 1; draw B;
    if n = 1:
        label("$1$", center B);
        label("$T_1$", center W);
        label("$T_1$", center M);
        label.bot("$1$", point 1/2 of W);
    elseif n < N - 1:
        label("$1" for i=2 upto n: & "+" & decimal i endfor & "$", center B);
        label("$T_" & decimal n & "$", 1/2[point 1 of W, point n + 3/2 of W]);
        label("$T_" & decimal n & "$", 1/2[point 1 of M, point n + 3/2 of W]);
        label.bot("$" & decimal n & "$", point 1/2 of W);
    elseif n = N:
        label("$T_n = 1 + 2 + \cdots + n$", center B);
        label("$T_n$", 1/2[point 1 of W, point n + 3/2 of W]);
        label("$T_n$", 1/2[point 1 of M, point n + 3/2 of W]);
        label.bot("$n$", point 1/2 of W);
        label.rt("$n+2$", point 3/2 of W shifted (0, u));
    fi

endfor
label.top(btex $T_n = 1 + 2 + \cdots + n$ \quad implies\quad
          $\displaystyle T_1 + T_2 + \cdots + T_n = \frac{n(n+1)(n+2)}6$
          etex, point 5/2 of bbox currentpicture shifted 34 up);
label.bot(btex $3\left(T_1+T_2+\cdots+T_n\right) = (n+2) \cdot T_n $
          etex, point 1/2 of bbox currentpicture shifted 34 down);
label.bot(btex $\displaystyle T_1+T_2+\cdots+T_n =
          \frac{(n+2)}{3} \cdot \frac{n(n+1)}{2} = \frac{n(n+1)(n+2)}6$
          etex, point 1/2 of bbox currentpicture shifted 21 down);
```

Sums of triangular numbers I

$$T_n = 1 + 2 + \cdots + n \quad \text{implies} \quad T_1 + T_2 + \cdots + T_n = \frac{n(n+1)(n+2)}{6}$$



$$3\,(T_1 + T_2 + \cdots + T_n) = (n+2) \cdot T_n$$

$$T_1 + T_2 + \cdots + T_n = \frac{(n+2)}{3} \cdot \frac{n(n+1)}{2} = \frac{n(n+1)(n+2)}{6}$$

— Monte J. Zerger

```
numeric t, N; t = 0; N = 5;
input arrow_label
input isometric_projection
ipscale := 14; set_projection(25, -30);
picture P[], r_cube;
r_cube = cube(Reds 8 4, Reds 8 2, background);
P7 = image(
    for n=1 upto N:
        for i = n downto 1:
            for j = 1 upto i:
                draw r_cube shifted p(2n, j, i);
            endfor
        endfor
    endfor
);
P1 = image(
for n=1 upto N:
    for i = n downto 1:
        for j = 1 upto i:
            draw r_cube shifted p(n, j, i);
        endfor
    endfor
endfor
);
P8 = image(
    draw P1;
    arrow_label(p(N + 1/2, 1, N + 1), p(N + 1/2, N + 1, N + 1), "$n$", 0);
);

path a; a = (left--right) scaled 18;
drawarrow a;
label.lft(P7, point 0 of a shifted 16 left);
label.rt(P8, point 1 of a);
% more juggling of the drawing order for other pictures...

label.bot(P9, point 1/2 of bbox currentpicture shifted 21 down);
label.bot(P10, point 1/2 of bbox currentpicture shifted 21 down);

label.top(btex $T_n = 1 + 2 + \cdots + n$ \quad implies\quad
  $\displaystyle T_1 + T_2 + \cdots + T_n = \frac{n(n+1)(n+2)}6$
  etex, point 5/2 of bbox currentpicture shifted 34 up);
```

Sums of triangular numbers II

$$T_n = 1 + 2 + \cdots + n \quad \text{implies} \quad T_1 + T_2 + \cdots + T_n = \frac{n(n+1)(n+2)}{6}$$







— Roger B. Nelsen

```
path t; t = for i=1 upto 3: 85up rotated 120i -- endfor cycle;
string s[];
s0 = "$\scriptstyle n$";
s1 = "$\scriptstyle n-1$";
s2 = "$\cdot$";
s3 = "$\cdot$";
s4 = "3";
s5 = "2";
s6 = "1";
s7 = "$\scriptstyle n+2$";
numeric N; N = 6;
for p=0 upto 2:
    t := t rotated 120;
    picture P;
    P = image(
        label(s0, point 0 of t);
        for n = 1 upto N:
            for i = 0 upto n:
                label(s[n], (i/n)[point -n/N of t, point n/N of t]);
            endfor
        endfor
    );
    draw P shifted (180p, 0);
endfor
label("$+$", (90, 30));
label("$+$", (270, 30));
label("$=$", (90, -150));
picture P;
P = image(
    label(s7, point 0 of t);
    for n = 1 upto N:
        for i = 0 upto n:
            label(s7, (i/n)[point -n/N of t, point n/N of t]);
        endfor
    endfor
);
draw P shifted (180, -180);

label.top(btex $T_n = 1 + 2 + \cdots + n$ \quad implies\quad
  $T_1 + T_2 + \cdots + T_n = \frac16 n(n+1)(n+2)$
  etex, point 5/2 of bbox currentpicture shifted 34 up);

label.bot(btex $3\left(T_1 + T_2 + \cdots + T_n\right) = T_n \cdot (n+2)$
  etex, point 1/2 of bbox currentpicture shifted 34 down);
```

Sums of triangular numbers III

$$T_n = 1 + 2 + \cdots + n \quad \text{implies} \quad T_1 + T_2 + \cdots + T_n = \tfrac{1}{6}n(n+1)(n+2)$$

$$\begin{array}{c} 1 \\ 1 \quad 2 \\ 1 \quad 2 \quad 3 \\ 1 \quad 2 \quad 3 \quad \cdot \\ 1 \quad 2 \quad 3 \quad \cdot \quad \cdot \\ 1 \quad 2 \quad 3 \quad \cdot \quad \cdot \quad n{-}1 \\ 1 \quad 2 \quad 3 \quad \cdot \quad \cdot \quad n{-}1 \quad n \end{array} + \begin{array}{c} n \\ n{-}1 \quad n{-}1 \\ \cdot \quad \cdot \quad \cdot \\ \cdot \quad \cdot \quad \cdot \quad \cdot \\ 3 \quad 3 \quad 3 \quad 3 \quad 3 \\ 2 \quad 2 \quad 2 \quad 2 \quad 2 \quad 2 \\ 1 \quad 1 \quad 1 \quad 1 \quad 1 \quad 1 \quad 1 \end{array} + \begin{array}{c} 1 \\ 2 \quad 1 \\ 3 \quad 2 \quad 1 \\ \cdot \quad 3 \quad 2 \quad 1 \\ \cdot \quad \cdot \quad 3 \quad 2 \quad 1 \\ n{-}1 \quad \cdot \quad \cdot \quad 3 \quad 2 \quad 1 \\ n \quad n{-}1 \quad \cdot \quad \cdot \quad 3 \quad 2 \quad 1 \end{array}$$

$$= \begin{array}{c} n{+}2 \\ n{+}2 \quad n{+}2 \\ n{+}2 \quad n{+}2 \quad n{+}2 \\ n{+}2 \quad n{+}2 \quad n{+}2 \quad n{+}2 \\ n{+}2 \quad n{+}2 \quad n{+}2 \quad n{+}2 \quad n{+}2 \\ n{+}2 \quad n{+}2 \quad n{+}2 \quad n{+}2 \quad n{+}2 \quad n{+}2 \\ n{+}2 \quad n{+}2 \quad n{+}2 \quad n{+}2 \quad n{+}2 \quad n{+}2 \quad n{+}2 \end{array}$$

$$3\,(T_1 + T_2 + \cdots + T_n) = T_n \cdot (n+2)$$

```
input isometric_projection
ipscale := 24; set_projection(25, -30);
picture cb[], P[];
cb1 = cuboid(-1, 2, 1, Blues 8 4, Blues 8 2, background);
cb2 = cuboid(-1, 4, 3, Blues 8 4, Blues 8 2, background);
cb3 = cuboid(-2, 4, 3, Blues 8 4, Blues 8 2, background);
P1 = image(
    draw cb1;
    for i = -1 upto 1:
        draw cb1 shifted p(i, 0, 0) shifted 144 right;
    endfor
    drawoptions(withcolor Blues 8 7);
    label.bot("$\scriptstyle 1$", p(-1/2, 0, 0));
    label.lrt("$\scriptstyle 1$", p(0, 0, 1/2));
    label.rt("$\scriptstyle 2$", p(0, 1, 1));
    for i = -1 upto 1:
        label.bot("$\scriptstyle 1$", p(i-1/2, 0, 0) shifted 144 right);
    endfor
    label.lrt("$\scriptstyle 1$", p(1, 0, 1/2) shifted 144 right);
    label.rt ("$\scriptstyle 2$", p(1, 1,   1) shifted 144 right);
    drawoptions();
    label("$= \qquad \frac13$", (64, 21));
);
P2 = image(
    draw cb3 shifted 20 left;

    draw cb2 shifted 80 right;
    draw cb3 shifted p(-1, 0, 0) shifted 240 right;
    for i = 0 upto 2:
        draw cb2 shifted p(i, 0, 0) shifted 240 right;
    endfor
    label("$= \quad \frac13$", (150, 64));
    label("$\frac13$", (-78, 64));
    label("$+$", (38, 64));
    drawoptions(withcolor Blues 8 7);
    label.llft("$\scriptstyle n-1$", p(-3/4,0,0)) shifted  20 left;
    label.lrt ("$\scriptstyle n$",   p(0,0,3/2))  shifted  20 left;
    label.rt  ("$\scriptstyle n+1$", p(0,5/2,3))  shifted  20 left;

    label.bot ("$\scriptstyle 1$",   p(-1/2,0,0)) shifted  80 right;
    label.lrt ("$\scriptstyle n$",   p(0,0,3/2))  shifted  80 right;
    label.rt  ("$\scriptstyle n+1$", p(0,5/2,3))  shifted  80 right;

    label.llft("$\scriptstyle n-1$", p(-7/4,0,0)) shifted 240 right;
    label.lrt ("$\scriptstyle n$",   p(2,0,3/2))  shifted 240 right;
    label.rt  ("$\scriptstyle n+1$", p(2,2,3))    shifted 240 right;
    label.urt ("$\scriptstyle n+2$", p(-1/2,4,3)) shifted 240 right;
    label.bot ("$\scriptstyle 1$",   p(-1/2,0,0)) shifted 240 right;
    label.bot ("$\scriptstyle 1$",   p(1/2,0,0))  shifted 240 right;
    label.bot ("$\scriptstyle 1$",   p(3/2,0,0))  shifted 240 right;
    drawoptions();
);
draw P1; draw P2 shifted 240 down;
label.top("$(1\times2)+(2\times3)+(3\times4) + \cdots + (n-1)n = \frac13 (n-1) n (n+1)$",
    point 5/2 of bbox currentpicture shifted 34 up);
```

Sums of oblong numbers I

$$(1 \times 2) + (2 \times 3) + (3 \times 4) + \cdots + (n-1)n = \frac{1}{3}(n-1)n(n+1)$$





— T. C. Wu

```
input isometric_projection
ipscale := 24; set_projection(28, -28);
numeric i; i = 0;
picture c[];
forsuffixes s = Reds, Greens, Oranges, Blues:
    c[incr i] = cube(s 8 4, s 8 3, s 8 2);
endfor

picture P[];
def make_boxes(expr n, X, Y, Z, Sx, Sy, Sz) =
    P[n] = image(
        for x = 1 upto X:
            for y = 0 upto Y-1:
                for z = Z-1 downto 0:
                    draw c[n] shifted p(x, y, z);
                endfor
            endfor
        endfor
    );
    P[11n] = image(draw P[n];
    label.lft(Sy, p(0, 1/2Y, 0));
    label.bot(Sx, p(1/2X, 0, 0));
    label.lrt(Sz, p(X, 0, 1/2Z));
) enddef;

make_boxes(1, 2, 1, 3, "2", "1", "3");
make_boxes(2, 2, 3, 3, "2", "3", "3");
make_boxes(3, 3, 4, 3, "3", "4", "3");
make_boxes(4, 5, 4, 3, "$n+1$\strut", "$n$", "3");

P5 = image(
    draw P4 shifted p(0, 0, 3);
    draw P2; draw P1 shifted p(0, 3, 0);
    draw P3 shifted p(2,0,0);
);

numeric s; s = 80;
label(P11, (-2s,5/2s)); label("$+$", (-s,5/2s));
label(P22,   (0,5/2s)); label("$+$", (+s,5/2s));
label(P33, (+2s,5/2s));

label("$+$", (-2.8s, 0));
label(P44,   (-3/2s, 0));
label("$=$",      (0, 0));
label(P5,     (3/2s, 0));

label.top(btex $3 \bigl(1\times2 + 2\times3 + 3\times4  + \cdots + n(n+1) \bigr)
    = n (n+1) (n+2)$ etex, point 5/2 of bbox currentpicture shifted 34 up);
```

Sums of oblong numbers II

$$3(1 \times 2 + 2 \times 3 + 3 \times 4 + \cdots + n(n+1)) = n(n+1)(n+2)$$



— Sidney H. Kung

```
input isometric_projection
ipscale := 16; set_projection(24, -28);
picture P[], c[]; numeric i; i = 0;
forsuffixes s = Reds, Greens, Oranges, Blues:
    c[incr i] = cube(s 8 4, s 8 3, s 8 2);
endfor
P1 = image(for i=1 upto 4:
    numeric z; z = 3i * sqrt(i);
    for j = i downto 1:
        for k = 1 upto i:
            draw c[i] shifted p(-j, k, z);
        endfor
    endfor
    for k=1 upto i-1:
        for j=i-k downto 1:
            draw c[i] shifted p(-j - 1/2 k, i + k, z);
        endfor
    endfor
endfor);
P2 = image(for i=1 upto 4:
    numeric z; z = 3i * sqrt(i);
    for j = i downto 1:
        for k = 1 upto i:
            draw c[i] shifted p(-j, k, z);
        endfor
    endfor
    for k=1 upto i-1:
        for j=i-k downto 1:
            draw c[i] shifted p(-j, i, z -k);
        endfor
    endfor
endfor);
P3 = image(path base, lid;
    base = p(-1,1,1) -- p(-6, 1, 1) -- p(-6, 1, 5) -- p(-1, 1, 5) -- cycle;
    lid = base shifted p(0,4,0);
    drawoptions(dashed withdots scaled 1/2);
    draw base; for i=0 upto 3: draw point i of base -- point i of lid; endfor
    drawoptions();
    for z = 4 downto 1:
        for x = z downto 1:
            for y = 1 upto 4:
                draw c[max(y,z)] shifted p(-x, y, z);
            endfor
        endfor
    endfor
    draw lid dashed withdots scaled 1/2;
    label.lrt("$n$", point -1/2 of base);
    label.rt ("$n$", 1/2[point -1 of base, point -1 of lid]);
    label.urt("$n+1$", point 5/2 of lid));

draw P1; draw P2 shifted (120, -50); draw P3 shifted (380, -50);

label.top(btex $\displaystyle
  \frac{1\cdot2}2 + \frac{2\cdot5}2 + \frac{3\cdot8}2 + \cdots + \frac{n(3n-1)}2
  = \frac{n^2(n+1)}2$ etex, point 5/2 of bbox currentpicture shifted 34 up);
```

Sums of pentagonal numbers

$$\frac{1\cdot 2}{2}+\frac{2\cdot 5}{2}+\frac{3\cdot 8}{2}+\cdots+\frac{n(3n-1)}{2}=\frac{n^2(n+1)}{2}$$



— William A. Miller

```
numeric u, n; u = 21; n = 4;
picture P[];
P1 = image(
    fill unitsquare shifted -(1/2, 1/2) scaled (2n+1) scaled u withcolor Oranges 8 3;
    unfill unitsquare shifted -(1/2, 1/2) scaled u;
    for k=0 upto 3:
        for i=0 upto n-1:
            for j=i upto n-1:
                fill unitsquare shifted (j + 1/2, i + 1/2)
                                scaled u rotated 90k withcolor Blues 8 3;
            endfor
        endfor
    endfor
    for i=-n upto n+1:
        draw (left--right) scaled (n+1/2) shifted (0, i-1/2) scaled u;
        draw (down--up) scaled (n+1/2) shifted (i-1/2, 0) scaled u;
    endfor
    label.bot("$\left(2n+1\right)^2 = 8T_n + 1$",
      point 1/2 of bbox currentpicture shifted 8 down);
);
P2 = image(
    for k=0 upto 3:
        for i=1 upto n-1:
            for j=i upto n-1:
                fill unitsquare shifted (j, i - 1)
                                scaled u rotated 90k withcolor Blues 8 3;
                fill unitsquare shifted (i - 1, j)
                                scaled u rotated 90k withcolor Oranges 8 3;
            endfor
        endfor
    endfor
    for i=-n upto n:
        draw (left--right) scaled n shifted (0, i) scaled u;
        draw (down--up) scaled n shifted (i, 0) scaled u;
    endfor

    label.bot("$\left(2n\right)^2 = 8T_{n-1} + 4n$",
      point 1/2 of bbox currentpicture shifted 8 down);
);
interim labeloffset := 13;
label.top(P1, origin); label.bot(P2, origin);
label.top("$T_n = 1 + 2 + \cdots + n$ \quad \Rightarrow",
  point 5/2 of bbox currentpicture shifted 21 up);
```

On squares of positive integers

$$T_n = 1 + 2 + \cdots + n \quad \Rightarrow$$



$$(2n+1)^2 = 8T_n + 1$$



$$(2n)^2 = 8T_{n-1} + 4n$$

— Edwin G. Landauer

```
input isometric_projection
ipscale := 20;
set_projection(24, -42);
picture P[];
picture c;
c = cube(Blues 8 4, Blues 8 3, Blues 8 1);
for n=1 upto 3:
    P[n] = image(
    for x=0 upto 4:
        for z=3 downto 0:
            for y=0 upto 4:
                if (y=4) and (z < x): else:
                    draw c shifted p(x if n=2: + 3/4 x fi, y, z if n=3: + 3/4 z fi);
                fi
            endfor
        endfor
    endfor
    if n=1:
        label.lrt("$n$", p(4,0,2));
        label.llft("$n+1$", p(2,0,0));
        label.lft("$n+1$", p(-1,5/2,0));
    elseif n=2:
        label.llft("$n^2$",    p(4 + 12/4 - 1/2, 0,0));
        label.llft("$n^2+1$",  p(3 +  9/4 - 1/2, 0,0));
        label.llft("$\cdots$", p(2 +  6/4 - 1/2, 0,0));
        label.llft("$n^2+n-1$",p(1 +  3/4 - 1/2, 0,0));
        label.lft("$n^2+n$",  p(-1,0,0));
    elseif n=3:
        label.lrt("$n^2+n+1$", p(4, 0, 1/2));
        label.lrt("$n^2+n+2$", p(4, 0, 1/2 + 7/4));
        label.lrt("$\cdots$",  p(4, 0, 1/2 + 14/4));
        label.lrt("$n^2+2n$",  p(4, 0, 1/2 + 21/4));
    fi
    );
endfor
draw P1 shifted 144 up;
draw P2 shifted 144 left;
draw P3 shifted 120 right;
label.bot(btex\vbox{\openup 6pt \halign{\hfill $#$&${}#{}$&$#$\hfill\cr
1+2&=&3\cr
4+5+6&=&7+8\cr
9+10+11+12&=&13+14+15\cr
16+17+18+19+20&=&21+22+23+24\cr
&\vdots&\cr
n^2+(n^2+1)+\cdots+(n^2+n)&=&(n^2+n+1)+\cdots+(n^2+2n)\cr}} etex,
point 1/2 of bbox currentpicture shifted 13 down);
```

Consecutive sums of consecutive integers



$$
\begin{gathered}
1+2=3\\
4+5+6=7+8\\
9+10+11+12=13+14+15\\
16+17+18+19+20=21+22+23+24\\
\vdots\\
n^2+(n^2+1)+\cdots+(n^2+n)=(n^2+n+1)+\cdots+(n^2+2n)
\end{gathered}
$$

— Roger B. Nelsen

```
input paintball
numeric n; n = 6;
picture P[];
P1 = image(
    for i=1 upto n:
        for j=1 upto n:
            draw if j <= n+1-i: rball else: bball fi shifted (24i, 24j);
        endfor
    endfor
    draw (for i=1 upto n-1: (i, n+1-i) -- (i, n-i) -- endfor (n, 1))
        shifted (1/2, 1/2) scaled 24;

);
P2 = image(
    for i=1-n upto n:
        for j=1 upto n:
            if j < i+n+1:
                draw if i<1: rball else: bball fi shifted (24i, 24j);
            fi
        endfor
    endfor
    draw (origin -- (0, n)) shifted (1/2, 1/2) scaled 24;
);
draw P1 shifted 90 up;
draw P2 shifted 90 down;
label.rt("$\displaystyle \sum_{k=1}^n k + \sum_{k=1}^{n-1} k = n^2$", (160, 174));
label.rt("$\displaystyle \sum_{k=1}^n k + n^2 = \sum_{k=n+1}^{2n} k$", (160, -6));
```

Count the dots



$$\sum_{k=1}^{n} k + \sum_{k=1}^{n-1} k = n^2$$



$$\sum_{k=1}^{n} k + n^2 = \sum_{k=n+1}^{2n} k$$

— Warren Page

```
vardef trig(expr n, shade, edge) = image(
    for y = 1 upto n:
        for x = 1 upto n + 1 - y:
            path c; c = fullcircle shifted (2x-1, 2y-1) scaled 8;
            fill c withcolor shade;
            draw subpath (1/2, 3/2) of c
                 shifted - center c scaled 3/4 shifted center c
                 withcolor 3/4[shade, white];
            draw c;
        endfor
    endfor
    if edge:
        draw origin -- (16n, 0)
            for i = 1 upto n:
                -- 16(n + 1 - i, i) -- 16(n - i, i)
            endfor -- cycle;
    fi
)
enddef;

picture t; t = trig(5, Oranges 8 3, true);
picture P[];
P1 = image(draw t; draw t shifted 80 up; draw t shifted 80 right;
  draw trig(4, Blues 8 3, true) rotated 180 shifted (80, 80);
  label.bot("$3T_n + T_{n-1} = T_{2n}$",
    point 1/2 of bbox currentpicture shifted 8 down);
);
P2 = image(
  draw t rotated 180;
  draw t rotated 180 shifted 96 left;
  draw t rotated 180 shifted 96 down;
  draw trig(6, Greens 8 4, true) shifted -(96, 96);
  label.bot("$3T_n + T_{n+1} = T_{2n+1}$",
    point 1/2 of bbox currentpicture shifted 8 down);
);
P3 = image(draw t; draw t shifted 80 up; draw t shifted 80 right;
  draw trig(4, Blues 8 3, true) rotated 180 shifted (80, 80);
  draw t rotated 180 shifted (176, 176);
  draw t rotated 180 shifted ( 80, 176);
  draw t rotated 180 shifted (176, 80);
  draw trig(6, Greens 8 4, true) shifted (80, 80);
  label.bot("$T_{n-1} + 6T_n + T_{n+1} = (2n+1)^2$",
    point 1/2 of bbox currentpicture shifted 8 down);
);

label.ulft(P1, origin); label.urt (P2, origin); label.bot (P3, 21 down);
label.top("$T_n = 1 + 2 + \cdots + n$ \Longrightarrow",
  point 5/2 of bbox currentpicture shifted 13 up);
```

Identities for triangular numbers

$$T_n = 1 + 2 + \cdots + n \Longrightarrow$$



$3T_n + T_{n-1} = T_{2n}$



$3T_n + T_{n+1} = T_{2n+1}$



$T_{n-1} + 6T_n + T_{n+1} = (2n+1)^2$

```
numeric n, s; n = 5; s = 7;
path edge; edge = (origin for i=0 upto n-1:
    -- (-i, n-i) -- (-i-1, n-i)
  endfor -- (-n, 0) -- cycle) scaled s;

def make_trig(expr shade) = image(
  fill edge withcolor shade;
  for i=1 upto n:
    draw unitsquare xscaled (i-n-1) yscaled i scaled s withpen pencircle scaled 1/4;
  endfor
  draw edge;
) enddef;
picture t, u;
t = make_trig(Reds 8 2);
u = make_trig(Purples 8 2) rotated 180;
picture P[];
for k=1 step 2 until 7:
  P[k] = image(
    for i = 1 upto k:
      for j = 1 upto i:
        pair z; z = (1-j, 1-i) scaled n shifted (-floor(j/2), 0)
             if odd j: shifted (0, - floor (i/2))
             else: shifted (if odd i: 0 else: 1 fi, -floor((i-1)/2))
             fi scaled s;
        draw t shifted z;
        if odd j:
          if j > 1:
            draw u shifted z shifted (0, n*s + s)
          fi
        else:
          draw u shifted z shifted (0, n*s)
        fi;
      endfor
    endfor
    label.bot(
        if k=1: "$T_n$"
        else: "$" & decimal (k*k) & "T_n + " & decimal floor(k*k/8) &
              "=T_{" & decimal k & "n+" & decimal floor(k/2) & "}$"
        fi,
      point 1/2 of bbox currentpicture);
  );
endfor

draw P1; draw P3 shifted (70, 0); draw P5 shifted (175, -40); draw P7 shifted (300, -140);
label.bot("$\left(2k+1\right)^2T_n + T_k = T_{(2k+1)n+k}$",
    point 1/2 of bbox currentpicture shifted 21 down);
```

A triangular identity



$$(2k+1)^2 T_n + T_k = T_{(2k+1)n+k}$$

— Roger B. Nelsen

```
numeric u; u = 12;
path ring; ring = fullcircle scaled u; picture ball[];
ball1 = image(fill ring withcolor Spectral 4 1; draw ring);
ball2 = image(fill ring withcolor Spectral 4 2; draw ring);
ball3 = image(fill ring withcolor Spectral 4 3; draw ring);
ball4 = image(fill ring withcolor Spectral 4 4; draw ring);
picture P[]; interim bboxmargin := 10;
P1 = image(path H;
    H = (for i=0 upto 5: up rotated -60i -- endfor cycle) shifted down;
    draw ball1 shifted point 0 of H;
    for i=2 upto 5:
        path h; h = H scaled ((i-1) * 1.44 u);
        draw ball1 shifted point 1 of h;
        for j = 1 upto 4:
            for k = 1 upto i - 1:
                draw ball[j] shifted point j + k / (i-1) of h;
            endfor
        endfor
    endfor
    label.bot("$H_5$", point 1/2 of bbox currentpicture));
P2 = image(
    for i = 1 upto 9:
        for j = 1 upto i:
            draw ball[if i < 5: 4
                elseif (i > 5) and (j < i-4): 3
                elseif (i > 5) and (j > 5): 2
                else: 1 fi] shifted ((j - 1/2 i, -0.866025 i) * 1.44u);
        endfor
    endfor
    label.bot("$T_9$", point 1/2 of bbox currentpicture);
);
P3 = image(
    for i=1 upto 9:
        for j=1 upto 5:
            draw ball[if i < 6: if i + j < 7: 1 else: 2 fi
                          else: if i + j < 11: 3 else: 4 fi
                fi] shifted ((i, j) * 1.44u);
        endfor
    endfor
    label.bot("$5\times 9$", point 1/2 of bbox currentpicture);
);
interim labeloffset := 32;
label.ulft(P1, origin); label.urt(P2, origin); label.bot(P3, origin);
label.top(btex $
  \left.\vcenter{\openup 6pt\halign{$#$\hss&${}=#$\hss\cr
  H_n&1+5+\cdots+(4n-3)\cr
  T_n&1+2+\cdots+n\cr}}\right\}
  \Longrightarrow H_n = 3T_{n-1} + T_n = T_{2n-1} = n(2n-1)
  $ etex, point 5/2 of bbox currentpicture shifted 34 up);
```

Every hexagonal number is a triangular number

$$\left.\begin{aligned} H_n &= 1+5+\cdots+(4n-3) \\ T_n &= 1+2+\cdots+n \end{aligned}\right\} \Longrightarrow H_n = 3T_{n-1}+T_n = T_{2n-1} = n(2n-1)$$



H_5



T_9



5×9

```
numeric u; u = 16;
path dom; dom = unitsquare xscaled 2 scaled 2u;
numeric c; c = 0;
for n=1 upto 4:
  for k = 0 upto 3:
    for x = 0 upto n - 1:
      path d; d = dom shifted ((4x - 2n + 1, 2n-1) * u) rotated 90k;
      fill d withcolor if odd incr c: Blues else: Oranges fi [9][if odd n: 1 else: 2 fi];
      draw d;
    endfor
  endfor
endfor


label.top(btex
  \vbox{\openup 12pt\halign{\hss$\displaystyle #$\hss\cr
  1 + 4\times 2 + 8 \times 2 + 12 \times 2 + 16 \times 2 = 9^2\cr
  1 + 2 \sum_{k=1}^n \, 4k = \left(2n+1\right)^2\cr}}
  etex, point 5/2 of bbox currentpicture shifted 34 up);
```

One domino = two squares : concentric squares

$$1 + 4 \times 2 + 8 \times 2 + 12 \times 2 + 16 \times 2 = 9^2$$

$$1 + 2\sum_{k=1}^{n} 4k = (2n+1)^2$$



— Shirley A. Wakin

```
numeric u; u = 6;
def marked(expr i, N) =
  if (i = N) or (i mod 9 = 0): withpen pencircle scaled 3/4
  elseif i mod 3 = 0: withpen pencircle scaled 1/2
  else: withpen pencircle scaled 1/4 withcolor 1/2
  fi
enddef;
vardef frame(expr N, Closed) = image(
  for i = 0 upto N-1:
    for j = 0 upto N-1:
      path s; s = unitsquare shifted (i, j) scaled u;
      if j > i + floor(N / 2):
        if not Closed:
          s := s rotatedabout((N/2, N) scaled u, 180);
        fi
        fill s withcolor Blues 9 3;
      fi
    endfor
  endfor

  for i = 0 upto N:
    numeric minx; minx = if Closed or (i < floor(N / 2) + 1): 0 else: i - floor(N / 2) fi;
    numeric maxy; maxy = if Closed: N else: i + floor (N / 2) if i < N: + 1 fi fi;
    draw ((minx, 0) -- (N, 0)) shifted (0, i) scaled u marked(i, N);
    draw ((0, 0) -- (0, maxy)) shifted (i, 0) scaled u marked(i, N);
  endfor
  if not Closed:
    numeric m; m = floor (N / 2) + 1;
    draw ((0, m) -- (1, m)
      for i = 1 upto N - 1:
        -- (i, m+i) -- (i+1, m+i)
        hide(if (m + i) > N: draw ((i+1, m+i) -- (N, m+i)) scaled u marked(m+i, N); fi)
      endfor) scaled u withpen pencircle scaled 3/4;
  fi)
enddef;
for y = 0, -45u:
  numeric x; x = 0;
  for n=0 upto 3:
    picture f; f = frame(3**n, y=0); draw f shifted (x, y);
    x := x + 24 + xpart lrcorner f;
  endfor
endfor
label.bot("$1+9+\cdots+9^n = 1 + 2 + 3 + \cdots + (1+3+\cdots+3^n)$",
  point 1/2 of bbox currentpicture shifted 13 down);
```

Sums of consecutive powers of 9 are sums of consecutive integers



$$1 + 9 + \cdots + 9^n = 1 + 2 + 3 + \cdots + (1 + 3 + \cdots + 3^n)$$

— Roger B. Nelsen

```
numeric u; u = 16; path disc; disc = fullcircle scaled u;
vardef hexagon(expr N, Label) = image(
  numeric t; t = 1; fill disc withcolor Oranges 9 1; draw disc;
  for n = 1 upto N - 1:
    for r = 0 upto 5:
      for k = 1 upto n:
        path d; d = disc shifted (k/n)[(0, n*u) rotated 60r, (0, n*u) rotated (60r + 60)];
        fill d withcolor Oranges[8][n+1]; draw d; t := t + 1;
      endfor
    endfor
  endfor
  if Label:
    label.bot("$h_{" & decimal N & "} = " & decimal t & "$",
      point 1/2 of bbox currentpicture shifted 13 down);
  fi
) enddef;
picture P[];
P1 = image(
  draw hexagon(1, true);
  draw hexagon(2, true) shifted (5u, 1u);
  draw hexagon(3, true) shifted (12u, 2u);
  draw hexagon(4, true) shifted (21u, 3u);
);
input isometric_projection
set_projection(20, -45); ipscale := u;
picture ocube, halfbox; ocube = cube(Oranges 9 3, Oranges 9 1, Oranges 9 2);
halfbox = image(
  for i=1 upto 5: for j=1 upto 5:
    draw ocube shifted p(0, i, 5-j);
  endfor endfor
  for i=1 upto 4: for j=1 upto 5:
    draw ocube shifted p(i, 1, 5-j);
  endfor endfor
  for i=1 upto 4: for j=1 upto 4:
    draw ocube shifted p(i, j+1, 4);
  endfor endfor);

P2 = image(
  label.lft(hexagon(5, false), 40 left);
  label("\large $\iff$", origin);
  label.rt(halfbox, 40 right);
);

label.top(P1, 13 up); label.bot(P2, 13 down);

label.bot("$h_n = n^3 - (n-1)^3$", point 1/2 of bbox currentpicture shifted 13 down);
label.bot("$\therefore \quad h_1 + h_2 + \cdots + h_n = n^3$",
    point 1/2 of bbox currentpicture shifted 13 down);
```

Sums of hex numbers are cubes



$h_1 = 1$ $h_2 = 7$ $h_3 = 19$ $h_4 = 37$



$$h_n = n^3 - (n-1)^3$$

$$\therefore \quad h_1 + h_2 + \cdots + h_n = n^3$$

Sequences and series

```
vardef f(expr x) = (x, 1/256 mlog(1 + x)) enddef;
vardef fp(expr x) = if x=-1: up else: (1, 1/(1 + x)) fi enddef;

path ff;
ff = (f(-1/2){fp(-1/2)} for x=0 step 1/2 until 3: .. f(x){fp(x)} endfor) scaled 120;

interim bboxmargin := 0;
path xx, yy;
xx = subpath(0, 1) of bbox ff shifted (0, -ypart llcorner ff);
yy = subpath(0, -1) of bbox ff shifted (-xpart llcorner ff, 0);

path m[];
z0 = (xpart point 0 of xx, xpart point 0 of xx);
m0 = z0 -- -2z0;

z1 = point 5.5 of ff;
z2 = point 4 of ff;

m1 = xx scaled 2 rotated angle z1
     cutbefore subpath (-1, 1) of bbox ff cutafter subpath (1,3) of bbox ff;
m2 = xx scaled 2 rotated angle z2
     cutbefore subpath (-1, 1) of bbox ff cutafter (point 1 of m0 -- point 1 of m1);

draw m0 dashed evenly withcolor 1/2;
draw m1 withcolor 1/2;
draw m2 withcolor 1/2;
draw ff withcolor 2/3 red;
drawarrow xx;
drawarrow yy;

draw z1 -- (x1, 0) dashed withdots scaled 1/4;
draw z2 -- (x2, 0) dashed withdots scaled 1/4;

forsuffixes @=origin, z1, z2:
      draw @ withpen pencircle scaled dotlabeldiam;
    undraw @ withpen pencircle scaled 3/4 dotlabeldiam;
endfor

dotlabel.bot("$\frac1n$", (x1, 0));
dotlabel.bot("$\frac1{n+1}$", (x2, 0));

label.top("$y=x$", point 1 of m0);
label.top("$y=m_1x$", point 1 of m1);
label.top("$y=m_2x$", point 1 of m2);
label.lrt("$y=\ln(1+x)$", point 6.5 of ff) withcolor 2/3 red;
label.rt("$x$", point 1 of xx);
label.top("$y$", point 1 of yy);

label.bot(btex
\vbox{\openup 12pt\halign{\hfill $#$&\quad\Longrightarrow\quad $\displaystyle #$\hfill\cr
n \ge 1 & m_1 < m_2 < 1\cr
&\frac{\ln(1+1/n)}{1/n} < \frac{\ln(1+1/(n+1))}{1/(n+1)} < 1\cr
&\left(1+\frac1n\right)^n < \left(1+\frac1{n+1}\right)^{n+1} < e\cr}} etex,
point 1/2 of bbox currentpicture shifted 34 down);
```

A monotone sequence bounded by e



$$\begin{aligned} n \ge 1 \quad &\Longrightarrow \quad m_1 < m_2 < 1 \\ &\Longrightarrow \quad \frac{\ln(1+1/n)}{1/n} < \frac{\ln(1+1/(n+1))}{1/(n+1)} < 1 \\ &\Longrightarrow \quad \left(1+\frac{1}{n}\right)^n < \left(1+\frac{1}{n+1}\right)^{n+1} < e \end{aligned}$$

— Roger B. Nelsen

```
numeric u; u = 32;
vardef f(expr x) = 256 x / mlog(x) enddef;
path ff, xy, xx, yy;
numeric minx, s; minx = 39/32; s = 1/8;
ff = ((minx, f(minx)) for x = minx + s step s until 8: .. (x, f(x)) endfor) scaled u;
xx = 5 left -- (xpart point infinity of ff, 0);
yy = 5 down -- (0, ypart point 0 of ff);
xy = origin -- 5(u, u);

numeric e; e = 2.718281828459;
numeric x[];  x0 = 7; for i=1 upto 4: x[i] = f(x[i-1]); endfor
drawoptions(dashed evenly scaled 1/2 withpen pencircle scaled 1/4 withcolor 1/4);
  draw (u, 0) -- (u, ypart point 0 of ff);
  draw ((e, 0) -- (e, e) -- (0, e)) scaled u;
  draw ((x0, 0) -- (x0, f(x0))) scaled u;
  draw ((x1, 0) -- (x1, f(x1))) scaled u;
  draw ((x2, 0) -- (x2, f(x2))) scaled u;
drawoptions(dashed withdots scaled 1/4 withpen pencircle scaled 1/2 withcolor 1/2 blue);
  draw ((x0, f(x0)) for i=1 upto 3: -- (x[i], x[i]) -- (x[i], f(x[i])) endfor) scaled u;
drawoptions(withcolor Reds 8 7);
  draw ff; label.top("$y=\frac{x}{\ln(x)}$", point infinity of ff);
drawoptions(withcolor Oranges 8 7);
  draw xy; label.top("$y=x$", point infinity of xy);
drawoptions();
  drawarrow xx; label.rt("$x$", point 1 of xx);
  drawarrow yy; label.top("$y$", point 1 of yy);

numeric dx, dy; dx = -2.8; dy = -10;
draw TEX("$1$")   shifted ( 1 * u + dx, dy);
draw TEX("$e$")   shifted ( e * u + dx, dy);
draw TEX("$x_0$") shifted (x0 * u + dx, dy);
draw TEX("$x_1$") shifted (x1 * u + dx, dy);
draw TEX("$x_2$") shifted (x2 * u + dx, 2dy);
draw TEX("$x_3$") shifted (x3 * u + dx, 3dy);
draw TEX("$\vdots$") shifted (x4 * u + dx, 4.2dy);

draw TEX("$e$")   shifted (dy, e * u + dx);

label.bot(btex $\displaystyle
  x_0 > 1 \enspace\mathbin{\&}\enspace x_{n+1} = \frac{x_n}{\ln(x_n)}
  \quad\mathbin{\Longrightarrow}\quad \lim x_n = e
  $ etex, point 1/2 of bbox currentpicture shifted 34 down);
```

A recursively defined sequence for e



$$x_0 > 1 \quad \& \quad x_{n+1} = \frac{x_n}{\ln(x_n)} \quad \Longrightarrow \quad \lim x_n = e$$

— Thomas P. Dence

```
vardef do_boxes(expr u, r, n, t) = image(
  save a, b; path a, b;
  a = unitsquare xscaled r   yscaled 1 scaled u shifted -(u,u);
  b = unitsquare xscaled (1-r) yscaled r scaled u shifted point 1 of a;
  for i=1 upto n:
    draw a withpen pencircle scaled 1/4;
    draw b withpen pencircle scaled 1/4;
    numeric sf; sf = sqrt(r**(i-1));
    if sf > 1/4: if r = 1/2:
      label(TEX("$\frac12^" & decimal 2i & "$") scaled sf, center b);
      label(TEX("$\frac12" if i>1: & "^" & decimal (2i-1) fi & "$") scaled sf, center a);
    else:
      label(TEX("$r(1-r)" if i>1: & "^" & decimal (2i-1) fi & "$") scaled sf, center b);
      label(TEX("$r" if i>1: & "(1-r)^" & decimal (2i-2) fi & "$") scaled sf, center a);
    fi fi
    a := a scaled (1-r);
    b := b scaled (1-r);
  endfor
  draw unitsquare scaled -u;
  label.bot(t, point 1/2 of bbox currentpicture shifted 13 down);
)
enddef;

picture P[];

P1 = do_boxes(144, 1/2, 6, "$\frac12 + \frac14 + \frac18+ \cdots = 1$");
P2 = do_boxes(160, 1/3, 9, "$r + r(1-r) + r(1-r)^2 + \cdots = 1$");

label.top(P1, 7 up);
label.bot(P2, 7 down);
```

Geometric sums



$$\frac{1}{2} + \frac{1}{4} + \frac{1}{8} + \cdots = 1$$



$$r + r(1-r) + r(1-r)^2 + \cdots = 1$$

— Warren Page

```
input arrow_label
picture P[];
numeric a, r, u;
2a = u = 160; r = 0.81;
pair x;
x = origin;
path b[];
for i=0 upto 21:
    b[i] = unitsquare xscaled a yscaled u scaled (r**i) shifted x;
    x := point 1 of b[i];
endfor
x1 = 0; z1 = whatever[point 2 of b0, point 2 of b1];
y2 = 0; z2 = whatever[point 2 of b0, point 2 of b1];
path t; t = subpath (3, 2) of b1 -- point 2 of b0 -- cycle;

P1 = image(
    fill t withcolor Oranges 8 2;
    for i = 0 upto 21:
        fill b[i] withcolor Blues 8 1;
        %draw subpath (1, 3) of
        draw b[i];
        if i < 6:
            label.top("$" &
            if i=0: "a" &
            elseif i=1: "ar" &
            elseif (i>1) and (i<5): "ar^" & decimal i &
            elseif i=5: "\cdots" &
            fi "$", point 1/2 of b[i] shifted 16 down);
        fi
        if i < 5:
            label.top("$+$", point 1 of b[i] shifted 16 down);
        fi
    endfor
    label.lft("$1$", point -1/2 of b0);
    draw origin -- z1 -- z2 -- cycle withpen pencircle scaled 5/8;
    arrow_label(origin, z1, "$\frac1r$", -20);
);
P2 = image(
    fill t withcolor Oranges 8 2;
    draw t;
    label.bot("$ar$", point 1/2 of t);
    label.lft("$1-r$", point -1/2 of t);
);


draw P1;
draw P2 shifted (a, a);

label.bot(btex \vbox{\openup13pt\halign{\hss$\displaystyle #$&$\displaystyle {}=#$\hss\cr
\frac{a + ar + ar^2 + ar^3 + ar^4 + \cdots}{1/r}&\frac{ar}{1-r}\cr
 a + ar + ar^2 + ar^3 + ar^4 + \cdots &\frac{a}{1-r}\cr
\sum_{n=0}^\infty ar^n&\frac{a}{1-r}\cr
}} etex, point 1/2 of bbox currentpicture shifted 55 down);
```

Geometric series I



$$\frac{a + ar + ar^2 + ar^3 + ar^4 + \cdots}{1/r} = \frac{ar}{1-r}$$

$$a + ar + ar^2 + ar^3 + ar^4 + \cdots = \frac{a}{1-r}$$

$$\sum_{n=0}^{\infty} ar^n = \frac{a}{1-r}$$

— J. H. Webb

```
numeric r, u;
u = 180; r = 0.561;
pair P, Q, R, S, T;
S = origin; P = u * up; R = (u, u * r); Q = (u, u);
T = whatever[P, R]; ypart T = 0;
path pg; pg = origin -- (u, 0) -- (u, u*r) -- (0, u) -- cycle;
draw R -- Q -- P dashed evenly scaled 1/2;
for i=0 upto 12:
    draw pg;
    if i=0:
        label.rt("$1$", point -1/2 of pg);
        label.bot("\strut $1$", point 1/2 of pg );
    elseif i=1:
        label.rt("$r$", point -1/2 of pg);
        label.bot("\strut $r$", point 1/2 of pg);
    elseif i < 4:
        label.rt("$r^"&decimal i &"$", point -1/2 of pg);
        label.bot("\strut $r^"&decimal i &"$", point 1/2 of pg);
    elseif i = 4:
        label.bot("\strut $\cdots$", point 1/2 of pg);
    fi
    pg := pg shifted - point 0 of pg scaled r shifted point 1 of pg;
endfor

dotlabel.ulft("$P$", P);
dotlabel.urt ("$Q$", Q);
dotlabel.urt ("$R$", R);
dotlabel.llft("$S$", S);
dotlabel.lrt ("$T$", T);
label.top("$1$", 1/2[P, Q]);
label.rt ("$1-r$", 1/2[R, Q]);

label.bot(btex \vbox{\openup 13pt\halign{\hss $\displaystyle #$ \hss\cr
\triangle PQR \sim \triangle TSP\cr
\therefore\quad 1 + r + r^2 + r^3 + \cdots = \frac{1}{1-r}\cr}} etex,
point 1/2 of bbox currentpicture shifted 21 down);
```

Geometric series II



$$\triangle PQR \sim \triangle TSP$$

$$\therefore \quad 1 + r + r^2 + r^3 + \cdots = \frac{1}{1-r}$$

— Benjamin G. Klein and Irl C. Bivens

```
numeric u; u = 120; picture P[];
P1 = image(
  for i=1 upto 9:
    path s; s = unitsquare scaled u shifted -2(u, u) scaled (1/2 ** i);
    fill s withcolor Oranges 8 3;
    draw s;
    draw s shifted (point 1 of s - point 0 of s);
    draw s shifted (point 3 of s - point 0 of s);
    if i < 4:
      picture t; t = TEX("$1/" & decimal (2 ** i) & "$") scaled (1/i);
      label.top(t, point 1/2 of s - (0, 1/2i));
      label.top(t, point 1/2 of s + (1/2 ** i * u, - 1/2 i));
    fi
  endfor
  label.bot("$\displaystyle \frac14+\frac1{16}+\frac1{64}+\frac1{256}+\cdots=\frac13$",
    point 1/2 of bbox currentpicture shifted 13 down);
);
P2 = image(numeric u, r; u = 160; r = .6;
  for i = 0 upto 9:
    path s, t;
    s = unitsquare scaled (u * (1-r)) shifted -(u, u) scaled (r ** i);
    t = point 1 of s -- (0, ypart point 1 of s)
                     -- (0, ypart point 2 of s) -- point 2 of s -- cycle;
    fill s withcolor Greens 8 3; draw s;
    draw t; draw t rotatedabout(center s, 90);
    if i < 3:
      picture q, p;
      if i=0:
        q = TEX("$1-r$"); p = TEX("$r$");
      elseif i=1:
        q = TEX("$r(1-r)$") scaled 0.8; p = TEX("$r^2$") scaled 0.8;
      else:
        q = TEX("$r^2(1-r)$") scaled 0.45; p = TEX("$r^3$") scaled 0.45;
      fi
      label.top(q, point 1/2 of s - (0, 1/2i));
      label.top(p, point 1/2 of t - (0, 1/2i));
    fi
  endfor
  label.bot(btex
  \vbox{\openup 13pt\halign{\hss $\displaystyle # + \cdots$&$\displaystyle {}=#$\hss\cr
  (1-r)^2 + r^2(1-r)^2 + r^4(1-r)^2 & \frac{(1-r)^2}{(1-r)^2 + 2\times r(1-r)}\cr
  1 + r^2 + r^4 & \frac{1}{(1 - r)^2 + 2r(1-r)} = \frac{1}{1-r^2} \cr
  a + ar + ar^2 & \frac{a}{1-r} \cr}}
  etex, point 1/2 of bbox currentpicture shifted 13 down);
);
label.top(P1, 7 up); label.bot(P2, 7 down);
```

Geometric series III



$$\frac{1}{4}+\frac{1}{16}+\frac{1}{64}+\frac{1}{256}+\cdots=\frac{1}{3}$$



$$(1-r)^2+r^2(1-r)^2+r^4(1-r)^2+\cdots=\frac{(1-r)^2}{(1-r)^2+2\times r(1-r)}$$

$$1+r^2+r^4+\cdots=\frac{1}{(1-r)^2+2r(1-r)}=\frac{1}{1-r^2}$$

$$a+ar+ar^2+\cdots=\frac{a}{1-r}$$

— Sunday A. Ajose

```
numeric u, v;
u = 72; v = 305;
numeric n; n = 5;
draw (u, v) -- (u, 0) -- (0, 0) -- (0, v + 21);
draw (n * u, 0) -- (n * u, v / (n + 1));

for x = 1 upto n + 1:
    numeric y; y = (x-1)/x;
    if x <> n - 1:
        draw (0, v) -- (x * u, 0) withcolor 2/3 blue;
        draw (0, v * y) -- (u, v * y) dashed withdots scaled 1/2;
        draw (left--right) scaled 1/2 shifted (x, 0) scaled u;
    else:
        draw (left--right) scaled 1/4 shifted (x, 0) scaled u dashed withdots;
    fi
    if x < n - 1:
        label.lft("$" & decimal (x-1) & "/" & decimal x & "$", (0, v * y));
        label.bot("\strut $" & decimal x & "$", (x * u, 0));
    elseif x = n:
        label.lft("$(n-1)/n$", (0, v * y - 2));
        label.bot("\strut $n$", (x * u, 0));
    elseif x = n + 1:
        label.lft("$n/(n+1)$", (0, v * y + 2));
        label.bot("\strut $n+1$", (x * u, 0));
    fi
endfor
label.lft("$1$", (0, v));

vardef label_bar(expr a, b, z, t) =
    label.rt(t, 1/2[a, b] + z);
    drawarrow (z {left} .. origin {left}
        cutafter fullcircle scaled dotlabeldiam) shifted 1/2[a, b]
        withpen pencircle scaled 1/4;
enddef;

label_bar((n * u, 0), (n * u, v / (n + 1)), (1/2u, 1/4u),
  "$\displaystyle\frac{1}{n+1}$");
label_bar((u, (n-1)*v/n), (u, n*v/(n+1)), (1/2u, 1/8u),
  "$\displaystyle\frac{1}{n}\cdot\frac1{n+1}$");

label.bot(btex $\displaystyle
  \frac1{1\cdot2} + \frac1{2\cdot3} + \cdots + \frac1{n(n+1)} = \frac{n}{n+1}
  $ etex, point 1/2 of bbox currentpicture shifted 21 down);
```

Sum of reciprocals of successive integer products



$$\frac{1}{1 \cdot 2} + \frac{1}{2 \cdot 3} + \cdots + \frac{1}{n(n+1)} = \frac{n}{n+1}$$

— Roman W. Wong

```
path xx, yy, ff;
numeric u, v, s, minx, maxx; s = 1/8; minx = 1; maxx = 7; u = 42; v = 102;
ff = ((minx - s, 2/(minx - s)) for x = minx step s until maxx + 2s:
       .. (x, 2/x)
     endfor) xscaled u yscaled v;
xx = origin -- (maxx + 4s, 0) scaled u;
yy = origin -- (0, 2+s) scaled v;

for x = minx upto maxx - 1:
  path a, b, c; a = unitsquare xscaled u yscaled (-2v / (x * (x + 1)));
  b = a shifted (x * u , 2v /x);
  c = a shifted (-2u, 2v/x);

  forsuffixes @=b, c:
    fill @ withcolor 1/2[Oranges[8][x], white]; draw @;
  endfor
  draw point 0 of b -- (x * u, 0) dashed evenly scaled 1/2;
  draw (up--down) scaled 3/2 shifted (x * u, 0);
  if x = maxx - 1:
    label.top(btex \vbox{\openup4pt\halign{\hss $##$ \hss\cr
      1\cdot\left(\frac2n-\frac2{n+1}\right)\cr
      =\frac2{n(n+1)}\cr}} etex, point 1/2 of b shifted 8 up);
    label.bot("\strut $n$", (x * u, -3));
    label.bot("\strut $n+1$", (x * u + u, -3));
    draw (up--down) scaled 3/2 shifted (x * u + u, 0);
    draw point 1 of b -- (x * u + u, 0) dashed evenly scaled 1/2;
  elseif x = maxx - 2:
    label.top("$\cdots$", point 1/2 of b);
    label.bot("\strut $\cdots$", (x * u, -3));
  else:
    label.top("$\frac1{" & decimal (1/2 x * (x + 1)) & "}$", point 1/2 of b);
    label.bot("\strut $" & decimal x & "$", (x * u, -3));
  fi
endfor
z0 = (xpart point 0 of c, 0);
z1 = (xpart point 1 of c, 0);
draw point 0 of c -- z0 dashed evenly scaled 1/2;
draw point 1 of c -- z1 dashed evenly scaled 1/2;
draw 1.1[z0, z1] -- 1.1[z1, z0];

for y=0 upto 2:
  draw (left--right) scaled 3/2 shifted (0, y * v);
  label.lft("$" & decimal y & "$", (0, y * v));
endfor


draw ff withcolor Reds 8 7; label.top("$y=\frac2x$", point 0 of ff) withcolor Reds 8 8;
drawarrow xx; label.rt("$x$", point 1 of xx);
drawarrow yy; label.top("$y$", point 1 of yy);


label.top("$\frac11+\frac13+\frac16+\frac1{10}+\cdots+\frac2{n(n+1)}+\cdots=2$",
  point 5/2 of bbox currentpicture shifted 34 up);
```

Sum of reciprocals of triangular numbers

$$\frac{1}{1}+\frac{1}{3}+\frac{1}{6}+\frac{1}{10}+\cdots+\frac{2}{n(n+1)}+\cdots=2$$



— Roger B. Nelsen

```
numeric u; u = 300;
path ff; ff = ((15/16, 16/15) for x = 1 step 1/8 until 9/4: .. (x, 1/x) endfor) scaled u;
path xx; xx = (xpart point 0 of ff, 0) -- (xpart point infinity of ff, 0);
path yy; yy = (point 0 of xx -- point 0 of ff) shifted 21 left;


vardef gcd(expr a, b) = if b = 0: a else: gcd(b, a mod b) fi enddef;
vardef reduced_fraction(expr N, D) =
  save n, d, g; numeric n, d, g; g = gcd(N, D); n = N / g; d = D / g;
  if d = 1: "$" & decimal n & "$"
  else: "$\frac{" & decimal n & "}{" & decimal d & "}$" fi
enddef;
vardef fh(expr a) =
  save s, t; string s; numeric t;
  for n = 1 upto 8:
    t := 1/n - 1/(n+1);
    s := "$" if n>1: & "\frac1" fi & decimal n & "-" & "\frac1" & decimal (n+1) & "$";
    exitif abs(a-t) < eps;
  endfor s
enddef;

vardef partition(expr a, b, c, level) =
  if level > 0:
    save box; path box;
    save w, h; numeric w, h; w = abs(a-b); h = 1/b - c;
    if h > 0:
      box = unitsquare xscaled w yscaled h shifted (a, c) scaled u;
      fill box withcolor 1/4[Oranges[9][level], white]; draw box;
      if w >= 1/4: label(fh(w*h) , center box); fi
    fi
    partition(a, a + 1/2 w, c + h, level - 1);
    partition(a + 1/2 w, b, c + h, level - 1);
  fi
enddef;
partition(1, 2, 0, 6);

for x = 1 step 1/4 until 2:
  pair a, b; a = (x,0) scaled u; b = (xpart point 0 of yy, u / x);
  draw (up--down) scaled 3 shifted a;
  draw (left--right) scaled 3 shifted b;
  label.bot("\strut" & reduced_fraction(4x, 4), a shifted 4 down);
  label.lft("\strut" & reduced_fraction(4, 4x), b shifted 4 left);
endfor

draw ((1, 0) -- (1, 1)) scaled u; draw ((2, 0) -- (2, 1/2)) scaled u;
draw (left--right) scaled 3 shifted point 0 of yy;
label.lft("$0$", point 0 of yy shifted 4 left);

draw ff withcolor Reds 8 7;
label.ulft("$y=\frac1x$", point infinity of ff shifted 8 up) withcolor Reds 8 8 ;
drawarrow xx; label.rt("$x$", point 1 of xx);
drawarrow yy; label.top("$y$", point 1 of yy);
% ... plus the TeX labels, top and bottom
```

Alternating harmonic series



$\frac{1}{2} \times \left(\frac{2}{3} - \frac{1}{2}\right) = \frac{1}{3} - \frac{1}{4}$; $\quad \frac{1}{4} \times \left(\frac{4}{5} - \frac{2}{3}\right) = \frac{1}{5} - \frac{1}{6}$; $\quad \frac{1}{4} \times \left(\frac{4}{7} - \frac{1}{2}\right) = \frac{1}{7} - \frac{1}{8}$; $\quad \cdots$

$$\ln 2 = \int_1^2 \frac{1}{x}\,dx = \left(1 - \frac{1}{2}\right) + \left(\frac{1}{3} - \frac{1}{4}\right) + \cdots = 1 - \frac{1}{2} + \frac{1}{3} - \frac{1}{4} + \cdots$$

— Mark Finklestein

```
path xx, yy, arc; numeric r;
r = 300; xx = 30 left -- (20+r) * right; yy = xx rotated 90; arc = quartercircle scaled 2r;
numeric n, theta; n = 4; theta = 8;
for i = 0 upto n:
  z[i] = point (2i+1) * theta / 45 of arc;
endfor

vardef pin@#(expr p, o, z) =
    draw z+o..z withpen pencircle scaled 1/4; label@#(p, o+z);
enddef;

for i = 0 upto n:
  draw origin -- z[i]; draw (0, y[i]) -- z[i] dashed evenly;
  path t;
  if i > 0:
    t = z[i] -- (x[i], y[i-1]) -- z[i-1] -- cycle;
    if i <> n - 1:
      pin.urt("$2\sin\theta$", 8 unitvector(direction 5/2 of t rotated -90), point 5/2 of t);
      picture p;
      p = thelabel.lft("$2\sin\theta\cos"
      & if i = n: "2n" else: decimal 2i fi
      & "\theta$", point 1/2 of t);
      unfill bbox p; draw p;
    else:
      label.lft("$\dots$", point 1/2 of t);
    fi
  else:
    t = z0 -- (x0, 0);
    pin.urt("$\sin\theta$", 8 dir 10, point 1/2 of t);
  fi
  draw t;
endfor

draw arc withcolor Reds 8 7; drawarrow xx; drawarrow yy;
drawoptions(withcolor Blues 8 7);
label("$\theta$", 1/4 r * dir 1/2 theta);
for i=1 upto n:
  picture t;
  t = thelabel(if i+1=n: "$\dots$" else: "$2\theta$" fi, 1/4 r * dir (2i * theta));
  unfill bbox t; draw t;
endfor
label("$2\theta$", 48 dir (270 +  theta) shifted point 3/45 theta of arc);
label("$4\theta$", 32 dir (270 + 2theta) shifted point 5/45 theta of arc);
label("$\dots$",   22 dir (270 + 3theta) shifted point 7/45 theta of arc);
label("$2n\theta$",22 dir (270 + 4theta) shifted point 9/45 theta of arc);
drawoptions();
labeloffset := 8;
dotlabel.llft("$0$", origin);
dotlabel.bot("$1$", (r, 0));
dotlabel.lft("$\sin\,(2n+1)\theta$", (0, r * sind((2n+1)*theta)));

label.top(btex $\displaystyle
  \sin\,(2n+1)\theta = \sin\theta + 2\sin\theta \sum_{k=1}^{n} \cos 2k\theta
  $ etex, point 5/2 of bbox currentpicture shifted 21 up);
```

Sum of sines

$$\sin{(2n+1)\theta} = \sin\theta + 2\sin\theta \sum_{k=1}^{n} \cos 2k\theta$$



— J. Chris Fisher & E. L. Koh

Miscellaneous

```
numeric a, b, c, d; a = 233; b = 52; c = 85; d = 164;
z0 = origin; z1 = (a, 0); z2 = (a, d); z3 = (0, d); z4 = z2 + (c, 0);
z5 = (a, b); z6 = z2 + (c, b); z7 = (c, d); z8 = z2 + (0, b);

z9 = whatever[z5, z6]; y9 = y2;
z11 = whatever[z5, z6]; y11 = y0;
z10 = whatever[z6, z7]; x10 = x2;
z12 = whatever[z6, z7]; x12 = x0;

path t[];
t1 = z0 -- z11 -- z5 -- cycle; t5 = z11 -- z1 -- z5 -- cycle;
t2 = z7 --  z9 -- z6 -- cycle; t6 =  z9 -- z4 -- z6 -- cycle;
t3 = z0 -- z12 -- z7 -- cycle; t7 = z12 -- z7 -- z3 -- cycle;
t4 = z5 -- z10 -- z6 -- cycle; t8 = z10 -- z6 -- z8 -- cycle;
t0 = z2 -- z9 -- z6 -- z10 -- cycle;

fill t1 withcolor Oranges 8 3;  fill t7 withcolor Reds 8 3;
fill t2 withcolor Oranges 8 2;  fill t8 withcolor Reds 8 3;
fill t3 withcolor Blues 8 3;    fill t5 withcolor Greens 8 3;
fill t4 withcolor Blues 8 2;    fill t6 withcolor Greens 8 3;
draw t1;  draw t3;  draw t5;  draw t7;
draw t2;  draw t4;  draw t6;  draw t8;

fill t0 withcolor 1/2[Oranges 8 2, Blues 8 2]; draw t0;

dotlabel.llft("$(0, 0)$", z0);
dotlabel.lrt ("$(a, 0)$", z1);
dotlabel.llft("$(a, d)$", z2);
dotlabel.ulft("$(0, d)$", z3);
dotlabel.lrt("$(a+c, d)$", z4);
dotlabel.top("$(a+c, b+d)$", z6);
dotlabel.top("$(a, b+d)$", z8);

numeric s; s = 1/8;
picture ad, bc, pg, t;
ad = image(draw (z0--z1--z2--z3--cycle) scaled s);
bc = image(draw (z2--z4--z6--z8--cycle) scaled s);
pg = image(draw (z0--z5--z6--z7--cycle) scaled s);
t = btex $\left|\,
  \vcenter{\halign{\hss$#$\hss&\quad\hss$#$\hss\cr a&b\cr c&d\cr}}
  \,\right|=ad-bc={}$ etex;
t := image(draw t; label.rt(ad,        point 3/2 of bbox t));
t := image(draw t; label.rt("${}-{}$", point 3/2 of bbox t));
t := image(draw t; label.rt(bc,        point 3/2 of bbox t));
t := image(draw t; label.rt("${}={}$", point 3/2 of bbox t));
t := image(draw t; label.rt(pg,        point 3/2 of bbox t));
label.bot(t, point 1/2 of bbox currentpicture shifted 55 down);
```

A 2×2 determinant is the area of a parallelogram



$\begin{vmatrix} a & b \\ c & d \end{vmatrix} = ad - bc =$  $-$  $=$ 

— Solomon W. Golomb

```
numeric a, b, c, d; a = 144; d = 3/4 a; c = 1/3 a; b = 1/4 a;
path xx, yy; xx = 4 left -- (a+c) * right; yy = 4 down -- (b+d) * up;
z1 = whatever[(a,b), (a+c, b+d)]; y1 = d;
z2 = whatever[(a,b), (a+c, b+d)]; y2 = 0;
z3 = (a, 0) - z2;
picture P[];
P1 = image(
    draw (c,d) -- z1 -- z2 dashed withdots scaled 1/2;
    drawarrow xx; drawarrow yy;
    draw origin -- (a,b) -- (a+c, b+d) -- (c, d) -- cycle
        withpen pencircle scaled 3/4 withcolor Blues 8 8;
    dotlabel.rt("$(a, b)$", (a, b));
    dotlabel.ulft("$(c, d)$", (c, d));
);
P2 = image(
    draw z3 -- (a, 0) -- (a, d) -- (x3, d) -- cycle dashed withdots scaled 1/2;
    drawarrow xx; drawarrow yy;
    draw origin -- z2 -- z1 -- (c,d) -- cycle
        withpen pencircle scaled 3/4 withcolor Blues 8 8;
    dotlabel.rt("$(a, b)$", (a, b));
    dotlabel.ulft("$(c, d)$", (c, d));
);
P3 = image(
    fill unitsquare xscaled (c-x3) yscaled b shifted z3 withcolor Blues 8 2;
    drawoptions(dashed withdots scaled 1/2);
    draw (0, b) -- (a, b);
    draw (c, 0) -- (c, d) -- origin;
    drawoptions();
    drawarrow xx; drawarrow yy;
    draw z3 -- (a, 0) -- (a, d) -- (x3, d) -- cycle
        withpen pencircle scaled 3/4 withcolor Blues 8 8;
    dotlabel.rt("$(a, b)$", (a, b));
    dotlabel.ulft("$(c, d)$", (c, d));
);
P4 = image(
    fill unitsquare xscaled x3 yscaled (d-b) shifted (0, b) withcolor Blues 8 2;
    draw (x3, b) -- (x3, d) dashed withdots scaled 1/2;
    drawarrow xx; drawarrow yy;
    draw (0, d) -- (a, d) -- (a, 0) -- (c, 0) -- (c, b) -- (0, b) -- cycle
        withpen pencircle scaled 3/4 withcolor Blues 8 8;
    label.bot("$a$", (a, 0));  label.lft("\llap{$b$}", (0, b));
    label.bot("$c$", (c, 0));  label.lft("\llap{$d$}", (0, d));
);

interim labeloffset := 24;
label.ulft(P1, origin); label.urt(P2, origin);
label.llft(P3, origin); label.lrt(P4, origin);

label.top(btex \vbox{\halign{\hss\vrule width 0pt depth 12pt # \hss\cr
The area of the parallelogram determined by vectors $(a,b)$ and $(c,d)$ is\cr
$\left|\,\vcenter{\halign{\hss$#$\hss&\quad\hss$#$\hss\cr a&b\cr
c&d\cr}}\,\right|=\pm(ad-bc)$\cr}} etex, point 5/2 of bbox currentpicture);
```

Area of parallelogram

The area of the parallelogram determined by vectors (a, b) and (c, d) is

$$\begin{vmatrix} a & b \\ c & d \end{vmatrix} = \pm(ad - bc)$$



— Yihnan David Gau

```
numeric u; u = 144;
vardef f(expr x) = 7/8 (x*x*x - x) enddef; numeric minx, maxx, s; -minx = maxx = 5/4; s = 1/8;
path ff; ff = ((minx,f(minx)) for x = minx+s step s until maxx: .. (x, f(x)) endfor) scaled u;
% move it to a convenient location, so axes can go through origin
ff := ff shifted - 1.14 point 0 of bbox ff shifted 20 up;
path xx; xx = 8 left -- right scaled 1.1 xpart (urcorner ff - llcorner ff);
path yy; yy = xx rotated 90 cutafter (left--right) shifted (0, 1.1 ypart urcorner ff);

numeric a, b, a', b'; a = 1.4; a' = 3.7; b = length ff - 1/4 - a; b' = length ff - 1/4 - a';
z0 = point a  of ff; z1 = point b  of ff; z2 = point a' of ff; z3 = point b' of ff;

z4 = whatever[z2, z3]; z5 = whatever[z2, z3];
x4 = xpart point 0 of ff; x5 = xpart point infinity of ff;

x0 = x6 = x8; y6 = y2; x1 = x7 = x9; y7 = y3;
z8 = whatever[z4, z5]; z9 = whatever[z4, z5];
z10 = whatever[z4, z5] = whatever[z6, z7];

% dotlabels.top(0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10); % <--- in case of need

fill z2--z6--z8--cycle withcolor Oranges 8 2;
fill z3--z7--z9--cycle withcolor Blues 8 2;

fill z10 -- z2 -- z6 -- cycle withcolor Oranges 8 3;
fill z10 -- z3 -- z7 -- cycle withcolor Blues 8 3;

label.bot("\strut $a$",  (x8, 0)); draw (x8, 0) -- z8;
label.bot("\strut $b$",  (x7, 0)); draw (x7, 0) -- z7;
label.bot("\strut $a'$", (x2, 0)); draw (x2, 0) -- z2 dashed evenly;
label.bot("\strut $b'$", (x3, 0)); draw (x3, 0) -- z3 dashed evenly;

draw z2 -- z6 -- z7 -- z3;
drawoptions(withcolor Blues 8 7);
draw z4--z5; label.rt("$y=h(x)$", z5);
drawoptions(withcolor Oranges 8 7);
draw ff; label.top("$y=f(x)$", point infinity of ff);
drawoptions();
drawarrow xx; label.rt("$x$", point 1 of xx);
drawarrow yy; label.top("$y$", point 1 of yy);

picture mark; mark = image(forsuffixes $=left, right:
    draw (up--down) scaled 2 rotated -20 shifted 1/2 $;
endfor);
draw mark shifted (1/2(x0+x2), ypart point 0 of xx);
draw mark shifted (1/2(x1+x3), ypart point 0 of xx);

vardef pindotlabel@#(expr t, z, o) =
    draw z -- z+o cutbefore fullcircle scaled 2 dotlabeldiam shifted z
        withpen pencircle scaled 1/4;
    label@#(t, z+o); draw z withpen pencircle scaled dotlabeldiam;
enddef; interim dotlabeldiam := 2; interim labeloffset := 2;
pindotlabel.top ("$(a, h\kern-0.4pt(a))$",     z8, 12 dir 106);
pindotlabel.lft ("$(a, f\kern-0.4pt(a'))$",    z6,  9 dir 186);
pindotlabel.top ("$(a'\!, f\kern-0.4pt(a'))$", z2, 20 dir 76);
pindotlabel.top ("$(b, f\kern-0.4pt(b'))$",    z7, 14 dir 50);
pindotlabel.rt  ("$(b, h\kern-0.4pt(b))$",     z9, 12 dir 20);
pindotlabel.llft("$(b'\!, f\kern-0.4pt(b'))$", z3, 18 dir 240);

label.top(btex $\displaystyle
\frac12 \left(b-a\right)\left(f(a') + f(b')\right) =
\frac12 \left(b-a\right)\left(h(a) + h(b)\right)
$ etex, point 5/2 of bbox currentpicture shifted 42 up);
```

Gaussian quadrature as the area of either trapezoid

$$\frac{1}{2}(b-a)\left(f(a')+f(b')\right)=\frac{1}{2}(b-a)\left(h(a)+h(b)\right)$$



— Mike Akerman

```
input isometric_projection
set_projection(100/3, -45);
picture blue_cube; blue_cube = cube(Blues 8 4, Blues 8 2, background);
picture white_cube; white_cube = cube(background, background, background);

path case; picture nice_case;
case = p(0,1,0)--p(0,1,5)--p(0,6,5)--p(-5,6,5)--p(-5,6,0)--p(-5,1,0)--cycle;
nice_case = image(draw case withpen pencircle scaled 3/2; draw case withcolor 7/8);

picture P[]; P0 = image(
  draw p(-5,6,3) -- p(-4,6,3) -- p(-4,6,5) -- p(-4,5,5);
  draw p(-5,6,4) -- p(-4,6,4) -- p(-4,5,4) -- p(-4,5,5) -- p(-3,5,5) -- p(-3,6,5);
  draw nice_case;
);

vardef make_pattern(expr cube) =
  % grid on "walls"
  for i = 0 upto 5:
    draw p(-i, 1, 0) -- p(-i, 1, 5) -- p(-i, 6, 5);
    draw p(0, 1, i) -- p(-5, 1, i) -- p(-5, 6, i);
    draw p(-5, i+1, 0) -- p(-5, i+1, 5) -- p(0, i+1, 5);
  endfor
  % draw the cubes
  save x, z; numeric x, z;
  x = -4; z = 4;
  for k = 5, 5, 5, 5, 3,
          5, 5, 5, 4, 3,
          5, 5, 1, 1, 0,
          4, 1, 0, 0, 0,
          3, 0, 0, 0, 0:
    for y=1 upto k: draw cube shifted p(x, y, z); endfor
    z := z - 1;
    if z < 0:
      z := 4;
      x := x + 1;
    fi
endfor enddef;
P1 = image(make_pattern(white_cube); draw nice_case);
P2 = image(make_pattern(blue_cube); draw nice_case);

draw P0 shifted 160 up;
draw P1 shifted 100 left;
draw P2 shifted 100 right;
```

The problem of the calissons



— Guy David and Carlos Tomei