


1 Moorish


These designs are largely taken from the chapter on Moorish tiling patterns in Owen Jones' *Grammar of Ornament*, 1856.

Bats



```
input jones-colors
beginfig(1);
  path bat;
  bat = (0,0)--(1,1)--(5,1)--(3,3)--(1,3)--(1,5)--(0,4)--
        (-1,5)--(-1,3)--(-3,3)--(-5,1)--(-1,1)--cycle;
  vardef unit(expr shade) = image(for t=0, 180:
    fill bat scaled 3.6 rotated t withcolor shade;
  endfor) enddef;
  picture tile[]; tile0 = unit(navy); tile1 = unit(forest) rotated 90;
  numeric n, u, v; n = 4; (u,v) = urcorner tile0 - llcorner tile0;
  for i=-n upto n:
    for j=-n upto n:
      draw tile[(i+j) mod 2] shifted (i*u,j*v);
    endfor
  endfor
  picture p; p = currentpicture; fill bbox p withcolor ivory; draw p;
endfig;
```

Carré




```
input jones-colors
beginfig(1);
  path lozenge, triangle, box;
  lozenge = (left -- down -- right -- up -- cycle)
    yscaled 1/3 rotated -45 scaled 21;
  interim bboxmargin := 0; box = bbox lozenge;
  triangle = point 0 of box + 4 right -- point 1 of lozenge
    -- point 0 of box + 4 up -- cycle;

  vardef paint_tile(expr s) = image(
    fill box withcolor ivory;
    fill triangle withcolor gold;
    fill triangle rotated 180 withcolor gold;
    fill lozenge withcolor s;
  ) enddef;

  picture tile[];
  tile0 = paint_tile(lapiz);   tile2 = paint_tile(navy);
  tile1 = paint_tile(forest); tile3 = paint_tile(lapiz);

  (u, v) = urcorner tile0 - llcorner tile0;
  for i = 0 upto 11:
    for j = 0 upto 7:
      t := (floor (i/2) mod 2) + 2(floor (j/2) mod 2);
      draw tile[t] if odd (i+j): rotated 90 fi shifted (i*u, j*v);
    endfor
  endfor
endfig;
```

Cross




```

input jones-colors
beginfig(1);
  path cross;
  numeric r, t; t = 2; r = t-sqrt(2);
  cross = (for i=0 upto 3:
    ((t,0) -- (t-r, r) -- (r, r) -- (r, t-r)) rotated 90i --
  endfor cycle) scaled 9.6;

  pair u, v;
  u = point 0 of cross - point 8 of cross;
  v = point 4 of cross - point 12 of cross;
  n = 4;
  for i=-n upto n:
    for j=-n upto n:
      k := j + floor ((i + abs(j)) / (2 abs(j) + 1));
      fill cross shifted (i*u+j*v)
        withcolor if odd k: forest else: navy fi;
      draw cross shifted (i*u+j*v)
        withpen pencircle scaled 3 withcolor 1/2[ivory, white];
    endfor
  endfor
  picture P; P = currentpicture; fill bbox P withcolor ivory; draw P;
endfig;

```

Hex-star




```

input jones-colors
beginfig(1);
  numeric r; r = sqrt(1/3);
  path t; t = (r * dir 0 {dir 60} .. {dir 120} 2r * dir 60 {dir 240}
    .. r * dir 120 {dir 180} .. {dir 240} 2r * dir 180 {dir 360}
    .. r * dir 240 {dir 300} .. {dir 360} 2r * dir 300 {dir 120}
    .. cycle) rotated -30 scaled 21;
  numeric a; (a, whatever) = t intersectiontimes (origin -- infinity * up);
  path h; h = for i=0 upto 5: point a of t rotated 60i -- endfor cycle;
  path t'; t' = t rotatedabout(point 1 of t, 60);
  path h'; h' = h rotatedabout(point 1 of t, 60);
  path s; s = for i=0 upto 5:
    point i of h' -- 1/3[point i of h', point i+2 of h'] --
  endfor cycle;
  vardef unit(expr a, b) = image(
    fill t withcolor a; fill h withcolor ivory;
    fill t' withcolor ivory; fill s withcolor b;
    drawoptions(withpen pencircle scaled 1 withcolor 15/16);
    draw t; draw t'; draw h; draw s;
    drawoptions();
  ) enddef;
  pair u, v; u = origin rotatedabout(point 1 of t, 120); v = u rotated -60;
  numeric n; n = 6;
  for i = -n upto n:
    for j = -n upto n:
      draw unit(if odd (i+j): navy else: forest fi,
        if odd (i+j): gold else: lapiz fi)
        shifted (i*u + j*v);
    endfor
  endfor
  clip currentpicture to unitsquare shifted -(1/2,1/2) scaled 5in yscaled 0.618;
endfig;

```

Lifebelt



```

input jones-colors
beginfig(1);
    numeric u; u = 21;
    path s, t;
    s = unitsquare scaled u shifted 1/2(3u, -u) rotated 45;
    t = subpath (2,3) of s -- subpath (0,1) of s rotated 90 -- cycle;

    picture unit;
    unit = image(
        for i=0 upto 3:
            fill s rotated 90i withcolor navy;
            filldraw t rotated 90i withcolor gold;
        endfor
        drawoptions(withpen pencircle scaled 3/2 withcolor 15/16);
        for i=0 upto 3:
            draw s rotated 90i;
            draw subpath (1,2) of t rotated 90i;
        endfor
    )

```